
Introduction


I wrote this book because I’m excited about Elm’s potential to change how we approach web development. In some ways, it’s already done that by inspiring people to create JavaScript libraries like Redux and Vuex. Elm reminds me of the early days of Ruby on Rails with its focus on developer happiness, and its key features – functional programming, a static type system, immutable data, and the Elm architecture – allow me to write more expressive and more reliable code. 


If you are reading this book, I’m going to assume that you are comfortable with HTML, CSS and JavaScript, and you are already somewhat familiar with Elm and excited about its potential to redefine how we develop web applications. I assume that you’ve already learned that Elm is a functional language with immutable data and a mechanism for dealing with side effects, and you’ve gone through the Elm guide and experimented with some Elm code. I also assume that you’re familiar with the Elm architecture based on the data model, the view, and the update function. 


This book isn’t going to walk you through the Elm installation instructions, basic language syntax, and other preliminaries, because they are adequately covered by other sources. My goal is to help you get comfortable in the intermediate territory – the vast and always under-documented area where you need to write code to solve real world problems but don’t yet have enough experience to figure everything out by yourself. 


As Elm is growing in popularity, books, courses and a profusion of blog posts are becoming available, which is good to see. However, the vast majority of resources are still focused on the basics, and I found that working on real world projects very often leads me down a rabbit hole of mailing list threads and GitHub issue discussions. 


In this text, I want to focus on the practical aspects of Elm development like tools and dealing with bugs, and I’ll spend a large portion of the book walking you through writing a small but realistic application. My goal is to go beyond the basics and to provide you with solutions – or at least pointers for further research – for problems you might encounter when doing real work with Elm. 


Since this isn’t a beginner book, it requires some familiarity with Elm and its core libraries, and consequently the basics of functional programming.


Conversely, I should mention that some more advanced topics are beyond the scope of this book. For example, we are not going to look at code profiling and performance optimisation, or server side rendering, or the considerations for writing Elm packages. 


Resources for Newcomers to Elm


If you would like to read Practical Elm but all you know about Elm is that it’s a kind of tree that grows in the Northern Hemisphere, don’t worry: it will not take a lot of effort to get to the point where this book becomes useful to you. 


To become familiar with Elm, I recommend reviewing the Elm syntax and the Elm guide, as well as spending a bit of time writing basic programs based on the guide before reading this book. 


Once you go through these resources and experiment a little bit with writing Elm code, you can get back to this book to go beyond the basics.








Capabilities of Elm: Beyond Basic Syntax


Overview


Before we get into building an application, we need to build a bit of a foundation. In this chapter, we will look at some of the capabilities provided by Elm, and the functional programming techniques that it affords us. 


We will talk about functions: higher-order functions, combinators, point-free style and closures. We will also look at the various ways of combining multiple functions calls in a single expression. We will consider a couple of interesting features of record types. We will take a look at operators, and at some facets of using types. Lastly, you will learn about the pattern matching and destructuring options available in Elm. 


A Few Useful Bits of Functional Programming Theory


I’d like to highlight several functional programming concepts, both because they are ubiquitous and because being cognisant of them allows you to write better code. These are:



  	Higher-order functions

  	Combinators

  	Point-free style

  	Closures.




Higher-order functions and combinators


Functions which take one or more other functions as some of their arguments, or functions that return a function are called higher-order functions. 


If a higher-order function doesn’t have any free variables – that is, its definition consists only of a combination of its arguments, then it’s called a combinator. Note that it means it cannot rely on any functions other than whatever functions are passed in as arguments. 


An example of a combinator is the compose function: 



compose f g x = f (g x)

compose text Debug.toString [Json, Xml, PlainText]
-- [Json,Xml,PlainText] : String






The reason I bring this up is that thinking about your code in terms of combinators can help you come up with better abstractions which help you construct the solution from small generic pieces.  


When working with higher-order functions, you may sometimes find it useful to define type aliases for function types. For example, this signature is not very easy to decipher:



generatePlanTree : (Plan -> msg) -> (Plan -> msg)
    -> (String -> Decoder PlanJson -> PlanJson)
    -> Html msg






We can define a couple of aliases to make it more palatable: 



type alias MouseEventHandler msg = Plan -> msg
type alias PlanParser = String -> Decoder PlanJson -> PlanJson

generatePlanTree : MouseEventHandler msg -> MouseEventHandler msg
    -> PlanParser
    -> Html msg






Point-free style


Related to this is the so called point-free style, which means writing a function definition in a way that omits one or more argument names. “Point” in “point-free” actually stands for “argument” for historical reasons.


It is enabled by the fact that all functions are curried (meaning that they can be partially applied). Suppose we have a function to get a list of user IDs:



getIds users = List.map (\u -> u.userId) users






As we can partially apply map, we can just omit the users argument instead, and we can also pass the field accessor function directly to map:



getIds = List.map .userId 






This version of getIds is still a function that takes a list of users and returns a list of user IDs, except now there are no arguments being passed explicitly.


Similarly to combinators, point-free style can help you think in terms of reusable pieces. However, it’s best not to get carried away. While many functions can be rewritten to produce a point-free version, it’s not necessarily going to be more readable, especially if you combine lots of function calls, so please keep readability in mind as well. 


Closures


Once higher-order functions are in play, it becomes possible to create closures. A closure is a function together with a mapping of all its free variables (in Elm, it means the variables which are not arguments) to their values at the time of creating this function. 


It can seem a bit counterintuitive in a pure language where all variables are actually constants, but it’s still possible to create closures in Elm:



isSelectedUser : Int -> User -> Bool
isSelectedUser selectedUserId user = .userId user == selectedUserId

displayUsers currUser users = 
    let 
        isCurrentUser : User -> Bool
        isCurrentUser = isSelectedUser currUser.userId
    in
        ...






isCurrentUser in the let block is a function that takes a user and returns True if the user is the current user. The ID of the current user isn’t passed as an argument: its value is enclosed from the function’s environment – in this case, the partially applied isSelectedUser function.


Another way to create a closure would be to return a function:



stringRepeater n =
   \s -> String.repeat n s 






The function returned by stringRepeater is a closure which carries the value of n from its surrounding context.


Using Functions


Now I’d like to show you a few tricks to do with functions. 


Pipe operators


One of the potential readability issues with FP code is that because it consists entirely of function calls, you can end up with a huge number of parentheses: 



text (toString (List.filter canAddUsers (getActiveUsers projectId users)))






Elm provides two pipe operators, more properly called function application operators, to help you avoid too much resemblance to Lisp in your code:



(|>) : a -> (a -> b) -> b
(<|) : (a -> b) -> a -> b






Both of the pipe operators are equivalent to plain function application:



x |> f == f x
f <| x == f x






However, because they have different precedence and associativity, they allow us to write expressions without parentheses for the most part. The backward function application operator passes an argument to a function on its left hand side: 



text 
    <| toString 
    <| List.filter canAddUsers 
    <| getActiveUsers projectId users






Here, we’re producing a text element with the string returned by toString, which is in turn given the filtered list of active users. 


The forward function application operator passes an argument to the function on its right hand side: 



users 
    |> getActiveUsers projectId 
    |> List.filter canAddUsers
    |> toString
    |> text






We can read it like this: we take the active users, then filter them, then convert the resulting list to a string, and finally pass it to the text function. 


Depending on the particular expression and perhaps what you are used to, either backward or forward operator will produce more readable code. These operators are used practically everywhere in Elm so you will see a lot of them in this book. 


Function composition operators


Similarly to function application, there are two function composition operators: 



(<<) : (b -> c) -> (a -> b) -> a -> c
(>>) : (a -> b) -> (b -> c) -> a -> c






Again, the idea is to choose one that results in more readable code. For example, we can filter a list of users to find those that aren’t allowed to create new users in two ways:



List.filter (not << canAddUsers) [Engineer, Driver, Foreman, OfficeManager]

List.filter (canAddUsers >> not) [Engineer, Driver, Foreman, OfficeManager]






The first line with not on the left reads more naturally in this case.


Argument order


If you’re coming to Elm from JavaScript or another object oriented language, it’s worth pointing out that programming with curried functions requires a different order of arguments from what you may be used to. 


The main or the most “varying” argument, so to speak, should be placed last in the function definition. To give an example, List.map takes a function first and the list it’s applied to second. Maybe.withDefault takes the default value first, and the actual Maybe value second. 


In imperative and object-oriented programming, you would normally do the opposite. For example, in Underscore.js or Lodash (both JavaScript libraries), the map function takes the array as its first argument and the function to apply follows:



_.map([1, 2, 3], (num) => num * 3)






However, that is rather inflexible. Placing the main argument last facilitates building reusable blocks out of partially applied functions, and it also makes them easier to write in a point-free style. For example:



isOverThreshold threshold n = n > threshold
getOutliers = List.filter <| isOverThreshold 100

getOutliers [1, 2, 200, 3, 400, 4]






The argument order in both isOverThreshold and List.filter makes it easier for me to write getOutliers in a point-free style by partially applying both of those functions. If, say, List.filter took the list as its first argument, it wouldn’t work out quite so conveniently.


Flipping arguments


At times, we really do need to apply arguments in a different order. Consider the issue of checking a user’s set of permissions to verify that they are allowed to access a particular page or piece of functionality. For instance, we can have multiple user types, and we might want to list the subset of users who are allowed to add new users: 



type User 
    = Driver
    | Engineer
    | Foreman
    | OfficeManager
    | ProjectManager
    | Surveyor

-- ...

List.filter canAddUsers [Engineer, Driver, Foreman, OfficeManager]






In canAddUsers, we need to check the user type against the list of types which are allowed to add users:



canAddUsers u = List.member u [Foreman, OfficeManager, ProjectManager]






We can rewrite canAddUsers in point-free style using the flip combinator: 



canAddUsers = flip List.member [Foreman, OfficeManager, ProjectManager]






flip converts the function into a version with swapped arguments: 



flip : (a -> b -> c) -> b -> a -> c
flip f b a =  
    f a b






This combinator was provided by the Elm core library prior to Elm 0.19, but now you have to implement it yourself, or get it from some package. 


Using Records


There are a couple of interesting things about record types beyond the basic syntax.


Constructors


The name of a record type is referenced in two contexts. One is the name of a type alias, and the other one is a constructor function used to create values of this type: 



type alias User = { userId: Int, name: String }

User 1 "John"  -- User is a constructor function






For each record type alias, the compiler automatically generates a constructor function of the same name. For our User alias, it’s a function that takes an integer and a string, and returns a record:



User : Int -> String -> User






The constructor is a regular function so it can be partially applied:



User 1
-- <function> : String -> User






Using extensible records to restrict arguments


Elm has the concept of extensible records: 



type alias GenericNode a =
    { a | actualLoops : Int
    , actualRows : Int
    , actualStartupTime : Float
    , actualTotalTime : Float
    ...
    }

type alias CteNode 
    = GenericNode 
    { nameAlias : String
    }






In this example, a CteNode is a record with all the fields of GenericNode plus an additional alias field.


At first glance, this might look sort of like inheritance where you construct more specific types by extending generic ancestors. However, it’s not really meant to be used in this fashion, as hinted by the fact that type aliases for extensible records don’t get a constructor generated for them. The preferred solution in Elm is to use a nested record field for shared fields.


The main purpose of this syntax is actually to restrict arguments taken by functions. The reason this is useful is that it allows you to write simple functions focused on a specific task, which makes it easier to comprehend code and to debug it. 


For example, I could have a function which appends an alias name to the node title. Using extensible records, I can make it clear that this function works specifically with a node title and an alias:



type alias NodeTitle = String
type alias Aliased n = { n | nameAlias : String }

appendAlias : NodeTitle -> Aliased n -> NodeTitle 
appendAlias title {nameAlias} = 
    title ++ " (" ++ nameAlias ++ ")"






This function will only work with records which have the nameAlias field, but it’s also clear from the type annotation that it will not touch anything other than the alias when it’s passed a record containing other fields in addition to the alias field. This results in improved readability of the code. 


Operators as Functions


Operators are in fact functions in disguise, and the infix notation is just syntactic sugar. Any operator can be used as a function if it’s wrapped in parentheses: 



(*) 10 3 
-- Produces 30

(::) "a" ["b", "c"]
-- Produces ["a", "b", "c"]






Since they are functions, we can partially apply them, which is handy for using them with higher-order functions:



List.map ((*) 2) [1, 2, 3]
-- Produces [2, 4, 6]






Using Types


With regards to types, there are several things worth understanding beyond the basics:



  	Recursive types

  	Special type variables 

  	Phantom types

  	Opaque types.




I will show the first three in this section, but defer the discussion of opaque types till later in the book in order to be able to introduce them in the context of the sample application.


Recursive types


It’s possible to define recursive types. For example, we could define a generic Tree type:



type Tree a
    = Empty
    | Node a (List (Tree a))






Each node has a value and a list of child nodes. The type of child nodes is recursively defined as Tree a. We could create a value of this type like this: 



tree =
    Node 1 
        [ Node 2 []
        , Node 3 
            [ Node 4 []
            , Node 5 [] 
            ] 
        ]






This value represents the following binary tree: 



   1
  / \
 2   3
    / \
   4   5






Special type variables


Normally when we define a generic type, we use some lowercase name like a or msg as a type placeholder: 



map : (a -> msg) -> Cmd a -> Cmd msg






Elm has three special placeholder names which denote constrained sets of types and can be used instead of a regular placeholder: 



  	number

  	appendable

  	comparable




For example, we could constrain the tree type we defined before to only work with appendable types: 



type Tree appendable 
    = Empty
    | Node appendable (List (Tree appendable))






If I try to use this type with Int which is not appendable, the compiler will complain: 



tree : Tree Int 
tree = Node 1 []
-- Compile error






The only appendable types are strings and lists. The only number types are Int and Float. The set of comparable types consists of Int, Float, Char, String, lists, and tuples.


It’s possible to use the same kind of placeholder more than once in a type signature. Consider Set.map from the Elm core library:



-- Set
map : (comparable -> comparable2) -> Set comparable -> Set comparable2
map func set = 
    fromList (foldl (\x xs -> func x :: xs) [] set)






This function produces a new set from the input set by applying func to elements of set. Since set elements have to be comparable, func takes a value of a comparable type as its argument, and it also has to return a value of a comparable type. However, the input and output types are not necessarily the same, so it doesn’t make sense to write  (comparable -> comparable). In order to allow these sorts of functions, number, appendable and comparable can be suffixed with an alphanumeric sequence without changing their meaning. 


Phantom types


A phantom type is a generic type with one or more type variables which are not used in any of the constructors. For example: 



type Unit tag value = Unit value






While value is used in the constructor, tag isn’t, which makes Unit a phantom type. The interesting thing is that there is a way to use tag in order to get the compiler to detect incompatibility between values we want to distinguish.


Let’s define a couple of types for distance units which we will use as tags: 



type KmTag = KmTag
type MileTag = MileTag






So how do we use the Unit constructor? For that, we need to define corresponding functions: 



km : number -> Unit KmTag number
km = Unit

mile : number -> Unit MileTag number
mile = Unit






Note that both of them are simply aliasing the Unit constructor, but the key is their return types which are using different tags (KmTag and MileTag). Now, if we pass a value created with mile to a function that expects a value tagged with KmTag, we will get a compilation error:



showKm : Unit KmTag Float -> String
showKm (Unit d) = String.fromFloat d

distance = mile 12.34

s = showKm distance  -- Compilation error







Type Mismatch

The 1st argument to `showKm` is not what I expect:

53| s = showKm distance  -- Compilation error
               ^^^^^^^^
This `distance` value is a:

    Unit MileTag Float

But `showKm` needs the 1st argument to be:

    Unit KmTag Float






So the compiler is able to distinguish between these values, even though at runtime values created with either km or mile are the same boxed integers or floats. 


At first blush, this might seem like a really cool and useful concept, however in practice the situations where you might need phantom types are few and far between. In most cases, all you need is distinct types and perhaps also opaque types if you’re trying to constrain how values are created. 


For example, you might have account IDs and user IDs which are both integers but can’t be used interchangeably. In this situation, you don’t need phantom types, it’s enough to define two distinct types: 



type AccountId = AccountId Int
type UserId = UserId Int






However, if you want to distinguish between values which have the same runtime representation, and you have a shared set of operations you want to perform on all the different types of these values, then it’s a use case for phantom types. Consider what we’d have to do to perform calculations on our km values, such as addition:



type alias Km = Unit KmTag Float

addKm : Km -> Km -> Km
addKm (Unit d1) (Unit d2) = km (d1 + d2)






For every operation we’d like to do (addition, subtraction etc.), we need to write a function to unwrap the arguments, perform the operation on raw numbers and wrap the result back up. Then we would have to write all the same functions for distances expressed in miles. Of course, this would be tedious and error-prone, so in this case there’s a benefit in using phantom types. Even so, you may not need to roll your own implementation, and instead use the elm-tagged package which implements tagged types along with a number of convenience functions. 


Another use case for phantom types is when you want to use a particular runtime representation for reasons of performance or memory use, but still want compile-time enforcement of the semantics of different types. This approach is in development in the elm-css library, whereby various values are represented as boxed strings at runtime (using the phantom type type Value a = Value String) but are tagged with different record types at compile time in order to make sure that the values are only generated using the appropriate functions. As a result, you will get a compiler error if you try to pass a size value (generated with px 100 or similar) to a colour style, for instance. 


Wildcard Match, Destructuring, and Pattern Matching


Elm provides a number of language level facilities for working with data values. 


These are:



  	Wildcard match

  	Destructuring

  	Pattern matching




Wildcard match


The underscore (_) means a wildcard match in Elm, and it can appear both in function definitions and in case expressions.


For instance, the function called always from the Elm core library can be defined like this: 



always : a -> b -> a
always x _ = x






Using the underscore instead of a named argument signifies that the argument is ignored. always can then be used with map like this:



List.map (always 0) [1, 2, 3]

-- [0, 0, 0] : List Int






The underscore is also used in the default branch of case expressions: 



describeNum n = 
    case n of 
        0 -> 
            "none"
        1 -> 
            "one"
        _ -> 
            "many"






The difference between a regular variable and underscore is that underscore doesn’t create a binding between the value and the name, which means that you can have multiple underscores in a single expression: 



type Node = Table SchemaName TableName | Cte CteName

nodeType : Node -> String
nodeType node = 
    case node of 
        Table _ _ -> 
            "Table"
        Cte _ -> 
            "CTE"






Destructuring


The distinction between destructuring and pattern matching is somewhat blurred in Elm, because these things frequently go together. When I talk about destructuring, I mean binding parts of a composite value to separate identifiers: 



let 
    (firstName, lastName) = ("John", "Doe")
in
lastName   
-- Doe : String






Destructuring works in function definitions:



second (_, snd) = snd

second (0, 1)
-- 1 : number






Records can be destructured too, but in this case you have to use matching field names rather than arbitrary variable identifiers. For example, this length function will work on any records which have fields called x and y of suitable types:



length { x, y } = sqrt <| toFloat <| x^2 + y^2






It can be applied to 2D vectors which just have x and y as well as 3D vectors with x, y and z fields. However, if you provide a type annotation for this function with a specific type alias, then the function will only work with that particular record type:



type alias Vector = { x : Int, y : Int }

length : Vector -> Float
length { x, y } = sqrt <| toFloat <| x^2 + y^2

length { x = 1, y = 2, z = 3 }  -- compile error






If you want to have a type annotation that lets it work with different record types, you need to write it using the extensible record syntax:



length : { r | x : Int, y : Int } -> Float
length { x, y } = sqrt <| toFloat <| x^2 + y^2

length { x = 1, y = 2, z = 3 }  -- OK






Alternatively, you can use a parametrised alias to achieve the same result: 



type alias Vector r = { r | x : Int, y : Int }

length : Vector r -> Float
length { x, y } = sqrt <| toFloat <| x^2 + y^2






Additionally, it’s possible to name the whole record while destructuring some of the fields: 



displayNode ({ nodeType, relationName } as node) =
    case ( nodeType, relationName ) of 
        ( "Seq Scan", "projects" ) -> 
            highlightNode Color.blue node
        ( _, "projects" ) -> 
            highlightNode Color.green node
        _ -> 
            displayPlainNode node






In the above example, we can refer both to nodeType and relationName fields as well as the whole record (as node), which is useful because we need to pass the whole record to other functions. Note that the parentheses around the argument are required, otherwise the compiler will complain. I’ll talk about how this case statement works in a minute, when we get to pattern matching. 


Beyond tuples and records, you can even do nested destructuring for tuples within tuples or records within tuples: 



isWithinBounds ( maxX, maxY, { x, y } ) =
    x < maxX && y < maxY

isWithinBounds ( 10, 10, { x = 9, y = 9 } )






One more use of destructuring is to extract the carried value from a type:



type User = User String

userId : User -> String
userId (User u) = u

userId (User "A")
-- A : String






This also works in a let expression: 



userName : User -> String
userName user = 
    let 
        (User name) = user
    in 
        String.toLower name






The last thing I’d like to point out here is that anonymous functions work just like regular functions with regards to destructuring: 



( "John", "Doe" )
    |> \( a, b ) -> a ++ " " ++ b
-- John Doe : String

{ first = "John", last = "Doe" } 
    |> \{ first, last } -> first ++ " " ++ last
-- John Doe : String

(\(User u) -> u) (User "A")
-- A : String






Pattern matching


Pattern matching is a mechanism for choosing the branch of code to execute based on the type or value of a given expression. Elm provides only one construct for pattern matching: the case expression. It is nonetheless a fairly versatile tool because it can be used to match custom types (also known as union types), values and lists. 


Let’s look at a few examples to see how this works. The most frequent situation is where you need to take different actions based on different values of a custom type, like handling different messages in the update function: 



case msg of
    ChangePlanText s ->
        ( { model | currPlanText = s }, Cmd.none )

    SubmitPlan ->
        ( { model | currPage = DisplayPage }, Cmd.none )

    _ ->
        ( model, Cmd.none )






While matching a particular value, you can also name the data it contains. In this example, ChangePlanText carries a string which we name s and then use it to update the model. The last branch is a catch-all: it will match any value not matched by the preceding branches. 


We are also able to match values of built-in types: 



case userCount of
    0 -> 
        "nobody"
    1 -> 
        "1 person"
    _ -> 
        "2 or more people"






We can even combine custom type and contained value checks:



case userCount of 
    Just 0 -> 
        "No users at the moment"
    Just 1 -> 
        "Only 1 person"
    Just _ -> 
        "2 or more people"
    Nothing -> 
        "No users ever"






By employing tuples, we can also handle more complex logic. Here is the case statement for highlighting nodes based on their properties which I showed before:



displayNode ({nodeType, relationName} as node) =
    case (nodeType, relationName) of 
        ("Seq Scan", "projects") -> 
            highlightNode Color.blue node
        (_, "projects") -> 
            highlightNode Color.green node
        _ -> 
            displayPlainNode node






By constructing a tuple out of the destructured record fields, we can choose to do different things based on the combination of the two field values. If the node type is “Seq Scan” and the relation name is “projects”, then we highlight in blue. If it’s any other node type for the “projects” relation, then we highlight in green, while any other combination of values doesn’t get highlighted. This is a compact way to express conditional logic. 


Finally, we can use the case expression to pattern match lists: 



describeNode node = 
    case node.children of
        [] -> 
            "Leaf node"
        [ _ ] -> 
            "1 child node"
        [ _, _ ] -> 
            "2 child nodes"
        _ -> 
            "Multiple child nodes"






In this example, the pattern is the number of elements in the list: zero, one or two.


Alternatively, lists can be pattern matched using the cons operator: 



describeNode node = 
    case node.children of
        [] -> 
            "Leaf node"
        [ a ] -> 
            "1 child node: " ++ a
        a::b::_ -> 
            "2 or more child nodes"






The last branch matches lists with two or more elements.


Summary


In this chapter, we looked into the capabilities of the Elm language and a few functional programming concepts. 


We talked about some of the functional programming concepts and techniques made available by Elm: higher-order functions, combinators, point-free style and closures. You learned some of the specifics of working with functions and records, as well as operators and types. Finally, we talked about the tools available for working with data at the language level and handling conditional logic: wildcard matches, destructuring, and pattern matching.








Building an Application: Tools, UI, JSON Parsing


Overview


In order to put the knowledge of Elm to use, we are going to dive into building an application in this chapter. 


I’ve chosen a PostgreSQL EXPLAIN output visualiser as an example, both to avoid the ubiquitous to-do lists and picture sharing apps, and to allow me to demonstrate how to deal with JSON parsing. JSON parsing is one of the most common difficulties people have with Elm. 


PostgreSQL is a relational database. It has an EXPLAIN command, which shows a query plan used by the database to execute any given SQL statement. This is very useful when a query runs too slowly and needs to be optimised. Analysing the query plan allows the developer or database administrator to see how PostgreSQL goes about producing the result for a given query, and where the bottlenecks are happening. Then steps can be taken to improve the performance, perhaps by adding indexes or by rewriting the query to avoid expensive operations. 


However, query plans can be quite extensive for non-trivial queries, in which case reading the raw output of the EXPLAIN command becomes rather difficult. Hence our visualiser will take the raw output supplied by the user (in JSON format) and show something easier to analyse. 


We’ll start by looking at some of the tools available to make the process of creating applications more convenient. Then we will begin building out the UI of the query execution plan visualiser. We will add some interactivity and use that to learn about handling events. Finally, we will start processing realistic data, which means that we will need to start breaking up our growing code base into chapters, and learn how to parse JSON data.


Tools


While Elm projects are simple enough that you can get started with just a text editor, there are several tools that can help you be more productive. 


Ellie


Ellie is an online Elm editor that allows you to write small programs and run them right in your browser. It also allows you to share them via a link. 


It gives you the ability to import packages and customise the HTML your Elm program is embedded into. 


It’s an excellent tool for exploration and experimentation, even when you’re already experienced with Elm.  



  elm-json



Even though its name suggests something to do with JSON, elm-json is actually an essential tool for managing dependencies in your elm.json files.


You can install it from NPM: 



npm install -g elm-json






elm-json new allows you to create elm.json for a new project (either an application or a package). 


elm-json install will add a dependency to your file, for example:



elm-json install elm/http






It’s also possible to add a specific version of a package like elm/http@2.0.0.


This tool doesn’t actually download dependencies though – that’s still a job for elm itself. 


The most useful feature is the ability to upgrade dependencies, albeit only for applications (packages aren’t supported yet). When you run elm-json upgrade, it executes its version constraint solver to work out how to bump patch and minor version of the dependences in a mutually compatible way. It’s also possible to upgrade major versions by adding the --unsafe flag.


Finally, elm-json tree will show you the entire dependency graph in tree form. 


These are the main commands, but there are more features such as support for test dependencies and uninstalling packages. 



  elm-live



The standard problem of web development is the change-reload-retry cycle: you make a change, then you have to reload the page and then get the application back into the state you want to test. While Elm includes elm-reactor by default, it doesn’t do live reloading when you change the source, and has other limitations. elm-live is a tool which provides live reloading functionality. 


You can install it from NPM:



npm install -g elm-live






Then you start it in your project directory:



elm-live src/Main.elm --open 






It will build the source, start a web server and open your application in the browser. If there are errors, it displays the output of elm make.


While it doesn’t help with preserving the state of the application between reloads, automatic rebuilding and reloading is still a great convenience. 


elm-live has a bunch of options in addition to --open. There are options to set a custom port for the server or to allow pushState functionality for URL handling in single-page applications. It’s worth checking out its documentation on GitHub for details. 



  create-elm-app



A production web application requires a few tools besides the development server: a tool like Webpack to minimise the JavaScript, some way of running tests and so on. You can set this up yourself using tools that you prefer for each specific task, but if you would like a more integrated experience, then you can use create-elm-app.


create-elm-app wraps a bunch of tools, among them the Elm compiler itself, elm-test, webpack, web pack-dev-server, and Babel with ES6.


Once again, you can install it from NPM:



npm install -g create-elm-app






Then, to create a scaffold for a new application, run this:



create-elm-app elmapp 






Since it is modelled on the create-react-app tool, create-elm-app doesn’t have any configuration options – all you give it is a project name. 


The development server can be started with the start command in the project directory: 



elm-app start 






The server has live reload functionality, so once you change a source file, it will automatically recompile and reload the page in the browser. 


To create an optimised production build with minified JavaScript and CSS, you need to run the build command:



elm-app build






The generated files are written to the build directory. 


To run Elm tests which are located in tests/Tests.elm, you use the test command: 



elm-app test 






Finally, create-elm-app provides an escape hatch for when you want to stop using it and go with a customised setup:



elm-app eject 






This command writes all the configuration into your project directory and installs the dependencies. 



  elm-format



elm-format is a tool for code formatting. It doesn’t have any options but instead aims to format the code in a canonical Elm style. 


While it’s quite popular, I understand it may not be your cup of tea. I hadn’t used elm-format to begin with, but after I tried it, I found it to be quite convenient because I no longer had to think about formatting and indentation. Particularly when you have multiple developers working on the code base, it’s great to have consistent formatting enforced automatically. 


elm-format can be installed from NPM:



npm install -g elm-format






Then you run it on a file or directory:



elm-format src/ --yes 






The --yes flag makes it skip all the confirmation prompts. 


One aspect of elm-format behaviour that you should be aware of is that it effectively removes your ability to expose all symbols in a module. If you write module M exposing (..), elm-format will replace it with an explicit list of symbols (module M exposing (a, b, c)). However, as it doesn’t update this list when you go on to add, remove or rename symbols after this change, you’re now stuck with maintaining the list of exposed symbols manually. In this instance, elm-format makes a semantic change to your source code (rather than simply a visual one). 


In Visual Studio Code, elm-format is run by the Elm extension. If you use WebStorm, you can set up a file watcher for files with the .elm extension to run elm-format every time you save a file. If you use another editor or IDE, check out the elm-format documentation as there are many integration options available.



  elm-analyse



elm-analyse is a tool that detects some cases of suboptimal code as well as some potential bugs. It can detect things like duplicate imports, missing type annotations, unused variables, the use of Debug module functions and quite a few other patterns. elm-analyse makes sense for larger codebases with multiple developers working on them. 


elm-analyse can be installed with yarn and also with npm:



npm install -g elm-analyse






Then you can run elm-analyse without arguments in your project directory to get a list of suggestions. 


It also has a web interface. If you run it with the -s flag, you can see the report in the browser at http://localhost:3000. The web interface provides some additional information about modules and dependencies. 


It’s possible to do some customisation by adding an elm-analyse config file. You can change the list of performed checks, ignore some of the source files, or add a list of words which will trigger a warning if found in the code. 



  elm-debug-transformer



As you’ll see in the “Squashing Bugs” chapter, Elm allows you to log values to the browser console. The values are converted to text, which can get difficult to read when logging complex values (like the whole model, for example). 


To help with this, elm-debug-transformer is a tool that replaces  console.log with a modified version which parses any Elm values it detects into an interactive tree with expandable nodes. 


It can be installed via yarn or npm. If you install from NPM (npm install elm-debug-transformer), you can then include the script in your index.html and register the console.log wrapper:



<script src="node_modules/elm-debug-transformer/dist/elm-console-debug.js"></script>
<script>
ElmConsoleDebug.register()
</script>






In Chrome, you can get even nicer output by enabling custom formatters in the Dev Console preferences.


Elm Package Catalog


When you’re still finding your way in the Elm ecosystem, finding packages can be a bit difficult. To help you explore available packages, I set up Elm Package Catalog where they are organised into categories. It includes all of the Elm 0.19-compatible packages (which currently number around 700). 


Design Considerations


Once the tools are set up and the initial project config is in place, we can get started on the code for the application. 


Elm applications are often single page applications, where the navigation between different UI views is driven by JavaScript and the DOM is dynamically updated within the same page. We are going to go with this model. 


As I mentioned at the start of the chapter, we’re working on an application which will visualise PostgreSQL query plans supplied by the user. A query plan is a tree of nodes representing various steps in the calculation. For example, if we have a query like this: 



select * from projects 
join zones using (project_id) 
order by zone_id;






then the query execution plan might be something like this:



Sort
  Hash Join
    Seq Scan on zones
    Hash
      Seq Scan on projects






This tree represents dependencies, meaning that before the sort, PostgreSQL has to do a hash join, and in order to do a hash join, it needs to perform a sequential table scan on zones, and so on.


PostgreSQL can output the query plan as a JSON string representing the tree of nodes, so that’s what we are going to use as input, and we will convert it into something that’s easier to analyse. 


The simplest possible implementation I can think of is one page where the user pastes in the query plan, and another page that shows the tree of nodes produced from the JSON: 




  
    [image: Initial page setup]
    Initial page setup
  




We can use this as a starting point, and then we can extend it with other features such as allowing the user to save plans or view the list of saved plans. Elm makes this kind of incremental approach easier because you can refactor with confidence knowing that the compiler will help you find any naming or type inconsistencies. 


Of course, there are many other considerations in building a web application, such as whether to break it up into multiple HTML pages, or whether to supplement client side rendering with server side rendering, or whether to allow the application to work offline in some form, or how to deal with network errors. Most of these design decisions are beyond the scope of this book because they are not Elm-specific, although we will look at things like the implementation side of things for error handling. 


UI Construction


Let’s start laying down some UI. We have some options here:



  	Use the html package from the Elm core library combined with some approach for styling

  	Use the elm-ui package which deals with both layout and styling

  	Use one of the widget frameworks such as elm-mdc.




I will describe the first two options in more detail.



html package and styling


The Elm core library includes a package for constructing HTML which is predictably named elm/html. It’s a straightforward wrapper around HTML tags:



import Html exposing (..)
import Html.Attributes exposing (..)
import Html.Events exposing (..)

-- ...

viewLogin : Html Msg
viewLogin = 
    div [ class "login" ] 
    [ input [ placeholder "Organisation", onInput SetOrganisation, value "" ] []
    , input [ placeholder "User name", onInput SetUsername, value "" ] []
    , input [ placeholder "Password"
            , type_ "password"
            , onInput SetPassword
            , value "" 
            ]  
            []
    , button [ onClick Login ] [ text "Login" ]
    ]






The package provides functions for most of the HTML tags like div, input and button. Each of these functions takes an array of HTML attributes, for example class or placeholder, and an array of child elements. 


In addition to attributes, you can add event handlers like onInput and onClick which result in calls to the update function with the specified message value. 


When it comes to styling, there is a plethora of options: 



  	The html package allows you to add inline styles to elements

  	You can write styles in a plain old CSS stylesheet outside of Elm code, and apply styles to HTML elements via ID and class attributes supplied in your view functions

  	You can rely on elm-css to add styles to elements

  	You can use elm-bootstrap which is a wrapper over html functions with baked-in Bootstrap styles.




Inline styles and external stylesheets


We can add a bit of inline styling to the previous example like this (note the extra attributes added to inputs and button):



inputStyles : List (Attribute msg)
inputStyles =
    [ style "font-size" "1.2em"
    , style "font-weight" "bold"
    , style "display" "block"
    ]


buttonStyles : List (Attribute msg)
buttonStyles =
    [ style "font-weight" "bold"
    , style "padding" "10px"
    , style "background-color" "f2fae8"
    ]


viewLogin : Html Msg
viewLogin =
    div [ class "login" ]
        [ input
            ([ placeholder "Organisation"
             , onInput SetOrganisation
             , value ""
             ]
                ++ inputStyles
            )
            []
        , input
            ([ placeholder "User name"
             , onInput SetUsername
             , value ""
             ]
                ++ inputStyles
            )
            []
        , input
            ([ placeholder "Password"
             , type_ "password"
             , onInput SetPassword
             , value ""
             ]
                ++ inputStyles
            )
            []
        , button ([ onClick Login ] ++ buttonStyles) [ text "Login" ]
        ]






This can probably be taken quite far and provides some potential for combining and reusing styles via Elm functions. However, the expected way of using the html package is that the majority of styles are defined in a separate CSS file instead, and applied via a class or ID attribute. 



  elm-css



We can pull the styling a bit further into Elm using the rtfeldman/elm-css package. This package provides replacements for html package functions, and allows you to add CSS styles to elements in a way that leverages type safety:



import Css exposing (..)
import Html
import Html.Styled exposing (..)
import Html.Styled.Attributes exposing (..)
import Html.Styled.Events exposing (..)

-- ...

-- Reusable style fragments
largerFont : Style
largerFont =
    batch
        [ fontSize (Css.em 1.2)
        , fontWeight bold
        ]


inputStyle : List Style
inputStyle =
    [ largerFont
    , display block
    ]


-- A reusable button with styling applied
loginButton : List (Attribute msg) -> List (Html msg) -> Html msg
loginButton =
    styled button
        [ fontWeight bold
        , padding (px 10)
        , backgroundColor (rgb 200 200 200)
        , hover
            [ backgroundColor (rgb 100 100 100)
            , textDecoration underline
            ]
        ]


viewLogin : Html Msg
viewLogin =
    div [ class "login" ]
        [ input [ placeholder "Organisation"
                , onInput SetOrganisation
                , value ""
                , css inputStyle 
                ] 
                []
        , input [ placeholder "User name"
                , onInput SetUsername
                , value ""
                , css inputStyle 
                ] 
                []
        , input [ placeholder "Password"
                , type_ "password"
                , onInput SetPassword
                , value ""
                , css inputStyle 
                ] 
                []
        , loginButton [ onClick Login ] [ text "Login" ]
        ]


main : Html.Html Msg
main =
    viewLogin |> toUnstyled






elm-css provides replacements for html functions which support applying styles with elm-css functions. For the most part, using them is just a matter of importing different modules: Html.Styled, Html.Styled.Attributes and Html.Styled.Events. Then, the nodes can have styles produced by the css function added to the attribute list:



input 
    [ placeholder "Organisation"
    , onInput SetOrganisation
    , value ""
    , css inputStyle 
    ]






Also note that in the above example, every CSS attribute has a corresponding function, and attribute values are expressed via functions too: fontSize (Css.em 1.2). This makes it possible for the compiler to lend a helping hand and rule out some invalid styles. 


Styles can be reused either by combining them into a single style with batch (used when defining largerFont in the example at the start of this section), or even by naming a list of styles (eg inputStyle above) and then using it in multiple places. 


Another option for reuse is demonstrated by loginButton. By calling styled with button and a list of styles, we are producing a value which can be used in place of a plain button but has styles already applied. 


elm-css combines the styles you define for individual elements into a single stylesheet, and applies the styles to elements via generated class names. 


The package also allows you to create standalone styles with the help of the Css.Global module:



import Css exposing (..)
import Css.Global exposing (..)
import Html
import Html.Styled

-- ... 

styles : Html.Html msg
styles =
    global
        [ class "nonElmContent"
            [ descendants
                [ a [ color <| hex "348aa7" ]
                , each [ h1, h2, h3, h4, h5, h6 ] 
                    [ fontFamilies 
                        [ "Helvetica", .value sansSerif ] 
                    ]
                , code [ fontFamilies 
                           [ "Inconsolata", .value monospace ]
                       , fontSize (Css.rem 1)
                       ]
                ]
            , property "pointer-events" "auto"
            ]
        ]
        |> Html.Styled.toUnstyled






Css.Global.global returns the CSS in a style node which can then be added to your views. This can be useful for styling parts of the page which are not generated by Elm. 



  elm-bootstrap



Yet another option for styling is the rundis/elm-bootstrap package. Continuing with the login form example, we can write something like this:



import Bootstrap.Button as Button
import Bootstrap.Form as Form
import Bootstrap.Form.Checkbox as Checkbox
import Bootstrap.Form.Input as Input
import Html exposing (..)
import Html.Attributes exposing (..)

-- ...

viewLogin : Html msg
viewLogin =
    Form.form []
        [ h1 [] [ text "Please login" ]
        , Form.group []
            [ Form.label [ for "organisation" ] [ text "Organisation" ]
            , Input.text [ Input.id "organisation" ]
            ]
        , Form.group []
            [ Form.label [ for "username" ] [ text "User name" ]
            , Input.email [ Input.id "username" ]
            ]
        , Form.group []
            [ Form.label [ for "password" ] [ text "Password" ]
            , Input.password [ Input.id "password" ]
            ]
        , Checkbox.checkbox [ Checkbox.id "remember" ] "Remember me"
        , Button.button [ Button.primary ] [ text "Submit" ]
        ]






In order for this to work, we also need to add the actual Bootstrap CSS to the HTML file:



<link rel="stylesheet" 
href="https://stackpath.bootstrapcdn.com/bootstrap/4.3.1/css/bootstrap.min.css"
integrity="sha384-ggOyR0iXCbMQv3Xipma34MD+dH/1fQ784/j6cY/iJTQUOhcWr7x9JvoRxT2MZw1T" 
crossorigin="anonymous">






Depending on the components you use, you might additionally need to add several pieces of JavaScript:



<script src="https://code.jquery.com/jquery-3.3.1.slim.min.js"
integrity="sha384-q8i/X+965DzO0rT7abK41JStQIAqVgRVzpbzo5smXKp4YfRvH+8abtTE1Pi6jizo" 
crossorigin="anonymous"></script>
<script src="https://cdnjs.cloudflare.com/ajax/libs/popper.js/1.14.7/umd/popper.min.\
js"
integrity="sha384-UO2eT0CpHqdSJQ6hJty5KVphtPhzWj9WO1clHTMGa3JDZwrnQq4sF86dIHNDz0W1" 
crossorigin="anonymous"></script>
<script src="https://stackpath.bootstrapcdn.com/bootstrap/4.3.1/js/bootstrap.min.js"
integrity="sha384-JjSmVgyd0p3pXB1rRibZUAYoIIy6OrQ6VrjIEaFf/nJGzIxFDsf4x0xIM+B07jRM" 
crossorigin="anonymous"></script>	






You can get up-to-date instructions for setup from the Bootstrap site.


All of these are definitely workable approaches. Since the learning curve for them is not particularly steep and there are plenty of examples and tutorials for these packages online, I’d like to put more focus on another package called elm-ui. 



  elm-ui



elm-ui takes a very different approach to UI from what we’ve seen so far in this chapter. Instead of providing some kind of wrappers for HTML and CSS, its goals are to step away from HTML and CSS in Elm altogether, treating them as just an output target, and to provide as many compile time guarantees as possible. In other words, elm-ui is to HTML and CSS what Elm is to JavaScript. 


This package provides a powerful alternative approach to generating both layout and styles which is based on four ideas:



  	Getting the compiler to verify as much of the layout and styling as possible by defining them in Elm code, rather than doing it in HTML and CSS

  	Enabling all of the reuse and refactoring capabilities of Elm for UI code

  	A simplified approach to styles and, in particular, layout; styles are specified locally for each element.

  	Context independence: elements and their attributes are expected to behave the same regardless of the surrounding context (which is often not the case for CSS). 




We all know that HTML and CSS can at times be extremely counterintuitive when building user interfaces, which is due to their original purpose as document layout languages. elm-ui goes in a direction which is more suitable to interactive applications, and in my opinion, it’s very promising. 


Beside the fact that CSS layouts can be a huge time sink when building a UI, if you’re working with plain HTML and CSS, or using one of the packages we talked about earlier, then you typically end up with an awkward mix of layout and visual styling in some places, and an awkward separation between parts of the layout in other places, which means that you have to keep moving around the code as you work on the layout. This isn’t convenient. 


In elm-ui, all of the layout and visual styling is done within your view function, with the “gross morphology” of layout made explicit through the functions exposed by this package (as you will see in code samples shortly). Each element in the layout can be assigned style attributes, similarly to how we added inline styles when using the html package. 


I think it’s best to demonstrate all of this with code, so let’s dive in! 


Creating a basic UI


Things start out innocently enough. Here is the boilerplate for our program, so far without a whiff of elm-ui:



module Main exposing (main)

import Browser
import Html


type Page
    = InputPage


type Msg
    = NoOp


type alias Model =
    { currPage : Page
    }


type alias Flags =
    ()


init : Flags -> ( Model, Cmd Msg )
init _ =
    ( { currPage = InputPage
      }
    , Cmd.none
    )


-- SUBSCRIPTIONS


subscriptions : Model -> Sub Msg
subscriptions model =
    Sub.none


-- UPDATE


update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        _ ->
            ( model, Cmd.none )


-- VIEW


view : Model -> Browser.Document Msg
view _ =
    { title = "VisExp"
    , body = []
    }


main : Program Flags Model Msg
main =
    Browser.document
        { init = init
        , update = update
        , view = view
        , subscriptions = subscriptions
        }






I’m definitely going to structure the UI code as multiple pages: one for input, another for displaying the visualised tree, another one to show a list of saved plans later on, etc. I’ve added the Page type to reflect that. Correspondingly, the model just keeps track of the page to begin with, and the init function initialises the model with the only available value. init has to take an argument representing flags passed to the Elm program on initialisation. You’ll see how they can be used later, and for now we will just ignore this argument. 


I need a Msg type, as always, but I don’t know what the messages are going to be, so I’ve just put a dummy message constructor in there for now. Further down, I set up no-op subscriptions and the update function and defined the most primitive view imaginable. Finally, I tied it all together in a call to Browser.document, which requires that the view function returns Browser.Document Msg – a record containing title (used for the HTML title), and another field called body, which is a list of Html Msg elements. Since the body is an empty list, if we run this, all we will see is a blank page titled “VisExp”. 


Adding a navigation bar


The next step is to add a little bit of layout. Let’s start with a navigation bar at the top.


First, I need to import one of the elm-ui modules: 



import Element exposing (..) 






I’m exposing everything for simplicity here, but you may want to use a qualified import because of potential naming conflicts. Then, I can change my view function to be:



view : Model -> Browser.Document Msg
view _ =
    { title = "VisExp"
    , body =
        [ layout [] <| text "VisExp"
        ]
    }






layout converts an elm-ui Element msg value into an Html msg value which can be used to populate the body field of Browser.Document msg, among other things. Its first argument is a list of Attribute msg values. We’ll see examples of these shortly as we continue to expand the UI. I have provided its second argument by generating the simplest Element msg possible – plain text – with the help of Element.text.


What we have now is the minimal working program that combines Browser.document with elm-ui. Let’s expand it to display a navigation bar at the top of the page:



view : Model -> Browser.Document Msg
view _ =
    { title = "VisExp"
    , body =
        [ layout [] <|
            column [ width fill, spacingXY 0 20 ]
                [ navBar
                , el [ centerX ] <| text "VisExp"
                ]
        ]
    }






I’ve used a new function: Element.column. Note that it has a similar signature to layout. It takes a list of attributes and a list of child elements (rather than a single element). 


The elm-ui package has 5 primary layout functions:



  	column

  	row

  	paragraph

  	textColumn

  	table




Predictably, column and row arrange their child elements in columns and rows, whereas paragraph produces a bunch of wrapped inline elements, which is handy for laying out text interspersed with images or highlights. textColumn provides a way of stringing multiple paragraphs together. Lastly, table is a way of generating tables (with headers). 


There are also a couple of less common layouts: wrappedRow and indexedTable. 


In the code snippet above, I’m laying out the contents of the page in a column, with the navBar above the rest of the content. I’ve specified the width of the column to be 100% with width fill, and there will be 20 pixels between elements. 


The other function I used for content below the navigation bar is el, the elm-ui analog to div in the core Html package, and a key building block for layouts. It takes two arguments: 



  	a list of attributes

  	a child element.




The child element is still just plain text for now, and I’ve given el a centerX attribute to centre it horizontally on the page. 


The remaining piece to define is navBar, which is going to look like this:



blue : Color
blue =
    rgb255 52 101 164

navBar : Element Msg
navBar =
    row
        [ width fill
        , paddingXY 60 10
        , Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
        , Border.color blue
        ]
        [ el [ alignLeft ] <| text "VisExp"
        , el [ alignRight ] <| text "Menu"
        ]






It’s a row containing two text elements: “VisExp” and “Menu”. I’ve used alignLeft and alignRight attributes on the elements to push them to the opposite sides of the page. The row itself has a few attributes as well: width fill makes it take up 100% of the width of its parent element, paddingXY 60 10 means that the content of the row is offset by 60 pixels on the left and right, as well as 10 pixels on top and bottom. 


This row of elements sits inside the column we defined earlier. I find that it’s quite intuitive to think about the layout as a hierarchy of rows and columns, and it’s a very flexible mechanism that can be used to manage complicated layouts. 


I should also point out that in contrast to CSS, elm-ui doesn’t have a concept of margins. Instead, you can specify padding for a given element as well as spacing between its child elements:




  
    [image: Padding and spacing are defined on the parent element]
    Padding and spacing are defined on the parent element
  




Lastly, as I want the navigation bar to be visually separated from the content below, I’ve given the row a single-pixel border on the bottom, which is set to be blue – with the colour defined using the Element.rgb255 helper function.


As you can see, element attributes are a mix of layout information and visual styling. The advantage over CSS is that they are part of the definition of the element, and since they are in a List, you can use all of the List functions as well as Elm functions and records to organise and combine attributes. 


Visual styles are broken down into a number of categories with corresponding modules such as Background and Border. Each category has a number of functions like Border.color or Border.width. To make the above example compile, we need to add an import: 



import Element.Border as Border






The result looks like this: 

    Layout with a navigation bar
  




Adding an input text area


Next, let’s add a textarea for pasting the EXPLAIN output into. We need an extra import for input elements:



import Element.Input as Input






This module import has to be qualified because it has a text function which would conflict with the one that produces plain text. 


After adding the import, we can update the view:



lightCharcoal : Color
lightCharcoal =
    rgb255 136 138 133

inputPage : Model -> Element Msg
inputPage model =
    column
        [ width (px 800)
        , spacingXY 0 10
        , centerX
        ]
        [ Input.multiline
            [ height (px 300)
            , Border.width 1
            , Border.rounded 3
            , Border.color lightCharcoal
            , padding 3
            ]
            { onChange = ChangePlanText
            , text = model.currPlanText
            , placeholder = Nothing
            , label =
                Input.labelAbove [] <|
                    text "Paste the EXPLAIN output in JSON format:"
            , spellcheck = False
            }
        , Input.button
            [ Background.color green
            , Border.color darkGreen
            , Border.rounded 3
            , Border.widthEach { bottom = 3, top = 0, right = 0, left = 0 }
            , Font.bold
            , Font.color white
            , paddingXY 20 6
            , alignRight
            , width (px 200)
            , height (px 40)
            ]
            { onPress = Just SubmitPlan
            , label = el [ centerX ] <| text "Go!"
            }
        ]


view : Model -> Browser.Document Msg
view model =
    { title = "VisExp"
    , body =
        [ layout [] <|
            column [ width fill, spacingXY 0 20 ]
                [ navBar
                , inputPage model
                ]
        ]
    }






There’s quite a lot going on here now! Let’s see the result before walking through this code: 




  
    [image: VisExp page with navbar and textarea]
    VisExp page with navbar and textarea
  




If you look at view, you can see that we now have inputPage below navBar. inputPage is itself a column which I set to be 800px wide and centred horizontally on the page. This column contains a text area produced by Input.multiline, as well as a “Go!” button. 


Both Input.multiline and Input.button have a bunch of attributes which are hopefully straightforward to understand. There are a few new colours used for these attributes, but I’m not going to show the definitions for them (you can see them in the sample code you got along with the book).


The second argument of multiline is a bit involved. It’s a record with 5 fields:



  	
onChange, a change handler

  	
text, which is the content of the text area

  	
placeholder, which is optional and appears inside the text area if it’s not Nothing


  	
label, which is another elm-ui element wrapped in a placement function (in this case, labelAbove)

  	
spellcheck, a boolean flag we set to False because we’re going to paste in JSON. 




The change handler requires a message which can carry a string, so I have to add that to my Msg type:



type Msg
    = NoOp    
    | ChangePlanText String






Input.button also takes a record as its second argument. onPress is a Maybe value; supplying Nothing means the button is disabled. We supplied Just SubmitPlan so we have to add SubmitPlan to the Msg type as well:



type Msg
    = NoOp    
    | ChangePlanText String
    | SubmitPlan






We also need to supply the label for the button in the form of an elm-ui element. 


This is all we need to display the input page, but so far it doesn’t do anything as we haven’t added anything to update to handle the new messages. 


Basic Event Handling


If we were dealing with JavaScript, we might want to attach functions to particular event handlers. But in Elm architecture, we specify the message that a particular event may generate. The message may or may not carry data. 


I’ve already created two event handlers: in the text area definition, I said that the onChange event is going to generate ChangePlanText messages, and the “Go!” button will generate SubmitPlan messages when pressed. These messages then need to be handled in the update function in order to modify the model. 


Right now, we don’t do anything with the ChangePlanText messages, so let’s amend that. First, I need to add a field to store the current value of the input in the model, and to provide an initial value for it:



type alias Model =
    { currPage : Page
    , currPlanText : String
    }


init : Flags -> ( Model, Cmd Msg )
init _ =
    ( { currPage = InputPage
      , currPlanText = ""
      }
    , Cmd.none
    )






Second, the update function has to set this field:



update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        ChangePlanText s ->
            ( { model | currPlanText = s }, Cmd.none )

        NoOp ->
            ( model, Cmd.none )






While we’re dealing with events, let’s also add a new page which is going to show the parsed tree, and make the “Go!” button take us to that page:



type Page
    = DisplayPage
    | InputPage


update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        ChangePlanText s ->
            ( { model | currPlanText = s }, Cmd.none )

        NoOp ->
            ( model, Cmd.none )

        SubmitPlan ->
            ( { model | currPage = DisplayPage }, Cmd.none )






The following question is, how do we display the right page? Currently we have a single view which is always the same, so we need to do a bit of refactoring:



view : Model -> Browser.Document Msg
view model =
    let
        content =
            case model.currPage of
                DisplayPage ->
                    displayPage model

                InputPage ->
                    inputPage model
    in
    { title = "VisExp"
    , body =
        [ layout [] <|
            column [ width fill, spacingXY 0 20 ]
                [ navBar
                , content
                ]
        ]
    }






Let’s get displayPage to echo the input for now: 



displayPage : Model -> Element Msg
displayPage model =
    column [] [ text model.currPlanText ]






Albeit this is simplistic, we have nonetheless started learning how to lay out pages, how to add visual styling, and how to weave messages throughout the application in order to react to events. 


Modules


We have the beginnings of a functional UI, so now it’s time to start populating it with something useful. To do that, we need to take the JSON data supplied by the user and parse it into records so we can build up the tree for display. Before we get into that, we should take a short detour and talk about organising the code into files. 


In Elm, files and modules have a one-to-one correspondence. If you want to break out a piece of code into a separate file, you have to organise it as a module in that file. 


The general approach I would suggest is to start your project with a single file, Main.elm, and keep adding to it until you see that it’s becoming long and inconvenient to navigate, and you also identify some fairly independent piece of code that can be carved off into a module. 


JSON parsing is a good example of a piece of functionality that can be extracted. To begin with, all our code is in a single file, Main.elm. Now that we’re going to have multiple source files, it’s a good idea to put them in a directory called src, which is the usual convention.


In order to demonstrate the namespacing of modules, let’s imagine that we are considering parsing other input formats in addition to JSON in the future (for example, the plain text query plan output which can also be generated by Postgres). Then, it makes sense to create a directory called PlanParsers and to put our new JSON module there. The resulting source tree looks like this:



src
  PlanParsers
    Json.elm
  Main.elm






In Json.elm, we have to lead with this line:



module PlanParsers.Json exposing (..)






In order to make the functions defined in Json.elm available in Main.elm, we need to import our newly created module the same way as we’ve been importing third party modules: 



import PlanParsers.Json exposing (..) 






Note that the path to the file has to match up with the dot-separated name of the module, so the capital “P” in the directory name PlanParsers is required, and if we change the directory name, we have to update the module name correspondingly. 


Constraining exports


We’ve used the exposing keyword in two different contexts: in Json.elm when naming the module, and in Main.elm when importing it. 


In Plan.elm, the exposing keyword can be used to restrict what’s visible outside the module. For instance, let’s assume we have a few definitions:



type Plan
    = PCte CteNode
    | PResult ResultNode
    | PSort SortNode


type alias PlanJson =
    { executionTime : Float
    , plan : Plan
    , planningTime : Float
    , triggers : List String
    }


decodePlanJson : Decode.Decoder PlanJson
decodePlanJson =
    ...






Instead of exposing everything, we can whitelist the identifiers which are exposed by the module. For example, to expose just decodePlanJson and nothing else, we need to write:



module PlanParsers.Json exposing (decodePlanJson) 






We can also add type aliases and types into the mix: 



module PlanParsers.Json exposing (Plan, PlanJson, decodePlanJson)






This will make both PlanJson and Plan available for use in type annotations, and the PlanJson constructor will be made available as well. 


There is a nuance for types like Plan as opposed to type aliases. While we’ve exported the Plan constructor, we haven’t exposed any of the tags like PCte and PResult. So we will be able to use Plan as a type of record fields, and we will be able to use it in type signatures, but we will not be able to construct values of this type. This makes plan a so called opaque type. I will talk more about opaque types in the penultimate chapter. 


For now, I will just point out that if we want to be able to construct values of this type, then we need to change our whitelist:



module PlanParsers.Json exposing (Plan(..), PlanJson, decodePlanJson) 






Note the (..) after Plan – it signifies that we want to expose all of the constructors included in that type definition. 


Constraining imports


When importing the module, there are a few variations as well. 


The simplest import statement looks like this:



import PlanParsers.Json 






This is called a qualified import, and it makes all the identifiers exposed by the module available in the current file, but they have to be prefixed with the module name: 



PlanParsers.Json.decodePlanJson

PlanParsers.Json.PCte






This is a bit unwieldy, so we can also alias the name of the module with something easier to type:



import PlanParsers.Json as P 






Now we can write P.decodePlanJson and P.PCte. Alternatively, we can dispense with the module name altogether by using an open import:



import PlanParsers.Json exposing (..) 






Importing a lot of identifiers can cause name clashes with other modules, so instead we can use an open import with a list of specific identifiers that we need: 



import PlanParsers.Json exposing (PlanJson, decodePlanJson) 






This import statement means that we can use PlanJson and decodePlanJson but not PCte or anything else.


It’s even possible to write a combined import statement for extra flexibility: 



import PlanParsers.Json as P exposing (PlanJson)






This would allow you to use PlanJson without a prefix, while everything else would need to be prefixed with P..


Now that we know how to write a module, we can go back to handling JSON in the PlanParsers.Json module. 


JSON Parsing


As part of the Elm core library, we have Json.Decode and Json.Encode modules in the elm/json package for converting JSON strings to and from Elm values. They do have a slight limitation in that if you have a lot of fields, like we do with query plan nodes, it becomes a bit inconvenient to handle them. For this reason, I’m going to combine the core package with the NoRedInk/elm-json-decode-pipeline package, which is also a very popular choice for working with JSON.


First of all, let’s take a moment to understand how JSON parsing works in Elm. The main function for parsing JSON is called decodeString. It takes two parameters, a decoder and a string:



decodeString : Decoder a -> String -> Result Error a






A decoder provides a recipe for converting string data into a value of a particular type (note that a decoder is a value rather than a function). The Json.Decode module comes with a bunch of simple decoders which you can combine to decode arbitrary JSON structures. It has decoders for the primitive types like string, int, bool, as well as compound structures like array and dict, and some others. 


If decodeString succeeds, it returns a record of the type given by the decoder, otherwise it returns an error as an Error value, which can be converted to a string with Json.Decode.errorToString. For instance, suppose we have a simple JSON object: 



{ "Node Type": "CTE Scan", "Total Cost": 4.2 }






The decoder for this JSON will be:



import Json.Decode as Decode 

type alias Node =
    { nodeType : String, totalCoast : Float }

d : Decode.Decoder Node
d =
  Decode.map2 Node
    (Decode.field "Node Type" Decode.string)
    (Decode.field "Total Cost" Decode.float)






map2 is a way of combining multiple decoders into a single decoder.


Now, to actually decode a string, we need to tell the decodeString function to use this decoder:



json =
    """{ "Node Type": "CTE Scan", "Total Cost": 4.2 }"""

Decode.decodeString d json
-- Returns: Ok { nodeType = "CTE Scan", totalCost = 4 }






To take another example, decoding a JSON array looks like this: 



json = 
    """["one", "two", "three"]"""

Decode.decodeString (Decode.list Decode.string) json
-- Returns: Ok ["one","two","three"]






As you can see, writing decoders is a process of combining basic decoders according to the structure of your data. So let’s consider the structure of the input data in our application:



{
  "Plan": {
    "Node Type": "Nested Loop",
    "Total Cost": 6093.08,
    ...
    "Plans": [
      {
        "Node Type": "WindowAgg",
        "Total Cost": 96.67,
        ...
      },
      ...
    ] 
  },
  "Planning Time": 2.659,
  "Triggers": [],
  "Execution Time": 318.718
}






The top level consists of a “Plan” key plus some optional fields like “Execution Time” and “Planning Time”. That’s straightforward, but the value under the “Plan” key is more interesting: 



  	Each node has a set of attributes common to all nodes, as well as some node specific attributes. For example, every node has “Startup Cost” and “Total Cost”, but a “Sort” node also has “Sort Key” and “Sort Method” which are specific to it. 

  	The structure is recursive: a node may optionally contain a “Plans” field which is a list of nodes, each of which may have its own “Plans” field, and so on.




Let’s start writing the decoder from the top down, and bring in Json.Decode.Pipeline to help. Json.Decode.Pipeline has the concept of required and optional fields. In order to handle the top level structure, we need to do this:



import Json.Decode as Decode
import Json.Decode.Pipeline exposing (..)

type alias PlanJson =
    { executionTime : Float
    , plan : Plan
    , planningTime : Float
    , triggers : List String
    }


decodePlanJson : Decode.Decoder PlanJson
decodePlanJson =
    Decode.succeed PlanJson
        |> optional "Execution Time" Decode.float 0
        |> required "Plan" decodePlan
        |> optional "Planning Time" Decode.float 0
        |> optional "Triggers" (Decode.list Decode.string) []






succeed is a Json.Decode function which we use to kick off the process of building the decoder. We pass it the constructor of the type we want to construct from JSON, and then provide definitions for each field with the help of the pipe operator. 


Keep in mind that the order of fields here has to match the order of fields in the type alias. For each optional field, we need to supply a default value of the right type. 


For “Execution Time”, “Planning Time”, and “Triggers”, I’m using the basic decoders supplied in Json.Decode. For the “Plan” field, I have to supply a custom decoder (decodePlan) because it’s an object rather than a simple value. We will also need to define the Plan type which this decoder will produce. We will do it a bit later. 


The next step is to write decodePlan, and that requires dealing with all the different node types. There are several possibilities here: 



  	Write a single decoder which will be able to handle any kind of node by way of making all node specific attributes optional. This would probably work but it would be ugly, because we need to supply “special” default values for missing fields, and what’s a special value for an integer, for example? We could sidestep the issue of default values by wrapping field types in Maybe and decoding to Maybe values with something like optional "Plan Width" (Decode.map Just Decode.int) Nothing. But either way, the conditional logic required to deal with different nodes further down the line isn’t likely to be pretty. 

  	We could think of specific nodes as extended versions of a shared set of attributes, which would be sort of like extensible records. However, using extensible records wouldn’t work because we need to pass the record constructor to decode, and you may recall from the previous chapter that extensible records don’t get constructors generated for them. Besides, extensible records are meant for restricting arguments rather than modelling data.

  	The generally preferred Elm solution for shared fields is to nest them in a field, so that’s what we are going to do. 




Let’s define some types first: 



type Plans
    = Plans (List Plan)

type alias CommonFields =
    { nodeType : String
    , plans : Plans
    , relationName : String
    , schema : String
    , startupCost : Float
    , totalCost : Float
    }

type alias CteNode =
    { common : CommonFields
    , alias_ : String
    , cteName : String
    }

type alias ResultNode =
    { common : CommonFields
    , parentRelationship : String
    }

type alias SeqScanNode =
    { common : CommonFields
    , alias_ : String
    , filter : String
    , relationName : String
    , rowsRemovedByFilter : Int
    }

type alias SortNode =
    { common : CommonFields
    , sortKey : List String
    , sortMethod : String
    , sortSpaceUsed : Int
    , sortSpaceType : String
    }






There are many more attributes and node types, so this is far from exhaustive, but it’s sufficient for demonstration purposes. We have a set of common fields listed in the CommonFields alias, and each of the aliases for a specific node type complements the common field with a few extra fields. 


Note that CommonFields contains a field of type Plans. This is simply a list of plans, but due to the way Elm deals with recursion as a strict language, Plans has to be a single-constructor custom type rather than a type alias. 


Next, let’s attempt to write a decoder for a ResultNode:



decodeResultNode : Decode.Decoder Plan 
decodeResultNode = 
    Decode.succeed ResultNode
        |> ???
        |> required "Parent Relationship" Decode.string






We need to replace the question marks with a recipe for decoding the common fields and populating the common record field with their values. Neither required nor optional methods we’ve seen so far are suitable for that, because they only deal with a single JSON field. Instead, we need to use another Json.Decode.Pipeline function called custom which takes a decoder as its argument:



decodeResultNode : Decode.Decoder Plan 
decodeResultNode = 
    Decode.succeed ResultNode
        |> custom decodeCommonFields
        |> required "Parent Relationship" Decode.string






Decoders for other node types will follow the same pattern. Remember that the ordering of operations is important when constructing a decoder. Since we called custom first, it will be used to populate the first field of ResultNode, which is common. Luckily, if you don’t get the order right, you will see an error at compile time rather than at run time (unless both fields happen to be of the same type, which is something to watch out for).


decodeCommonFields is a straightforward decoder, with the exception of the second field, “Plans”: 



decodeCommonFields : Decode.Decoder CommonFields
decodeCommonFields =
    Decode.succeed CommonFields
        |> required "Node Type" Decode.string
        |> optional "Plans" decodePlans (Plans [])
        |> optional "Relation Name" Decode.string ""
        |> optional "Schema" Decode.string ""
        |> required "Startup Cost" Decode.float
        |> required "Total Cost" Decode.float






The decodePlans function has to deal with the recursive nature of the data. Again, let’s try to write something. There is a basic decoder for lists, so clearly we need to use that:



decodePlans : Decode.Decoder Plans
decodePlans =
    Decode.list decodePlan






Here, we run into the need to define the concept of a “node plan”. But we have many possible node plans! We need to deal with that both in terms of types (note that I referred to Plan in the decodePlans type annotation), and in terms of decoding within the singular decodePlan, which we still have to write. 


At type level, the way to describe something that can take a number of different forms is a custom type:



type Plan
    = PCte CteNode
    | PResult ResultNode
    | PSeqScan SeqScanNode
    | PSort SortNode






At decoder level, there are two options:



  	We can use a function called oneOf from Json.Decode, which takes a list of decoders and tries them in sequence until it finds one that can be applied successfully. It makes the most sense to use it when the data can be in different formats. 

  	We can look at the node type and apply the appropriate node decoder based on it with the help of  Json.Decode.andThen, which allows us to create decoders dependent on previous results. 




Using oneOf, decodePlan looks like this:



decodePlan : Decode.Decoder Plan
decodePlan =
    Decode.oneOf
        [ decodeCteNode
        , decodeResultNode
        , decodeSeqScanNode
        , decodeSortNode
        , decodeGenericNode
        ]






Note that the generic decoder has to appear last in the list, otherwise every node would be successfully decoded as a generic node. 


The decoder above will produce correct results, however in our case, andThen is a better fit:



andThen : (a -> Decoder b) -> Decoder a -> Decoder b






We can pass it decodeNode, which will receive the node type string as its argument:



decodePlan : Decode.Decoder Plan
decodePlan =
    Decode.field "Node Type" Decode.string
        |> Decode.andThen decodeNode






decodeNode can then select the appropriate decoder based on the node type: 



decodeNode : String -> Decode.Decoder Plan
decodeNode nodeType =
    case nodeType of
        "CTE Scan" ->
            decodeCteNode

        "Result" ->
            decodeResultNode

        "Seq Scan" ->
            decodeSeqScanNode

        "Sort" ->
            decodeSortNode






To a first approximation, we are done: we have a decoder for the top level structure, we have decoders for different node types, and we have come up with a way to decode the nested plans in each node. However, there are still a number of details to be taken care of to make this compile and work properly. 


Let’s try using these decoders from Main.elm:



import Json.Decode
import PlanParsers.Json exposing (..)

-- ...

displayPage : Model -> Element Msg
displayPage model =
    let
        tree =
            case Json.Decode.decodeString decodePlanJson model.currPlanText of
                Ok planJson ->
                    text "Success"

                Err err ->
                    text <| Json.Decode.errorToString err
    in
    column [] [ tree ]






This will make the compiler complain about missing types and functions. One reason is that we haven’t yet implemented the decoders used in decodePlan.


Let’s take decodeResultNode; we’ve already implemented a part of it like this:



decodeResultNode : Decode.Decoder Plan
decodeResultNode =
    Decode.succeed ResultNode
        |> custom decodeCommonFields
        |> required "Parent Relationship" Decode.string






However, this actually returns Decode.Decoder ResultNode whereas what we need (as per the type annotation) is a Decode.Decoder Plan. We can achieve that by applying Decode.map:



decodeResultNode : Decode.Decoder Plan
decodeResultNode =
    let
        innerDecoder =
            Decode.succeed ResultNode
                |> custom decodeCommonFields
                |> required "Parent Relationship" Decode.string
    in
        Decode.map PResult innerDecoder






The decoders for other node types are very similar: 



decodeCteNode : Decode.Decoder Plan
decodeCteNode =
    let
        innerDecoder =
            Decode.succeed CteNode
                |> custom decodeCommonFields
                |> required "Alias" Decode.string
                |> required "CTE Name" Decode.string
    in
        Decode.map PCte innerDecoder


decodeSeqScanNode : Decode.Decoder Plan
decodeSeqScanNode =
    let
        innerDecoder =
            Decode.succeed SeqScanNode
                |> custom decodeCommonFields
                |> required "Alias" Decode.string
                |> optional "Filter" Decode.string ""
                |> required "Relation Name" Decode.string
                |> optional "Rows Removed by Filter" Decode.int 0
    in
        Decode.map PSeqScan innerDecoder


decodeSortNode : Decode.Decoder Plan
decodeSortNode =
    let
        innerDecoder =
            Decode.succeed SortNode
                |> custom decodeCommonFields
                |> required "Sort Key" (Decode.list Decode.string)
                |> required "Sort Method" Decode.string
                |> required "Sort Space Used" Decode.int
                |> required "Sort Space Type" Decode.string
    in
        Decode.map PSort innerDecoder






The compiler also rightfully complains about the incomplete case expression that handles node types: 



decodeNode : String -> Decode.Decoder Plan
decodeNode nodeType =
    case nodeType of
        "CTE Scan" ->
            decodeCteNode

        "Result" ->
            decodeResultNode

        "Seq Scan" ->
            decodeSeqScanNode

        "Sort" ->
            decodeSortNode






Naturally, we need to add a catch-all branch: 



decodeNode : String -> Decode.Decoder Plan
decodeNode nodeType =
    case nodeType of
        "CTE Scan" ->
            decodeCteNode

        "Result" ->
            decodeResultNode

        "Seq Scan" ->
            decodeSeqScanNode

        "Sort" ->
            decodeSortNode

        _ ->
            decodeGenericNode






We need to amend the Plan type correspondingly, and to implement this generic node decoder:



type Plan
    = PCte CteNode
    | PGeneric CommonFields
    | PResult ResultNode
    | PSeqScan SeqScanNode
    | PSort SortNode

-- ...

decodeGenericNode : Decode.Decoder Plan
decodeGenericNode =
    Decode.map PGeneric decodeCommonFields






Finally, decodePlans needs to be tweaked to produce a Plans value rather than List Plan:



decodePlans : Decode.Decoder Plans
decodePlans =
    Decode.map Plans <| Decode.list decodePlan






And now we’re really done!


JSON-to-Elm


As decoders involve some boilerplate, there is some help available with generating them. JSON-to-Elm helps you generate the Elm code for JSON decoders and encoders based on a sample JSON object that you provide. This saves you time on writing boilerplate, although you may well want to customise the resulting Elm for your specific needs. 


For example, I could supply this JSON as the input:



{ 
  "createdAt": "2017-12-16",
  "id": "1",
  "name": "Project query plan",
  "planText": "{}"
}






Assuming that I also provide SavedPlan as the top level type alias name, I get these decoders and encoders generated for me: 



import Json.Encode
import Json.Decode exposing (field)

type alias SavedPlan =
    { createdAt : String
    , id : String
    , name : String
    , planText : String
    }


decodeSavedPlan : Json.Decode.Decoder SavedPlan
decodeSavedPlan =
    Json.Decode.map4 SavedPlan
        (field "createdAt" Json.Decode.string)
        (field "id" Json.Decode.string)
        (field "name" Json.Decode.string)
        (field "planText" Json.Decode.string)


encodeSavedPlan : SavedPlan -> Json.Encode.Value
encodeSavedPlan record =
    Json.Encode.object
        [ ("createdAt",  Json.Encode.string <| record.createdAt)
        , ("id",  Json.Encode.string <| record.id)
        , ("name",  Json.Encode.string <| record.name)
        , ("planText",  Json.Encode.string <| record.planText)
        ]






JSON-to-Elm also has the ability to generate decoders using the JSON.Decode.Pipeline package (although at the time of publishing it’s out of date, so you’ll need to change decode to succeed manually): 



decodeSavedPlan : Json.Decode.Decoder SavedPlan
decodeSavedPlan =
    Json.Decode.Pipeline.decode SavedPlan
        |> Json.Decode.Pipeline.required "createdAt" (Json.Decode.string)
        |> Json.Decode.Pipeline.required "id" (Json.Decode.string)
        |> Json.Decode.Pipeline.required "name" (Json.Decode.string)
        |> Json.Decode.Pipeline.required "planText" (Json.Decode.string)






Displaying the Results


The last thing I’d like to do in this chapter is adding the UI to show the results of our hard work on the JSON decoders. 


At the moment, we only have a stub of the result page: 



displayPage : Model -> Element Msg
displayPage model =
    let
        tree =
            case Json.Decode.decodeString decodePlanJson model.currPlanText of
                Ok planJson ->
                    text "Success"

                Err err ->
                    text <| Json.Decode.errorToString err
    in
    column [] [ tree ]






I have to point out that doing the decoding in the view is not a good design! It should really happen once in the update function when the plan text is submitted by the user. If the plan text can’t be decoded successfully, then the user should remain on the input page and not even get to the display page. However, it calls for fairly substantial changes which would complicate the update function more than is pedagogically desirable, and obscure other aspects of the code. For this reason, I’m going to settle for providing this warning, and will not demonstrate a more realistic implementation in the book. 


We need to add some code to generate a tree of nodes from the parsed data consisting of Plan and Plans values. Let’s call the function that does this planNodeTree, and let’s call it from displayPage:



displayPage : Model -> Element Msg
displayPage model =
    let
        tree =
            case Json.Decode.decodeString decodePlanJson model.currPlanText of
                Ok planJson ->
                    planNodeTree planJson.plan

                Err err ->
                    [ text <| Json.Decode.errorToString err ]
    in
    column [] tree






Here is how this function can be implemented:



planNodeTree : Plan -> List (Element Msg)
planNodeTree plan =
    let 
        nodeTypeEl nodeType = 
            el [ Font.bold ] <| text nodeType        

        treeNode node nodeDetails = 
            [ el
                [ Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
                , Border.color lightBlue
                , mouseOver [ Background.color lightYellow ]
                , padding 4
                ] <|
                paragraph [] ((nodeTypeEl node.common.nodeType) :: nodeDetails)
            , childNodeTree node.common.plans
            ]
    in 
    case plan of
        PCte cteNode ->
            treeNode cteNode
                [ text " on "
                , el [ Font.italic ] <| text cteNode.cteName
                , text <| " (" ++ cteNode.alias_ ++ ")"
                ]

        PGeneric genericNode ->
            treeNode { common = genericNode }
                []

        PResult resultNode ->
            treeNode resultNode
                []

        PSeqScan seqScanNode ->
            treeNode seqScanNode 
                [ text " on "
                , el [ Font.italic ] <| text seqScanNode.relationName
                , text <| " (" ++ seqScanNode.alias_ ++ ")"
                ]

        PSort sortNode ->
            treeNode sortNode  
                [ text " on "
                , el [ Font.italic ] <| text <| String.join ", " sortNode.sortKey
                ]






We need to pattern-match on the tag of the plan value, and show the appropriate details for each type of node. There is some scaffolding that’s going to be the same regardless of node type, so I’ve described it within treeNode in the let. 


Note the mouseOver attribute used in defining the first element in treeNode. This attribute is the elm-ui analogue of the CSS pseudo-class hover. It allows us to highlight rows when the cursor hovers over them. We could also achieve it by sending messages on relevant mouse events, and changing the background colour according to a flag in the model, but using mouseOver is much simpler.  


The second element in the list returned by treeNode calls another function called childNodeTree. This function converts the child nodes contained in the plans field into elm-ui elements. It can be defined like this: 



childNodeTree : Plans -> Element Msg
childNodeTree (Plans plans) =
    column [ paddingEach { left = 20, right = 0, top = 0, bottom = 0 } ] <|
        List.concatMap planNodeTree plans






planNodeTree and childNodeTree are mutually recursive, so the end result of calling planNodeTree on the root node is that the whole plan tree gets rendered as elm-ui elements:

Now that we have a basic tree of nodes displaying, it would be nice to show all the attributes for a selected node. Since there are quite a few of them, I don’t want to insert the attributes inline into the tree. Instead, when the user hovers the mouse over a node, I’d like to populate a panel on the right hand side with attributes for that node. 


We can achieve that by handling mouseenter and mouseleave events. First, let’s add a couple of messages for these events:



type Msg
    = NoOp
    | ChangePlanText String
    | MouseEnteredPlanNode Plan
    | MouseLeftPlanNode Plan
    | SubmitPlan






MouseEnteredPlanNode and MouseLeftPlanNode need to carry a Plan value because we want to show both common and node specific attributes. 


We’ll keep track of the selected node in the model as Maybe Plan to deal with the situation when nothing is selected (the init value has to be modified accordingly): 



type alias Model =
    { currPage : Page
    , currPlanText : String
    , selectedNode : Maybe Plan
    }

init : Flags -> ( Model, Cmd Msg )
init _ =
    ( { currPage = InputPage
      , currPlanText = ""
      , selectedNode = Nothing
      }
    , Cmd.none
    )

-- ...

update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        ChangePlanText s ->
            ( { model | currPlanText = s }, Cmd.none )

        MouseEnteredPlanNode plan ->
            ( { model | selectedNode = Just plan }, Cmd.none )

        MouseLeftPlanNode commonFields ->
            ( { model | selectedNode = Nothing }, Cmd.none )

        NoOp ->
            ( model, Cmd.none )

        SubmitPlan ->
            ( { model | currPage = DisplayPage }, Cmd.none )






Then, we need to update the displayPage function to show an attribute panel: 



displayPage : Model -> Element Msg
displayPage model =
    let
        tree =
            case Json.Decode.decodeString decodePlanJson model.currPlanText of
                Ok planJson ->
                    planNodeTree planJson.plan

                Err err ->
                    [ text <| Json.Decode.errorToString err ]

        details =
            case model.selectedNode of
                Nothing ->
                    [ text "" ]

                Just plan ->
                    detailPanelContent plan
    in
    row [ width fill, paddingEach { top = 20, left = 0, right = 0, bottom = 0 } ]
        [ column [ width (fillPortion 7), height fill, alignTop ] tree
        , column
            [ width (fillPortion 3 |> maximum 500)
            , height fill
            , alignTop
            , padding 5
            , Border.widthEach { left = 1, right = 0, top = 0, bottom = 0 }
            , Border.color Color.grey
            ]
          <|
            details
        ]






There is a new value called details defined in the let; it deals with the selected node. If selectedNode has a value, then it calls a new function detailPanelContent to generate a list of elements to populate the detail panel. This function needs to handle different node types similarly to planNodeTree: 



detailPanelContent : Plan -> List (Element msg)
detailPanelContent plan =
    let
        attr name value =
            wrappedRow [ width fill ]
                [ el
                    [ width (px 200)
                    , paddingEach { right = 10, left = 10, top = 3, bottom = 3 }
                    , alignTop
                    ]
                  <|
                    text name
                , paragraph [ width fill, Font.bold, scrollbarX ] [ text value ]
                ]

        header name =
            el [ paddingEach { top = 10, bottom = 5, left = 10, right = 0 } ] <|
                el
                    [ Font.bold
                    , Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
                    , Border.color lightGrey
                    ]
                <|
                    text name

        commonAttrs common =
            [ attr "Startup cost" <| String.fromFloat common.startupCost
            , attr "Total cost" <| String.fromFloat common.totalCost
            , attr "Schema" common.schema
            ]
    in
    case plan of
        PCte node ->
            commonAttrs node.common

        PGeneric node ->
            commonAttrs node

        PResult node ->
            commonAttrs node.common

        PSeqScan node ->
            commonAttrs node.common
                ++ [ header "Filter"
                   , attr "Filter" node.filter
                   , attr "Width" <| String.fromInt node.rowsRemovedByFilter
                   ]

        PSort node ->
            commonAttrs node.common
                ++ [ header "Sort"
                   , attr "Sort Key" <| String.join ", " node.sortKey
                   , attr "Sort Method" node.sortMethod
                   , attr "Sort Space Type" node.sortSpaceType
                   , attr "Sort Space Used" <| String.fromInt node.sortSpaceUsed
                   ]






We also have to update the UI code in planNodeTree to generate messages when the mouse cursor enters or leaves a node. To do that, we need to import a module called Element.Events, which provides us two attributes: onMouseEnter and onMouseLeave. We will use these attributes in the definition of treeNode within planTreeNode:



planNodeTree : Plan -> List (Element Msg)
planNodeTree plan =
    let 
        nodeTypeEl nodeType = 
            el [ Font.bold ] <| text nodeType        

        treeNode node nodeDetails = 
            [ el
                [ Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
                , Border.color lightBlue
                , mouseOver [ Background.color lightYellow ]
                , padding 4
                , onMouseEnter <| MouseEnteredPlanNode plan
                , onMouseLeave <| MouseLeftPlanNode plan                
                ] <|
                paragraph [] ((nodeTypeEl node.common.nodeType) :: nodeDetails)
            , childNodeTree node.common.plans
            ]
    in 
    ...






Finally, we have something that’s capable of displaying useful information:

    Plan tree with an attribute panel shown on hover
  




Summary


In this chapter, we started writing code for the query execution plan visualiser. We looked at some of the available productivity tools. We started constructing a basic UI with a little bit of interactivity. We looked at how the code can be broken down into modules. In the remainder of the chapter, we dealt with JSON parsing and displaying parsing results in a UI with basic interactivity.








Growing the Application: Server Requests, JS Interop, Code Organisation


Overview


In this chapter, we are going to add a few features to our application. As happens in real life, this doesn’t result in a nice linear progression, so this chapter will address a few different facets of Elm. 


You will learn about using commands to perform operations with side effects. You will also see how to make requests to a server and then handle responses. You will find out how to use ports, commands, subscriptions and flags to exchange data with JavaScript. Finally, I will show an approach to refactoring and organising the application code as it grows. 


New Functionality


Let’s say that the main feature we want to add is a page where a user can view their saved query plans. This means that we need to allow users to login first, and then we need to implement an interaction with the server to retrieve previously saved plans. I’m going to omit the code for saving a new plan because it doesn’t require any new concepts – it’s still mainly about making requests and handling responses. I will also defer implementing user registration until the next chapter.


Some new UI is going to be required: a login page and a page that lists saved plans. These pages need to be wired into the application (the update function, messages, and so on). At this point, it’s a good idea replace the “Menu” label in the navigation bar with an actual working menu button so we can switch between pages. 


Adding the menu


We are going to use a new style of button for the menu. In contrast to the green button we concocted before, this one is going to be of a more subdued grey colour, along with a few other visual tweaks. This gives me an opportunity to demonstrate how elm-ui styles can be organised. Let’s create a new module called Attr, and put the attributes for the two button types in there: 



module Attr exposing (..)

import Color exposing (..)
import Element exposing (..)
import Element.Background as Background
import Element.Border as Border
import Element.Font as Font


greenButton : List (Attribute msg)
greenButton =
    [ Background.color green
    , Border.color darkGreen
    , Border.rounded 3
    , Border.widthEach { bottom = 3, top = 0, right = 0, left = 0 }
    , Font.bold
    , Font.color white
    , paddingXY 20 6
    ]


greyButton : List (Attribute msg)
greyButton =
    [ Background.color lightGrey
    , Border.color grey
    , Border.rounded 3
    , Border.widthEach { bottom = 1, right = 1, top = 0, left = 0 }
    , Font.bold
    , Font.color darkCharcoal
    ]






Note that greenButton and greyButton are simply lists of attributes. If we had more complex styles, then we could define functions that set different colours based on their arguments, or add extra items like padding, or join multiple lists together. The point is, we can use all of Elm’s facilities to manage attributes without unnecessary duplication. 


Typically, I start by defining attributes directly in the view code, and only extract them into a module like Attr when the same attributes are used in multiple places, which is exactly the same approach as for any other code. 


Now we can change the definition of the “Go!” button on the input page to use the attributes from Attr:



Input.button 
    (Attr.greenButton ++ [ width (px 200), height (px 40), alignRight ])
    { onPress = Just SubmitPlan
    , label = el [ centerX ] <| text "Go!"
    }






Next, let’s add the “Menu” button to the navigation bar: 



navBar : Element Msg
navBar =
    row
        [ width fill
        , paddingXY 60 10
        , Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
        , Border.color blue
        ]
        [ el [ alignLeft ] <| text "VisExp"
        , Input.button (Attr.greyButton ++ [ padding 5, alignRight, width (px 80) ])
            { onPress = Just ToggleMenu
            , label = el [ centerX ] <| text "Menu"
            }
        ]






Now, we have two different button styles in the application: 

A few straightforward changes are needed to wire in the menu panel, which I’m not going to show code for:



  	Adding a ToggleMenu message

  	Adding isMenuOpen boolean flag to the model

  	Toggling isMenuOpen in response to the ToggleMenu message in the update function.




Following these changes, we can add the menu panel itself: 



menuPanel : Model -> Element Msg
menuPanel model =
    let
        items =
            [ el [ pointer, onClick CreatePlan ] <| text "New plan" 
            , el [ pointer, onClick RequestLogin ] <| text "Login"
            ]

        panel = 
            column
                [ Background.color Color.white
                , Border.widthEach { left = 1, right = 0, top = 0, bottom = 0 }
                , Border.color Color.grey
                , Border.shadow 
                    { offset = ( 0, 0 )
                    , size = 1
                    , blur = 10
                    , color = Color.lightCharcoal 
                    }
                , Font.bold
                , Font.color Color.darkCharcoal
                , Font.family [ Font.sansSerif ]
                , width <| fillPortion 1
                , height fill
                , paddingXY 20 20
                , spacingXY 0 20
                ]
                items

        overlay = 
            el [ width <| fillPortion 4, height fill, onClick ToggleMenu ] none
    in
    if model.isMenuOpen then
        row [ width fill, height fill ] [ overlay, panel ]

    else
        none






If model.isMenuOpen is True, then we return the menu panel, otherwise we return a special element called none which produces no HTML. Actually, there is more than just the panel. Note that we defined a row [ width fill, height fill ], meaning that it will fill the whole viewport. This row contains two elements: overlay with a width <| fillPortion 4 attribute, and panel with width <| fillPortion 1. This means that overlay will take up 80% of the width of the page, and the panel will take up the remaining 20% on the right hand side. Both of the elements also have height fill, so they’ll extend to the full height of the viewport. The overlay is a transparent element that covers the rest of the page not obscured by the menu panel. It allows us to dismiss the panel when clicking somewhere outside it, because clicking on the overlay sends a ToggleMenu message. 


How do we make sure that the overlay and the menu panel appear in front of the content, and stay in front even if we scroll the page? To achieve that, elm-ui has a special attribute called inFront, which we can use in the view function: 



view : Model -> Browser.Document Msg
view model =
    let
        content =
            case model.currPage of
                DisplayPage ->
                    displayPage model

                InputPage ->
                    inputPage model
    in
    { title = "VisExp"
    , body =
        [ layout [ inFront <| menuPanel model ] <|
            column [ width fill, spacingXY 0 20 ]
                [ navBar
                , content
                ]
        ]
    }






When inFront is applied to the top-level layout, it’s equivalent to using position: fixed in CSS, so the overlay and menu panel will appear on top of other content, and cover the whole viewport even if the page is scrolled. 


This is what the application looks like with the menu open: 

Adding the login page


Next up, let’s add a login page. Once again, I’m going to skip over some scaffolding: 



  	Adding LoginPage to the Page custom type

  	Adding password and userName string fields to the model

  	Adding the lastError string field to the model for login errors

  	Adding ChangePassword String and ChangeUserName String messages to deal with input on the login page.




Then, a very simple login page can be implemented like this:



loginPage : Model -> Element Msg
loginPage model =
    column [ paddingXY 0 20, spacingXY 0 10, width (px 300), centerX ]
        [ Input.username Attr.input
            { onChange = ChangeUserName
            , text = model.userName
            , label = Input.labelAbove [] <| text "User name:"
            , placeholder = Nothing
            }
        , Input.currentPassword Attr.input
            { onChange = ChangePassword
            , text = model.password
            , label = Input.labelAbove [] <| text "Password:"
            , placeholder = Nothing
            , show = False
            }
        , Input.button Attr.greenButton
            { onPress = Just StartLogin
            , label = el [ centerX ] <| text "Login"
            }
        , el Attr.error <| text model.lastError
        ]

The view function needs to be expanded to include LoginPage: 



view : Model -> Browser.Document Msg
view model =
    let
        content =
            case model.currPage of
                DisplayPage ->
                    displayPage model

                InputPage ->
                    inputPage model

                LoginPage -> 
                    loginPage model
    in
    { title = "VisExp"
    , body =
        [ layout [ inFront <| menuPanel model ] <|
            column [ width fill, spacingXY 0 20 ]
                [ navBar
                , content
                ]
        ]
    }






Finally, the update function needs to handle a bunch of new messages, including RequestLogin to switch to the login page, and CreatePlan to switch to the input page: 



update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        ChangePassword s ->
            ( { model | password = s }, Cmd.none )

        ChangePlanText s ->
            ( { model | currPlanText = s }, Cmd.none )

        ChangeUserName s ->
            ( { model | userName = s }, Cmd.none )

        CreatePlan ->
            ( { model | currPage = InputPage, currPlanText = "" }, Cmd.none )

        MouseEnteredPlanNode plan ->
            ( { model | selectedNode = Just plan }, Cmd.none )

        MouseLeftPlanNode commonFields ->
            ( { model | selectedNode = Nothing }, Cmd.none )

        NoOp ->
            ( model, Cmd.none )

        RequestLogin ->
            ( { model | currPage = LoginPage, password = "", userName = "" }
            , Cmd.none 
            )

        StartLogin ->
            ( model, Cmd.none )

        SubmitPlan ->
            ( { model | currPage = DisplayPage }, Cmd.none )

        ToggleMenu ->
            ( { model | isMenuOpen = not model.isMenuOpen }, Cmd.none )






With the UI in place, we can move on to making the application submit user credentials to the server. This requires using commands – something we have neatly avoided so far, other than using Cmd.none as in the version of update shown above. 


Commands


Commands provide a way to carry out asynchronous operations with side effects. When you issue a command by returning it from the update function, it causes the Elm runtime to kick off the corresponding operation and subsequently return the result back to your program via a message:




  
    [image: Commands trigger operations in the runtime which eventually result in messages]
    Commands trigger operations in the runtime which eventually result in messages
  




As normal functions in Elm are pure, there’s no way to express something like making a call to a server or generating a random number in a regular function, because every call to such a function might return a different value. Instead, anything that has side effects is done via commands.    


The update function returns a command in a tuple with the changed model. We have actually been returning a command all along, except it’s been Cmd.none which does nothing:



    ToggleMenu ->
        ( { model | isMenuOpen = not model.isMenuOpen }, Cmd.none )






Server Requests


Server requests are made by sending particular commands to the Elm runtime. 


Let’s say that the server accepts POST requests with a JSON body containing credentials to the login endpoint, and GET requests to the plans endpoint:

The plans request has to include a custom header with the user’s session ID, and all responses are provided in JSON format.


In order to make requests, we need to use the elm/http package from the Elm core library. 


Making a login request


To send a login request to the server, we need to update the StartLogin branch of the update function (recall that the StartLogin message is generated by pressing the “Login” button):



update msg model =
    case msg of
        ...
        StartLogin ->
            ( model, login model.userName model.password ) 
        ...






login is a regular pure function that happens to return a command: 



login : String -> String -> Cmd Msg
login userName password =
    let
        body =
            Http.jsonBody <|
                Json.Encode.object
                    [ ( "userName", Json.Encode.string userName )
                    , ( "password", Json.Encode.string password ) 
                    ]

        responseDecoder =
            Json.Decode.field "sessionId" Json.Decode.string
    in
    Http.post
        { url = serverUrl ++ "login"
        , body = body
        , expect = Http.expectJson FinishLogin responseDecoder
        }






Http.post returns a command that will generate a message once the request is complete. The http package provides functions such as get, post and a more general-purpose request to generate commands that trigger HTTP requests. Since we need to make a POST request to login, we’ve used Http.post here. 


The expect field in the argument determines how the response will be handled, and is a value of type Expect msg. The http package provides several functions to construct these values: expectString, expectJson, expectBytes and expectWhatever (which ignores the response). In this case, the session ID returned by the server is a JSON value, so we used expectJson, which has this signature: 



expectJson : (Result Error a -> msg) -> Decoder a -> Expect msg






The first argument to expectJson is a function that takes a Result value and returns a message, and the second argument is a JSON decoder.


How do we figure out what to pass to it? Type variable a in this signature stands for the type of response data that will be sent by the server. In response to this request, the server will send the session ID as a string, so a is String. Result is a generic type with two type variables:



type Result error value 
    = Ok value
    | Err error






We now know that we need to use Http.Error and String in place of its type variables: (Result Http.Error String -> msg). We also know that message constructors are functions that return a message value, so if we add a FinishLogin message with a result payload to the Msg type, its constructor with the type signature (Result Http.Error String) -> Msg will be just what we need:



type Msg
    ...
    | FinishLogin (Result Http.Error String)   
    ...






The type of responseDecoder is Decoder String so we can use it as the second argument to expectJson. 


As a result of returning the command produced by Http.post in the update function, the Elm runtime will kick off the request behind the scenes. Once the request completes, either successfully or otherwise, we are going to get a FinishLogin message, so let’s handle it: 



update msg model =
    case msg of
        ...
        FinishLogin (Ok sessionId) ->
            ( { model | sessionId = Just sessionId, currPage = InputPage }
            , Cmd.none 
            )

        FinishLogin (Err error) ->
            ( { model | lastError = httpErrorString error }, Cmd.none )
        ...






The interesting thing here is that we can use pattern matching and handle both a successful result and an error within the case expression in update – no need for a nested case!


The error value is of the type Http.error which is a custom type encoding various failure scenarios. httpErrorString is a function that converts this type to a string: 



httpErrorString : Http.Error -> String
httpErrorString error =
    case error of
        Http.BadBody message ->
            "Unable to handle response: " ++ message

        Http.BadStatus statusCode ->
            "Server error: " ++ String.fromInt statusCode

        Http.BadUrl url ->
            "Invalid URL: " ++ url

        Http.NetworkError ->
            "Network error"

        Http.Timeout ->
            "Request timeout"






lastError is used on the login page to display the error message when a login request fails.  


We’ll keep the session ID returned by the server in the model (as Maybe String) so we can send it along with a request for the saved plans. 


Making a request for saved plans


As we have the session ID in the model now, we can add the request for saved plans as well. We’ll kick off this request along with switching to the saved plans page:  



update msg model =
    case msg of
        ...
        RequestSavedPlans ->
            ( { model | currPage = SavedPlansPage }
            , getSavedPlans model.sessionId 
            )
        ...







getSavedPlans : Maybe String -> Cmd Msg
getSavedPlans sessionId =
    Http.request
        { method = "GET"
        , headers = 
            [ Http.header "SessionId" <| Maybe.withDefault "" sessionId ]
        , url = serverUrl ++ "plans"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectJson FinishSavedPlans decodeSavedPlans
        }






Even though it’s a GET request, we have to use Http.request rather than Http.get because this request requires a custom HTTP header with the session ID. The request function takes a record with various parameters. 


You’ll notice that I’m being a bit cavalier with the session ID added to the request headers. If the session ID isn’t set, then I send an empty string supplied by Maybe.withDefault. I can claim that getSavedPlans will never be called without a valid session ID, and that the server will reject a request without it anyway, and leave it at that. However, I want to highlight that this is the kind of design decision that eventually leads to bugs. In the next chapter, we’ll talk about some ways of avoiding bugs at the point of software design, and improve this bit of code. 


As part of constructing the request, I’ve used a JSON decoder called decodeSavedPlans. It is similar to decoders we constructed earlier in the book: 



type alias PlanVersion =
    { version : Int
    , createdAt : String
    , planText : String
    }


type alias SavedPlan =
    { id : String
    , name : String
    , versions : List PlanVersion
    }


decodePlanVersion : Decode.Decoder PlanVersion
decodePlanVersion =
    Decode.succeed PlanVersion
        |> required "version" Decode.int
        |> required "createdAt" Decode.string
        |> required "planText" Decode.string


decodeSavedPlans : Decode.Decoder (List SavedPlan)
decodeSavedPlans =
    Decode.list
        (Decode.succeed SavedPlan
            |> required "id" Decode.string
            |> required "name" Decode.string
            |> required "versions" (Decode.list decodePlanVersion)
        )






The SavedPlan record contains a list of plan versions, each of which in turn contains a query plan. The idea is that the user can save several versions of the plan against a single query as they work on optimising it. 


Correspondingly, the FinishSavedPlans message has to carry a list of SavedPlan values wrapped in Result:



type Msg
    ...
    | FinishSavedPlans (Result Http.Error (List SavedPlan))
    ...






The handling of the FinishSavedPlans message is no different from FinishLogin (savedPlans is a field of type List SavedPlan):



update msg model =
    case msg of
        ...
        FinishSavedPlans (Ok savedPlans) ->
            ( { model | savedPlans = savedPlans }, Cmd.none )

        FinishSavedPlans (Err error) ->
            ( { model | lastError = httpErrorString error }, Cmd.none )
        ...






Displaying the saved plans


The saved plans page provides an opportunity to use elm-ui table. The function ends up being fairly long, but hopefully straightforward: 



savedPlansPage : Model -> Element Msg
savedPlansPage model =
    let
        annotateVersion name planVersion =
            { version = planVersion.version
            , planText = planVersion.planText
            , createdAt = planVersion.createdAt
            , name = name
            }

        annotateVersions savedPlan =
            List.map (annotateVersion savedPlan.name) savedPlan.versions

        tableAttrs =
            [ width (px 800)
            , paddingEach { top = 10, bottom = 50, left = 10, right = 10 }
            , spacingXY 10 10
            , centerX
            ]

        headerAttrs =
            [ Font.bold
            , Background.color Color.lightGrey
            , Border.color Color.darkCharcoal
            , Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
            , centerX
            ]
    in
    table tableAttrs
        { data = List.concatMap annotateVersions model.savedPlans
        , columns =
            [ { header = el headerAttrs <| text "Plan name"
              , width = fill
              , view =
                    \plan ->
                        el
                            [ Font.underline
                            , mouseOver [ Font.color lightCharcoal ]
                            , onClick <| ShowPlan plan.planText
                            ]
                        <|
                            text plan.name
              }
            , { header = el headerAttrs <| text "Creation time"
              , width = fill
              , view = .createdAt >> text
              }
            , { header = el headerAttrs <| text "Version"
              , width = fill
              , view = .version >> String.fromInt >> text
              }
            ]
        }






Here is how it appears in the browser: 

The table function takes a list of attributes, as usual, followed by a record containing the data along with column definitions. To populate the data field, we transform model.savedPlans into a list of records containing three fields: plan name, plan creation date, and plan version. 


A column definition consists of an element for the header, a width specification as an attribute, and a view function which takes one of the data records and converts it into an element that will form one of the table cells. The first column of our table will show plan names as links that generate ShowPlan messages when clicked. In update, we will update the model to show the corresponding plan: 



update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        ...
        ShowPlan planText ->
            ( { model | currPlanText = planText, currPage = DisplayPage }
            , Cmd.none 
            )
        ...






Additionally, the view function needs to be updated to handle the new page option – I’m not going to show that as it’s a trivial change.


Finally, we need to add a link to the new page in the menu panel to make it accessible. We’re going to vary the content of the menu based on the presence of a session ID in the model, which indicates that the user has logged in:



menuPanel : Model -> Element Msg
menuPanel model =
    let
        items =
            [ el [ pointer, onClick CreatePlan ] <| text "New plan" ]
                ++ (case model.sessionId of
                        Just _ ->
                            [ el [ pointer, onClick RequestSavedPlans ] <|
                                text "Saved plans"
                            , el [ pointer, onClick RequestLogout ] <| text "Logout"
                            ]

                        Nothing ->
                            [ el [ pointer, onClick RequestLogin ] <| text "Login" ]
                   )
        ...






Following these changes, the “Saved plans” link will be present in the menu after the user has logged in, and will take the user to the saved plans page we’ve just created. 


Data Exchange with the Browser and JS Code


An Elm program exists within the browser environment, and usually needs to interact with the browser functionality or supporting JavaScript code. For this reason, Elm provides facilities to interface with the outside world. Before we get into code examples, I’d like to give an overview of the data exchange facilities. Elm provides a language level mechanism for exchanging data with the browser and JavaScript: 




  
    [image: Data exchange facilities mediated by the runtime]
    Data exchange facilities mediated by the runtime
  




It’s mediated by the Elm runtime so it’s conceptually quite similar to the way commands work, and in fact, commands are involved here as well. This mechanism is driven by ports, subscriptions, and flags. Let’s take a quick look at each of these.


Ports


Ports work in tandem with commands and subscriptions. They define what data your program receives from its environment, and what data it sends out. 


From the point of view of Elm code, a port for outgoing data is defined as a function taking some data and returning a command. This function can then be used in update to generate a command. A port for incoming data is defined as a function returning a Sub value which is the type for subscriptions. 


Subscriptions


Subscriptions are a mechanism for receiving data from the outside environment, which for an Elm program consists of the browser and possibly JavaScript code. This means things like time, key presses, position updates, and data from JavaScript. 


A point to keep in mind is that both subscriptions and commands are descriptions of what you want to happen – in other words, they are data rather than actions. It’s up to the Elm runtime to take these descriptions and use them as instructions to perform actions. Based on the subscriptions, the runtime generates messages which flow back into the update function.


Flags


There is one more way to receive data from JavaScript besides subscriptions. An application can receive data at the start of execution, and incorporate it into its initial state. This data is referred to as “flags” in Elm as it’s similar to passing some flags into a program on the command line. 


Data validation and conversion


In the case of both incoming ports and flags, the data is validated to ensure that your Elm code doesn’t receive something unexpected. If the data fails validation, a runtime exception is thrown. Both incoming and outgoing ports allow all valid JSON values to be passed. 


Some type conversions happen when passing data through ports:



  	Booleans, strings, integers and floating point numbers convert trivially

  	
Maybe values: a Nothing appears as null on the JavaScript side, and a Just 1 appears as 1. 

  	Both lists and arrays on the Elm side correspond to JavaScript arrays

  	Records correspond to JavaScript objects

  	Tuples correspond to fixed-length, mixed-type JavaScript arrays

  	Values of type Json.Encode.Value correspond to arbitrary JSON.




Saving session ID to localStorage



In order to demonstrate the use of commands and flags, we will use localStorage to store the session ID which the server provides upon login. This allows the user to stay logged in after closing and reopening the page. 


As it’s a kind of side effect, we need to pass the session ID value out of the Elm program to JavaScript, then use JavaScript to save it in localStorage and also read it back out so it can be passed to the Elm program via flags on page load.


First of all, let’s define a port on the Elm side. Any module with port definitions has to be marked as a port module, eg port module Main exposing (main). This makes it easy to check whether a given module interacts with JavaScript. While ports can be declared in the Main module, and you can have multiple port modules, it’s usually a good idea to have a single dedicated port module because then it’s easy to see all of the possible ways your program can interact with JavaScript in one place. So I’m going to start a new module called Ports and declare the port there:



port module Ports exposing (..)

port saveSessionId : Maybe String -> Cmd msg






The definition of a port starts with the port keyword and looks like a function type annotation without a function definition. This function has to take a single argument of a suitable type (say Int or Maybe String), and return Cmd msg – the only kind of signature the compiler will allow for ports. 


Once I import the Ports module in Main.elm, I can use this port to send session ID values:



update msg model =
    case msg of
        ...
        FinishLogin (Ok sessionId) ->
            ( { model | sessionId = Just sessionId, currPage = InputPage }
            , saveSessionId <| Just sessionId 
            )
        ...






Subsequently, we need to setup the JavaScript side of things. This requires manually creating index.html which includes the generated JavaScript for our application because we will be using the app object:



<!doctype html>
<html>
<head>
    <title>VisExp</title>
    <script src="elm.js"></script>
</head>

<body>
<script>
    let app = Elm.Main.init()
</script>
</body>
</html>






elm.js can be generated with elm make (elm make src/Main.elm --output elm.js), or, if using elm-live, by running elm-live src/Main.elm -- --output elm.js. 


In order to receive session ID values that are sent from the Elm side, we need to subscribe to them:



    let app = Elm.Main.init()
    app.ports.saveSessionId.subscribe((sessionId) => {
        if (sessionId === null)
            localStorage.removeItem("sessionId")
        else
            localStorage.setItem("sessionId", sessionId)
    })






The app object returned by init has a ports property which contains a key for every outgoing port defined on the Elm side. The keys match port function names, so we’re interested in the saveSessionId key whose value is an object with a subscribe function. 


One thing to keep in mind: due to dead code elimination, unless you actually use a port in your Elm code, the corresponding key in app.ports will not be present. It isn’t enough just to declare the port. In fact, app.ports itself may not be defined if none of the ports are used on the Elm side. 


The subscribe function takes a callback which receives the value sent through the port, and we can save the session ID to localStorage in this callback. I also added a bit of code to clear out the saved value if the received value is null, which is something we can do on logout. 


As a result of these changes, the session ID will appear in localStorage after login. 


Passing saved session ID back to Elm


To address the reciprocal flow of information, we need to pass the saved session ID to Elm in the flags object. This object is passed as an additional property of the init argument:



    let app = Elm.Main.init({
        flags: {sessionId: localStorage.getItem("sessionId")}
        })






On the Elm side, we need to make a few changes to handle the flags. The Flags type alias needs to change from the unit type to a record, and we need to use flags in init to set the initial value of the session ID in the model:



type alias Flags =
    { sessionId : Maybe String
    }

init : Flags -> ( Model, Cmd Msg )
init flags =
    ( { currPage = InputPage
      , currPlanText = ""
      , isMenuOpen = False
      , lastError = ""
      , password = ""
      , savedPlans = []
      , sessionId = flags.sessionId
      , selectedNode = Nothing
      , userName = ""
      }
    , Cmd.none
    )






The structure of the flags object passed from JavaScript has to match the structure defined for the Flags type alias, otherwise you will get a runtime exception.


This addition allows us to use the session ID passed from JavaScript to initialise the sessionId field in the model, which closes the loop: now session ID is saved to localStorage on login and then loaded back from localStorage into the Elm program on page load. After this change, the login state of the application will be preserved across page reloads.


Connecting subscriptions to JavaScript code


There is one more piece of the puzzle, and that is how to send data from JavaScript to Elm at any time rather than just on page load. 


Let’s implement a way to dump the current state of the model to JavaScript console. We’ll do that by sending information into an incoming port, which will trigger a subscription on the Elm side. We’ll start by defining the incoming port, dumpModel, in the Ports module:



port module Ports exposing (dumpModel, saveSessionId)


port dumpModel : (() -> msg) -> Sub msg

port saveSessionId : Maybe String -> Cmd msg






The incoming port must be a function that takes one argument and returns a Sub msg. The argument is itself a function that takes a single argument which defines the type of data sent through the port, and returns msg. In our case, we don’t really want to send data, all we want to do is signal to the Elm program that we want the model, so I used the unit type () as the data type. 


To get this port to generate messages, we have to use it in the subscriptions function: 



subscriptions : Model -> Sub Msg
subscriptions model =
    dumpModel DumpModel






There is a nuance to do with DumpModel: it has to carry the unit type rather than nothing because of the requirements for the port type signature, so in the Msg type we have to add it as: 



type Msg
    ...
    | DumpModel ()
    ...






Rather than sending the model through an outgoing port, I’ll cheat a little and simply dump it to the console with the help of Debug.log:



update msg model =
    case msg of
        ...
        DumpModel () ->
            ( Debug.log "model" model, Cmd.none )
        ...






Keep in mind that this will not work if you use the --optimize flag to generate a production build of the application.


From the JavaScript side, we can trigger the logging by sending a value into the dumpModel port:



app.ports.dumpModel.send(null)






If we try it from the JS console in the browser, we get the model dumped back into the console as a string: 



model: { 
  currPage = InputPage, 
  currPlanText = "", 
  isMenuOpen = False, 
  lastError = "", 
  password = "", 
  savedPlans = [], 
  selectedNode = Nothing, 
  sessionId = Just "ABC", 
  userName = "" 
}






send calls can occur at any time in your JavaScript and can be used to send data to Elm. In this example we only sent null, but you can send JavaScript objects and values of primitive types to an appropriately defined port. 


Keeping the session alive with periodic requests


Another thing that’s sometimes needed for session management, particularly in the case of single page applications, is sending a periodic update to the server to let it know that the session is still active. It might be a little bit at odds with having persistent sessions, but let’s add it anyway because it allows us to take a look at working with time. 


The Elm core library includes the elm/time package, so we don’t need to mess with ports and JavaScript to achieve this. Once we import the Time module, we can add a subscription which will periodically generate a message for us:



subscriptions : Model -> Sub Msg
subscriptions model =
    Sub.batch
        [ dumpModel DumpModel
        , Time.every (100 * 1000) SendHeartbeat
        ]






Sub.batch allows us to combine multiple subscriptions into a single subscription. The Time.every call will result in the SendHeartbeat message being generated every 100 seconds (the first argument to every is the number of milliseconds). SendHeartbeat has to be defined to carry a Time.Posix value:



type Msg
    ...
    | SendHeartbeat Time.Posix
    ...






Then, in update, we need to generate a request in response to this message:



update msg model =
    case msg of
        ...
        SendHeartbeat _ ->
            ( model, sendHeartbeat model.sessionId )
        ...






The implementation of sendHeartbeat is very similar to the requests we implemented earlier:



sendHeartbeat : Maybe String -> Cmd Msg
sendHeartbeat sessionId =
    Http.request
        { method = "POST"
        , headers =
            [ Http.header "SessionId" <| Maybe.withDefault "" sessionId ]
        , url = serverUrl ++ "heartbeat"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectWhatever <| always NoOp
        }






We’re using Http.expectWhatever because we don’t care what response is returned, and passing it always NoOp which means that the request will always result in a NoOp message being generated, regardless of whether the request is successful. 


Receiving keyboard events


One more type of subscriptions available in the core library is keyboard events. Let’s use it to add keyboard shortcuts for creating a new plan and for jumping to the saved plans page. 


We need to import Browser.Events to get the necessary subscription:



import Browser.Events exposing (onKeyPress)






This module allows you to subscribe to a number of events in addition to key presses: animation frame, mouse events, window size changes, and visibility changes caused by the user switching to a different tab or window. 


We are going to add a subscription to keydown events: 



subscriptions : Model -> Sub Msg
subscriptions model =
    Sub.batch
        [ dumpModel DumpModel
        , Time.every (100 * Time.second) SendHeartbeat
        , onKeyPress <| keyDecoder model
        ]






onKeyPress requires a JSON decoder that will extract the relevant information from the event value (this is done to allow support for older browsers). The event value is a JSON representation of JavaScript KeyboardEvent. We can rely on the “key” property of the event:



keyDecoder : Model -> Json.Decode.Decoder Msg
keyDecoder model =
    Json.Decode.field "key" Json.Decode.string
        |> Json.Decode.map (keyToMsg model)
        
keyToMsg : Model -> String -> Msg
keyToMsg model s =
    case ( s, model.sessionId ) of
        ( "s", Just id ) ->
            RequestSavedPlans

        ( "n", _ ) ->
            CreatePlan

        _ ->
            NoOp        






We are creating two shortcuts: pressing “s” takes the user to the saved plans page, and pressing “n” takes the user to the input page to create a new plan. Since subscriptions takes the model as its argument, we can pass the model through to keyToMsg to only handle the “s” shortcut when the user is logged in.  


This implementation has a flaw in that these shortcuts are triggered even when typing into the input text area, so it would be better to use something like Alt+Shift+N/Option+Shift+N and Alt+Shift+S/Option+Shift+S as shortcuts. We can rewrite our decoder to handle such shortcuts:



keyDecoder : Model -> Json.Decode.Decoder Msg
keyDecoder model =
    Json.Decode.map2 Tuple.pair
        (Json.Decode.field "altKey" Json.Decode.bool)
        (Json.Decode.field "shiftKey" Json.Decode.bool)
        |> Json.Decode.andThen
            (\altAndShiftFlags ->
                case altAndShiftFlags of
                    ( True, True ) ->
                        Json.Decode.field "code" Json.Decode.string
                            |> Json.Decode.map (keyToMsg model)

                    _ ->
                        Json.Decode.succeed NoOp
            )

keyToMsg : Model -> String -> Msg
keyToMsg model code =
    case ( code, model.sessionId ) of
        ( "KeyS", Just id ) ->
            RequestSavedPlans

        ( "KeyN", _ ) ->
            CreatePlan

        _ ->
            NoOp            






Here, we’re also extracting the altKey and shiftKey flags from the keyboard event object, and using the code field instead of the key field. The code field contains values like “KeyS” and “KeyN” to indicate which key has been pressed. Note that the code field may not be a good solution if dealing with non-QWERTY keyboard layouts. I’ve only used it as an example. 


And now, we have talked about all the aspects of exchanging data with the browser and JavaScript, both via ports and via subscriptions made available by various packages. 


Summary


As we implemented additional functionality in the course of this chapter, you learned about commands as well as making requests to the server and handling responses. You saw examples of using ports, commands, subscriptions and flags to exchange data with JavaScript. We added a time-based subscription as well as a couple of keyboard shortcuts for navigation.








Structuring the Code: Extracting Modules and Organising Data (Plus Bonus: Input Validation)


Overview


As the code base grows, questions arise about how it can be structured to remain easy to understand and maintain.


A lot of developers coming to Elm have questions about breaking up the application into some sort of “components” with their own state. However, there’s an easier way to think about the problem that goes a long way and doesn’t require any new concepts: it’s all just types and functions organised into modules! 


If a function is getting too long, we can break it down into several functions. One level up, if a file/module is getting too long, we can break it into two or more modules to logically group functions and isolate some of the implementation details so that they can only be used within the bounds of the module. 


Let’s say Main.elm that we’ve built over the course of previous chapters has grown too large and is starting to become unmanageable. Which parts of the code would we extract into a module? There are four main approaches, in the order of complexity: 



  	Extract helper functions into a module

  	Extract a subset of view code into a module 

  	Extract non-UI functionality including state related functions into a module

  	Extract a subset of the model, update and view code into a module. 




The first approach is trivial, and we’ve already applied it when we created the PlanParsers.Json and Attr modules. The second approach allows us to extract reusable widget code into a separate module while leaving all of the logic to do with managing widget state and messages in place. The third approach addresses the issue of a growing Msg type and a complex update function along with extracting other functions. The last approach produces an approximation of what would be stateful components in other languages/frameworks, and is useful for managing larger code bases. 


However, the important thing to understand is that no matter what you do, you’re still operating in terms of functions and modules. You are still passing a single update function and a single view function to Browser.document or one of the other functions that generate a Program. All the messages the runtime generates will still go into that update function and nowhere else. Even if a module contains an update function, there is no mechanism for it to receive messages, so the notion of components in Elm is always somewhat artificial. 


So which approach is the “correct” one? The answer is: it depends on the kind of code that you have, and you may well end up using a mix of them in a single project.


Finally, zooming out a bit further, how do we go about organising modules? One approach is to organise code into views, requests, and data. Since many applications are organised into pages, another approach that works very well is to have a module per page. It’s also possible to delineate boundaries based on functional areas (authentication, registration, parsing and so on in the context of our sample application). A good thing about Elm is that it provides enough flexibility for you to have these options.  


In this chapter, we are going to examine different ways of splitting up our code into manageable parts. 


Extracting View Code


Let’s extract the code for displaying a tree of plan nodes into a separate module, while leaving all the related messages and parts of the update function in Main.elm. This approach is particularly useful if you have some kind of reusable UI widgets. 


Previously, we defined displayPage as follows: 



displayPage : Model -> Element Msg
displayPage model =
    let
        tree =
            case Json.Decode.decodeString decodePlanJson model.currPlanText of
                Ok planJson ->
                    planNodeTree planJson.plan

                Err err ->
                    [ text <| Json.Decode.errorToString err ]

        details =
            case model.selectedNode of
                Nothing ->
                    [ text "" ]

                Just plan ->
                    detailPanelContent plan
    in
    row [ width fill, paddingEach { top = 20, left = 0, right = 0, bottom = 0 } ]
        [ column [ width (fillPortion 7), height fill, alignTop ] tree
        , column
            [ width (fillPortion 3 |> maximum 500)
            , height fill
            , alignTop
            , padding 5
            , Border.widthEach { left = 1, right = 0, top = 0, bottom = 0 }
            , Border.color Color.grey
            ]
          <|
            details
        ]






Let’s move the code for rendering a tree as an Element into a render function in a file called PlanTree.elm. We can see from the displayPage function that render needs to take a plan (of type PlanJson) as well as the currently selected node (of type Maybe Plan) as its arguments: 



module PlanTree exposing (render)

import Color exposing (..)
import Element exposing (..)
import Element.Background as Background
import Element.Border as Border
import Element.Events exposing (..)
import Element.Font as Font
import PlanParsers.Json exposing (..)


type alias SelectedNode =
    Maybe Plan


render : PlanJson -> SelectedNode -> Element msg
render planJson selectedNode =
    let
        details =
            case selectedNode of
                Nothing ->
                    [ text "" ]

                Just plan ->
                    detailPanelContent plan
    in
    row [ width fill, paddingEach { top = 20, left = 0, right = 0, bottom = 0 } ]
        [ column [ width (fillPortion 7), height fill, alignTop ] <|
            planNodeTree planJson.plan
        , column
            [ width (fillPortion 3 |> maximum 500)
            , height fill
            , alignTop
            , padding 5
            , Border.widthEach { left = 1, right = 0, top = 0, bottom = 0 }
            , Border.color Color.grey
            ]
          <|
            details
        ]






If you look closely, you’ll notice that the return type of render is Element msg rather than Element Msg. The idea is that PlanTree should be generic with regards to messages because messages are still dealt with in the main module while the PlanTree module is only concerned with rendering the UI. 


Correspondingly, displayPage in Main.elm becomes much simpler: 



displayPage : Model -> Element Msg
displayPage model =
    case Json.Decode.decodeString decodePlanJson model.currPlanText of
        Ok planJson ->
            PlanTree.render planJson model.selectedNode

        Err err ->
            el [] <| text <| Json.Decode.errorToString err






We also need to move all the functions called by render: planNodeTree, childNodeTree, and detailPanelContent. When we move planNodeTree, the compiler will let us know that this presents us with a problem because it’s using message constructors, MouseEnteredPlanNode and MouseLeftPlanNode: 



planNodeTree : Plan -> List (Element msg)
planNodeTree plan =
    let
        nodeTypeEl nodeType =
            el [ Font.bold ] <| text nodeType

        treeNode node nodeDetails =
            [ el
                [ Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
                , Border.color lightBlue
                , mouseOver [ Background.color lightYellow ]
                , padding 4
                , onMouseEnter <| MouseEnteredPlanNode plan
                , onMouseLeave <| MouseLeftPlanNode plan
                ]
              <|
                paragraph [] (nodeTypeEl node.common.nodeType :: nodeDetails)
            , childNodeTree node.common.plans
            ]
    in
    ...






Since we don’t want either messages or parts of the update function to end up in PlanTree.elm, what we need to do is parametrise planNodeTree. It needs to take two functions of type Plan -> msg which we will pass to onMouseEnter and onMouseLeave. Consequently, we need to pass these down from render. Let’s introduce a generic Config record and make it an argument of render to achieve this: 



type alias Config msg =
    { onMouseEnteredNode : Plan -> msg
    , onMouseLeftNode : Plan -> msg
    }

render : Config msg -> PlanJson -> SelectedNode -> Element msg
render config planJson selectedNode =
    ...






planNodeTree and childNodeTree will also take a Config value as an argument:



childNodeTree : Config msg -> Plans -> Element msg
childNodeTree config (Plans plans) =
    column [ paddingEach { left = 20, right = 0, top = 0, bottom = 0 } ] <|
        List.concatMap (planNodeTree config) plans


planNodeTree : Config msg -> Plan -> List (Element msg)
planNodeTree config plan =
    let
        nodeTypeEl nodeType =
            el [ Font.bold ] <| text nodeType

        treeNode node nodeDetails =
            [ el
                [ Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
                , Border.color lightBlue
                , mouseOver [ Background.color lightYellow ]
                , padding 4
                , onMouseEnter <| config.onMouseEnteredNode plan
                , onMouseLeave <| config.onMouseLeftNode plan
                ]
              <|
                paragraph [] (nodeTypeEl node.common.nodeType :: nodeDetails)
            , childNodeTree config node.common.plans
            ]
    in
    ...






To close the loop, displayPage needs to construct and pass a config to render:



displayPage : Model -> Element Msg
displayPage model =
    let
        planTreeConfig =
            { onMouseEnteredNode = MouseEnteredPlanNode
            , onMouseLeftNode = MouseLeftPlanNode
            }
    in
    case Json.Decode.decodeString decodePlanJson model.currPlanText of
        Ok planJson ->
            PlanTree.render planTreeConfig planJson model.selectedNode

        Err err ->
            el [] <| text <| Json.Decode.errorToString err






With this, the refactoring is complete. To reiterate what we’ve done: 



  	All the code to render a plan tree is in a separate module exposing a single function, render


  	All the messages generated from the plan tree, and the logic to handle them are still defined in the Main module. 




Therefore, we’ve created something like a reusable, stateless UI widget. 


Extracting Non-UI Functionality


Next, let’s take a look at extracting a portion of non-UI code into a module. One part of functionality responsible for quite a lot of the code in our project is related to authentication. We can break it out into a module called Auth.elm. It’s a somewhat contrived example, because we’ll see later in this chapter that for our application, it makes more sense to extract modules per page. However, it’s useful to illustrate the approach. 


We can start by slicing up some of the types: Msg, Model and the associated bits like the init function:



module Auth exposing (Flags, Model, Msg(..), init)

import Http
import Json.Decode
import Json.Encode
import Ports exposing (..)
import Time


type Msg
    = NoOp
    | ChangePassword String
    | ChangeUserName String
    | FinishLogin (Result Http.Error String)
    | SendHeartbeat Time.Posix
    | StartLogin


type alias Model =
    { lastError : String
    , password : String
    , sessionId : Maybe String
    , userName : String
    }


init : Maybe String -> Model
init sessionId =
    { lastError = ""
    , password = ""
    , sessionId = sessionId
    , userName = ""
    }






The Msg type in Main.elm will now have a new kind of message instead of the messages we moved to Auth:



type Msg
    ...
    | Auth Auth.Msg
    ...






By analogy with handling common node fields in JSON types, we’ll use a nested field (auth) for this data in the main model:



type alias Model =
    { auth : Auth.Model
    , currPage : Page
    , currPlanText : String
    , isMenuOpen : Bool
    , lastError : String
    , savedPlans : List SavedPlan
    , selectedNode : Maybe Plan
    }


init : Flags -> ( Model, Cmd Msg )
init flags =
    ( { auth = Auth.init flags.sessionId
      , currPage = InputPage
      , currPlanText = ""
      , isMenuOpen = False
      , lastError = ""
      , savedPlans = []
      , selectedNode = Nothing
      }
    , Cmd.none
    )






As far as functions go, the login and sendHeartbeat functions obviously belong in Auth.elm. One change is required to make them work after relocation. We originally referred to serverUrl, a top-level value, in these functions. serverUrl now has to become an argument to these functions; it doesn’t make sense to move it from Main.elm to Auth.elm as it isn’t authentication-specific: 



login : String -> String -> String -> Cmd Msg
login serverUrl userName password =
    let
        body =
            Http.jsonBody <|
                Json.Encode.object
                    [ ( "userName", Json.Encode.string userName )
                    , ( "password", Json.Encode.string password )
                    ]

        responseDecoder =
            Json.Decode.field "sessionId" Json.Decode.string
    in
    Http.post
        { url = serverUrl ++ "login"
        , body = body
        , expect = Http.expectJson FinishLogin responseDecoder
        }


sendHeartbeat : String -> Maybe String -> Cmd Msg
sendHeartbeat serverUrl sessionId =
    Http.request
        { method = "POST"
        , headers =
            [ Http.header "SessionId" <| Maybe.withDefault "" sessionId ]
        , url = serverUrl ++ "heartbeat"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectWhatever <| always NoOp
        }






By the way, the login signature with three undifferentiated string arguments is starting to become rather confusing, isn’t it? This kind of thing is a hotspot for bugs. In the next chapter, we will see how it can be made more foolproof.


We also need to cut out the relevant portions of the update function into a new update in the Auth module: 



update : String -> Msg -> Model -> ( Model, Cmd Msg )
update serverUrl msg model =
    case msg of
        ChangePassword p ->
            ( { model | password = p }, Cmd.none )

        ChangeUserName name ->
            ( { model | userName = name }, Cmd.none )

        FinishLogin (Ok sessionId) ->
            ( { model | sessionId = Just sessionId, userName = "", password = "" }
            , saveSessionId <| Just sessionId
            )

        FinishLogin (Err error) ->
            ( { model | lastError = httpErrorString error }, Cmd.none )

        NoOp ->
            ( model, Cmd.none )

        SendHeartbeat _ ->
            ( model, sendHeartbeat serverUrl model.sessionId )

        StartLogin ->
            ( model, login serverUrl model.userName model.password )






Keep in mind that this update function is written in terms of Auth.Model and Auth.Msg. It doesn’t know anything about the messages or the model in Main. 


One wrinkle to iron out is the use of httpErrorString in this function. Since we still need it in Main as well, we can’t just relocate it to the Auth module. It’s time to introduce the ubiquitous Utils module and move httpErrorString there, so we can use it in both places.


As a result of moving authentication related messages, all the references to them in Main.elm need to be updated. For example, StartLogin becomes Auth Auth.StartLogin, ChangeUserName becomes Auth << Auth.ChangeUserName (because we need to construct a function that takes a string), model.sessionId becomes model.auth.sessionId and so on. The compiler will point all of these out so it’s a mostly mechanical transformation. After that, what remains is plugging Auth.update into the main update. We can do it like this:



update msg model =
    case msg of
        Auth authMsg ->
            let
                ( authModel, authCmd ) =
                    Auth.update serverUrl authMsg model.auth

                currPage =
                    case authMsg of
                        Auth.FinishLogin (Ok _) ->
                            InputPage

                        _ ->
                            model.currPage
            in
            ( { model | auth = authModel, currPage = currPage }
            , Cmd.map Auth authCmd
            )
        ...






Let’s look at the interesting aspects of this clause. First, I used a let/in expression inside the case branch, which demonstrates that it’s just an expression rather than some specific syntax in function definitions. 


Second, I used destructuring inside the let to pull apart the results returned by Auth.update into two values, authModel and authCmd. Then, I calculated the new value for currPage because the user should end up on the input page in case of a successful login, or stay on the login page otherwise. I did it here rather than in the Auth module because currPage really belongs in the main module, but at the same time its value depends on the login result. 


Finally, I had a little discrepancy in commands I had to take care of. The return type of Auth.update is ( Auth.Model, Cmd Auth.Msg ) but the return type of the main update is ( Model, Cmd Msg ). The model side of things is easy to deal with – we just need to update the auth field of the main model using the result of Auth.update: 



{ model | auth = authModel, currPage = currPage }






The command side is a bit trickier. I can’t just return a command provided by Auth.update, so I have to perform a conversion like this somehow: Cmd Auth.Msg -> Cmd Msg. I happen to have a function which converts Auth.Msg -> Msg – it’s the Auth constructor in the main Msg type. So I need to apply it in the context of Cmd, and that’s exactly what Cmd.map is for:



Cmd.map : (a -> msg) -> Cmd a -> Cmd msg






Given Auth as the first argument, (Cmd.map Auth) is a function that will do the required conversion for us: Cmd Auth.Msg -> Cmd Msg.


At this point, the refactoring can be considered complete. For bonus points, the time subscription should really go into the Auth module too, because it’s an implementation detail that’s only relevant to authentication. It would be a matter of adding an Auth.subscriptions function and returning the relevant value from there. 


If we were doing something more complicated and there was more code, then it might warrant breaking up the code into a few more modules, for example:



Auth
  Helpers.elm
  Requests.elm
  State.elm






However, for now, a single module is sufficient. 


Extracting a Whole Page


The last approach to structuring code involves extracting a subset of the model, update and view code into a module. This method applies both to extracting particularly complex portions of the UI (for example, an interactive data grid with sorting and customisable columns), as well as whole pages. 


If your application UI is organised into pages, as is often the case, then it’s quite natural to create a module for each page, and to define the page-specific messages, the corresponding update logic, and the subset of the state pertaining to the page in its module, along with the view code to render the UI of the page. 


The page-specific code then needs to be wired into the main Msg type as well as the main update function. In fact, we’ve already added this kind of wiring when we created the Auth module. The only difference is that at that point we stopped short of moving any view code into the module. 


For a practical example, let’s add a registration page to our application. It will be quite similar to the login page (but more complex) and will look like this:

Writing the page module


To begin with, we need to define a model and some messages in Pages/Registration.elm:



module Pages.Registration exposing (Model, Msg(..), Platform(..), init)

import Http

type Msg
    = ChangePassword String
    | ChangeRepeatPassword String
    | ChangeUserName String
    | FinishRegistration (Result Http.Error String)
    | SelectPlatform Platform
    | StartRegistration
    | ToggleAcceptTerms Bool


type Platform
    = Aws
    | Azure
    | Heroku
    | SelfHosted


type alias Model =
    { errors : List String
    , hasAcceptedTerms : Bool
    , password : String
    , platform : Maybe Platform
    , repeatPassword : String
    , userName : String
    }


init : Model
init =
    { hasAcceptedTerms = False
    , errors = []
    , password = ""
    , platform = Nothing
    , repeatPassword = ""
    , userName = ""
    }






Since we’re going to have some page-specific state, it has to reside somewhere in the main model, but where in particular? We could have a field for every page in the main model: 



type alias Model = 
    { ...
    , displayPageState : Display.Model
    , registrationPageState : Registration.Model
    , savedPlansPageState : SavedPlans.Model
    , ...
    }






Obviously, this is rather unwieldy as we have to modify this structure every time we add a page. But also, since we don’t need to retain page state for inactive pages, we’ll have to add brittle code to reset page state every time the user navigates to another page. Instead, let’s recall that we already have a currPage field in the model: 



type Page
    = DisplayPage
    | ...
    | RegistrationPage
    | SavedPlansPage


type alias Model =
    { ...
    , currPage : Page
    , ...
    }






We can actually modify the Page type so that currPage carries the page-specific state for whatever page happens to be active: 



type Page
    = DisplayPage Display.Model
    | ...
    | RegistrationPage Registration.Model
    | SavedPlansPage SavedPlans.Model






Some state, such as the session ID or the isMenuOpen flag, is global – it’s shared between the pages. In order to make it easier to manage, we’re going to move it to a sub-record (you’ll see why it’s useful shortly): 



type alias AppState =
    { auth : Auth.Model
    , currPlanText : String
    , isMenuOpen : Bool
    , lastError : String
    , serverUrl : String
    }


type alias Model =
    { appState : AppState
    , currPage : Page
    }






As a result, we end up with just two fields in the main model. Note that I moved serverUrl from a top-level value into an AppState field because it’s easier to pass it to the page-specific update functions this way (no need to make it a separate argument). 


Now, we can write the update function for the registration page. Let’s assume that the /register endpoint of the server returns a session ID to allow the user to be logged in straight away upon registration. This creates an interesting problem because this piece of data doesn’t belong in Registration.Model. We need to find a way to pass it over to model.auth.sessionId in the main model. 


Clearly, the update function has to return this value somehow, and we have several ways to package it: 



  	Return the session ID directly in a tuple with Model and Cmd Msg: ( Model, String, Cmd Msg )


  	Return the new appState: ( AppState, Model, Cmd Msg )


  	Define a custom type to enumerate a set of messages that could be used to signal some kind of action to the main update function, and return that: ( Model, Cmd Msg, PageMsg ).




The first approach is overly specific. Different pages will return different types of data potentially mapping to multiple fields in the AppState record, so the return value from each page will need to be handled in a different way. Let’s implement the second and third approaches, starting with returning a new appState value:



-- in Pages/Registration.elm

update : Msg -> AppState -> Model -> ( AppState, Model, Cmd Msg )
update msg ({ auth } as appState) model =
    case msg of
        ChangePassword p ->
            ( appState, { model | password = p }, Cmd.none )

        ChangeRepeatPassword p ->
            ( appState, { model | repeatPassword = p }, Cmd.none )

        ChangeUserName u ->
            ( appState, { model | userName = u }, Cmd.none )

        FinishRegistration (Ok sessionId) ->
            ( { appState | auth = { auth | sessionId = Just sessionId } }
            , model
            , saveSessionId <| Just sessionId
            )

        FinishRegistration (Err error) ->
            ( appState, { model | errors = [ toString error ] }, Cmd.none )

        SelectPlatform platform ->
            ( appState
            , { model | platform = Just platform }
            , Cmd.none
            )

        StartRegistration ->
            ( appState
            , model
            , register appState.serverUrl model.userName model.password
            )

        ToggleAcceptTerms val ->
            ( appState, { model | hasAcceptedTerms = val }, Cmd.none )






Since we are using AppState here, we need to move its definition into a separate module (eg Types.elm) that’s imported by both Main and Pages.Registration modules in order to avoid a circular dependency that would result from importing Main. 


Remember that this code is inside the Registration module so we are working with Registration.Msg and Registration.Model in this function. saveSessionId used in handling the FinishRegistration message is a port defined earlier. Finally, the register function used to generate a command in response to the StartRegistration message is very similar to the login function from the Auth module:



register : String -> String -> String -> Cmd Msg
register serverUrl userName password =
    let
        body =
            Http.jsonBody <|
                Json.Encode.object
                    [ ( "userName", Json.Encode.string userName )
                    , ( "password", Json.Encode.string password )
                    ]

        responseDecoder =
            Json.Decode.field "sessionId" Json.Decode.string
    in
    Http.post
        { url = serverUrl ++ "register"
        , body = body
        , expect = Http.expectJson FinishRegistration responseDecoder
        }






There is one slight improvement we can make to the update function. So far, its signature may lead us to believe that it can change any aspect of appState: 



update : Msg -> AppState -> Model -> ( AppState, Model, Cmd Msg )






We can’t tell from this signature whether this function will also toggle the side menu or change the server URL. But in fact, the only thing it changes is the sessionId. We can make the subset of fields which are used by this function explicit: 



type alias AppStateSubset a = 
    { a | serverUrl : String, sessionId : Maybe String }


update : Msg -> AppStateSubset a -> Model 
    -> ( AppStateSubset a, Model, Cmd Msg )
update msg appState model =
    ...






We don’t have the ability to specify different input and output subsets of AppState in the type signature, so we still can’t tell whether update is going to change the server URL unless we check all of the code in the function. But at least we can tell that it’s not going to open the side menu because isMenuOpen is not mentioned in AppStateSubset. This is probably more self-documenting, but it also means that you have to change the type alias every time you want to use another AppState field in this function. 


Subsequently, we can add the view code to complete the Registration module: 



page : Model -> Element Msg
page model =
    column
        [ paddingXY 0 20, spacingXY 0 10, width (px 300), centerX ]
    <|
        [ Input.email Attr.input
            { onChange = ChangeUserName
            , text = model.userName
            , placeholder = Nothing
            , label = Input.labelAbove [] <| text "Email:"
            }
        , Input.newPassword Attr.input
            { onChange = ChangePassword
            , text = model.password
            , placeholder = Nothing
            , show = False
            , label = Input.labelAbove [] <| text "Password:"
            }
        , Input.newPassword Attr.input
            { onChange = ChangeRepeatPassword
            , text = model.repeatPassword
            , placeholder = Nothing
            , show = False
            , label = Input.labelAbove [] <| text "Repeat password:"
            }
        , Input.radio
            [ padding 3, spacing 5 ]
            { onChange = SelectPlatform
            , selected = model.platform
            , label = Input.labelAbove [] <| text "Platform:"
            , options =
                [ Input.option Aws <| text "AWS"
                , Input.option Azure <| text "Azure"
                , Input.option Heroku <| text "Heroku"
                , Input.option SelfHosted <| text "Self-hosted"
                ]
            }
        , Input.checkbox
            [ padding 3 ]
            { onChange = ToggleAcceptTerms
            , checked = model.hasAcceptedTerms
            , label = Input.labelRight [] <| text "I accept the terms"
            , icon = Input.defaultCheckbox
            }
        , Input.button Attr.greenButton
            { onPress = Just StartRegistration
            , label = el [ centerX ] <| text "Register"
            }
        ]
            ++ List.map (text >> el Attr.error) model.errors






This is quite similar to the login page, but with a few extra input controls. 


Wiring the module into the main program


There are three sets of changes required to wire our newly created page module into the main module:



  	Updating the Msg type

  	Altering the view function to include the new page

  	Handling the new messages in the update function.




Just like we did with Auth, we need to add a new message constructor to wrap the Registration.Msg messages: 



type Msg
    ...
    | Register Registration.Msg
    ...






The view changes are fairly straightforward too:



view : Model -> Browser.Document Msg
view model =
    let
        content =
            case model.currPage of
                DisplayPage pageModel ->
                    Display.page model.appState pageModel
                        |> Element.map Display

                InputPage ->
                    inputPage model.appState

                LoginPage ->
                    loginPage model.appState.auth

                RegistrationPage pageModel ->
                    Registration.page pageModel
                        |> Element.map Register

                SavedPlansPage pageModel ->
                    SavedPlans.page pageModel
                        |> Element.map SavedPlans
    in
    { title = "VisExp"
    , body =
        [ layout [ inFront <| menuPanel model.appState ] <|
            column [ width fill, spacingXY 0 20 ]
                [ navBar
                , content
                ]
        ]
    }






This version of view assumes that display and saved plan pages have also been extracted into Display and SavedPlans modules, correspondingly. I will show these modules for completeness at the end of the chapter. Input and login pages are still created as before because they haven’t been moved into modules, but the rest of the pages are a bit more involved. For the registration page, we have: 



    RegistrationPage pageModel ->
        Registration.page pageModel
            |> Element.map Register






Remember that values of the Page type now carry the page state – that’s what pageModel refers to. Further, we pass this state to the page function you saw above. The reason that we then need to add |> Element.map Register is that the page function returns Element Msg where Msg is the message type defined in Registration.elm. However, the view function returns a different type, Browser.Document Msg. So, just like we had to translate between Cmd Auth.Msg and Cmd Msg with the help of Cmd.map when handling Auth messages, here we have to translate between Element Registration.Msg and Element Msg in order to create the correct value for the body field in the returned record. We achieve the translation with the help of Element.map. 


The last task is to amend the update function. There are a couple of changes; one is to create the registration page, and another to handle Register Registration.Msg messages:



update : Msg -> Model -> ( Model, Cmd Msg )
update msg ({ appState } as model) =
    case msg of
        ...
        RequestRegistration ->
            ( { model | currPage = RegistrationPage Registration.init }, Cmd.none )

        Register regMsg ->
            let
                ( newAppState, regModel, regCmd ) =
                    Registration.update regMsg appState ???

                newCurrPage =
                    case regMsg of
                        Registration.FinishRegistration (Ok _) ->
                            InputPage

                        _ ->
                            RegistrationPage regModel
            in
            ( { appState = newAppState, currPage = newCurrPage }
            , Cmd.map Register regCmd
            )        






The first one is straightforward, we just need to pass the initial state to RegistrationPage. With the second one, we run into a problem marked by the ??? where we need to pass the current state of the page to Registration.update. How can we get the current Registration.Model value from currPage? As currPage is a value of a custom type, it has to involve a case expression: if currPage happens to be holding a RegistrationPage Registration.Model value, then we use it, otherwise we’re in some kind of invalid state. Fortunately, we don’t need to insert this kind of nested case for every page. Instead, we can extend the case expression we already have in the update function:



update : Msg -> Model -> ( Model, Cmd Msg )
update msg ({ appState } as model) =
    case ( msg, model.currPage ) of
        ...
        ( Register regMsg, RegistrationPage pageModel ) ->
            let
                ( newAppState, regModel, regCmd ) =
                    Registration.update regMsg model.appState pageModel

                newCurrPage =
                    case regMsg of
                        Registration.FinishRegistration (Ok _) ->
                            InputPage

                        _ ->
                            RegistrationPage regModel
            in
            ( { appState = newAppState, currPage = newCurrPage }
            , Cmd.map Register regCmd
            )
        ...






Similarly, all other messages have to be paired up either with a page where they are supposed to occur, or with a wildcard match if they can occur on more than a single page. For example, we’ll also have: 



    -- Login page can be requested from any page via the menu
    ( RequestLogin, _ ) ->
        ( { model | currPage = LoginPage }, Cmd.none )
    ...
    -- A plan can only be submitted from the input page
    ( SubmitPlan, InputPage ) ->
        ( { model | currPage = DisplayPage Display.init }, Cmd.none )






Any other combinations of messages and pages will be handled by the catch-all clause which we can’t avoid in this situation: 



    ( _, _ ) ->
        ( model, Cmd.none )






One more wrinkle in handling registration messages is that we need to send the user to the input page once the registration has been successfully completed. However, we don’t get that information explicitly, so we need to either examine the value of appState.auth.sessionId or to consider whether the registration message indicates successful completion, which is what I’ve chosen to do: 



    newCurrPage =
        case regMsg of
            Registration.FinishRegistration (Ok _) ->
                InputPage

            _ ->
                RegistrationPage regModel






Either way, we end up examining implementation details rather than acting on an explicit signal. This demonstrates why it may be a more robust solution to return a special message instead of the updated appState value. Let’s reimplement Registration.update in this way.


Returning a message instead of the updated global state


We need to introduce a new type (in the Registration module) to signal to the main update what the desired action is, along with passing the necessary data: 



type PageMsg
    = DoNothing
    | FinishSuccessfully String






For the registration page, there is only one potential action: reacting to a successful registration. In all other cases, we’ll return DoNothing. Registration.update now changes slightly to return the page state, a command, and a message:



update : Msg -> { a | serverUrl : String } -> Model -> ( Model, Cmd Msg, PageMsg )
update msg { serverUrl } model =
    case msg of
        ChangePassword p ->
            ( { model | password = p }, Cmd.none, DoNothing )

        ChangeRepeatPassword p ->
            ( { model | repeatPassword = p }, Cmd.none, DoNothing )

        ChangeUserName u ->
            ( { model | userName = u }, Cmd.none, DoNothing )

        FinishRegistration (Ok sessionIdStr) ->
            ( model
            , saveSessionId <| Just sessionId
            , FinishSuccessfully sessionIdStr
            )

        FinishRegistration (Err error) ->
            ( { model | errors = [ httpErrorString error ] }, Cmd.none, DoNothing )

        SelectPlatform platform ->
            ( { model | platform = Just platform }
            , Cmd.none
            , DoNothing
            )

        StartRegistration ->
            ( model
            , register serverUrl model.userName model.password
            , DoNothing
            )

        ToggleAcceptTerms val ->
            ( { model | hasAcceptedTerms = val }, Cmd.none, DoNothing )






Note that I’ve used { a | serverUrl : String } in the type signature because we’re not using any part of AppState other than the server URL. Consequently, the Registration module no longer needs to know about AppState.


It’s now up to the main update function to modify the appState and currPage fields based on the message returned by the page update function: 



update : Msg -> Model -> ( Model, Cmd Msg )
update msg ({ appState } as model) =
    case ( msg, model.currPage ) of
        ...
        ( Register regMsg, RegistrationPage pageModel ) ->
            let
                ( regModel, regCmd, pageMsg ) =
                    Registration.update regMsg model.appState pageModel

                newModel =
                    case pageMsg of
                        Registration.FinishSuccessfully id ->
                            let
                                auth =
                                    appState.auth
                            in
                                { appState =
                                    { appState
                                        | auth =
                                            { auth | sessionId = Just id }
                                    }
                                , currPage = InputPage
                                }

                        Registration.DoNothing ->
                            { model | currPage = RegistrationPage regModel }
            in
            ( newModel, Cmd.map Register regCmd )
        ...






This clause is now very explicit as to how it changes both appState and currPage. Does it look a bit unwieldy? That’s because in Elm, nested record updates are intentionally inconvenient in order to encourage flatter record structures. In fact, we could have kept all of the auth fields directly in appState to simplify things here. However, since I needed to demonstrate a number of different things, it was useful to have more nesting. 


The update mechanism we’ve just implemented, consisting of a model with two fields (some variation of appState and currPage) combined with returning messages used to update appState, was popularised by Ossi Hanhinen and Matias Klemola under the name of “Taco model”. If you see people referring to tacos in relation to Elm code, that’s what they’re talking about. 


Adding Input Validation


This is a bit of a digression from the main topic of this chapter, however the registration page we’ve just created provides a convenient spot to talk about validation, as there are quite a few inputs on this page. 


A popular package for validation is rtfeldman/elm-validate. It’s based on the now-familiar approach of first constructing a validator which is a value rather than a function, and then applying it to some data. This is analogous to how JSON decoders work.


So far, we haven’t done anything to check the user input before kicking off the registration request: 



    StartRegistration ->
        ( model
        , register serverUrl model.userName model.password
        , DoNothing
        )






Let’s assume that we need to check several things: 



  	The user name has to be an email

  	The password can’t be blank

  	Both of the password fields have to match

  	The user has to select a platform that they use from the dropdown

  	The user has to tick the checkbox to confirm that they’ve accepted the terms.




Using elm-validate, we can translate these requirements into a validator in this fashion: 



import Validate exposing (..)

...

validator : Validator String Model
validator =
    Validate.all
        [ Validate.ifInvalidEmail .userName
            (\_ -> "Please enter a valid email address.")
        , Validate.ifBlank .password "Password can't be blank."
        , Validate.fromErrors
            (\model ->
                if model.password == model.repeatPassword then
                    []

                else
                    [ "Passwords must match." ]
            )
        , Validate.fromErrors
            (\model ->
                if model.platform /= Nothing then
                    []

                else
                    [ "Please select a platform." ]
            )
        , Validate.ifFalse .hasAcceptedTerms "Please accept the terms."
        ]






There are built-in validators for things like emails or blank inputs. For other things, like checking that the passwords match, we can define a custom validator by passing a function which returns a list of errors (or an empty list if the validation passed) to Validate.fromErrors. 


Note that the validator only works with the model, which makes elm-validate completely independent of the approach you’ve used to build the UI. 


To perform the validation, all we need to do is call the validate function and check whether it returns any errors: 



    StartRegistration ->
        case validate validator model of
            Ok validModel ->
                ( fromValid validModel
                , register serverUrl model.userName model.password
                , DoNothing
                )

            Err errors ->
                ( { model | errors = errors }, Cmd.none, DoNothing )






Extracting Display and Saved Plan Pages


As I mentioned earlier, some of the code in this and following chapters assumes that the display and saved plan pages are also extracted into their own modules. 


Here is what Pages/Display.elm looks like:



module Pages.Display exposing (Model, Msg(..), init, page, update)

import Attr
import Element exposing (..)
import Json.Decode
import PlanParsers.Json exposing (..)
import PlanTree


type Msg
    = MouseEnteredPlanNode Plan
    | MouseLeftPlanNode Plan


type alias Model =
    { selectedNode : Maybe Plan
    }


init : Model
init =
    { selectedNode = Nothing
    }


update : Msg -> Model -> Model
update msg model =
    case msg of
        MouseEnteredPlanNode plan ->
            { model | selectedNode = Just plan }

        MouseLeftPlanNode _ ->
            { model | selectedNode = Nothing }


page : { a | currPlanText : String } -> Model -> Element Msg
page appState model =
    let
        planTreeConfig =
            { onMouseEnteredNode = MouseEnteredPlanNode
            , onMouseLeftNode = MouseLeftPlanNode
            }
    in
    case Json.Decode.decodeString decodePlanJson appState.currPlanText of
        Ok planJson ->
            PlanTree.render planTreeConfig planJson model.selectedNode

        Err err ->
            el Attr.error <| text <| Json.Decode.errorToString err






And here is Pages/SavedPlans.elm: 



module Pages.SavedPlans exposing 
    (Model
    , Msg(..)
    , PageMsg(..)
    , getSavedPlans
    , init
    , page
    , update
    )

import Color exposing (..)
import Element exposing (..)
import Element.Background as Background
import Element.Border as Border
import Element.Events exposing (..)
import Element.Font as Font
import Http
import PlanParsers.Json exposing (..)
import Utils exposing (httpErrorString)


type Msg
    = FinishSavedPlans (Result Http.Error (List SavedPlan))
    | ShowPlan String


type PageMsg
    = DisplayPlan String
    | DoNothing


type alias Model =
    { lastError : String
    , savedPlans : List SavedPlan
    }


init : String -> Maybe String -> ( Model, Cmd Msg )
init serverUrl sessionId =
    ( { lastError = "", savedPlans = [] }, getSavedPlans serverUrl sessionId )


getSavedPlans : String -> Maybe String -> Cmd Msg
getSavedPlans serverUrl sessionId =
    Http.request
        { method = "GET"
        , headers =
            [ Http.header "SessionId" <| Maybe.withDefault "" sessionId ]
        , url = serverUrl ++ "plans"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectJson FinishSavedPlans decodeSavedPlans
        }


update : Msg -> Model -> ( Model, PageMsg )
update msg model =
    case msg of
        FinishSavedPlans (Ok savedPlans) ->
            ( { model | savedPlans = savedPlans }, DoNothing )

        FinishSavedPlans (Err error) ->
            ( { model | lastError = httpErrorString error }, DoNothing )

        ShowPlan planText ->
            ( model, DisplayPlan planText )


page : Model -> Element Msg
page model =
    let
        annotateVersion name planVersion =
            { version = planVersion.version
            , planText = planVersion.planText
            , createdAt = planVersion.createdAt
            , name = name
            }

        annotateVersions savedPlan =
            List.map (annotateVersion savedPlan.name) savedPlan.versions

        tableAttrs =
            [ width (px 800)
            , paddingEach { top = 10, bottom = 50, left = 10, right = 10 }
            , spacingXY 10 10
            , centerX
            ]

        headerAttrs =
            [ Font.bold
            , Background.color Color.lightGrey
            , Border.color Color.darkCharcoal
            , Border.widthEach { bottom = 1, top = 0, left = 0, right = 0 }
            , centerX
            ]
    in
    table tableAttrs
        { data = List.concatMap annotateVersions model.savedPlans
        , columns =
            [ { header = el headerAttrs <| text "Plan name"
              , width = fill
              , view =
                    \plan ->
                        el
                            [ Font.underline
                            , mouseOver [ Font.color lightCharcoal ]
                            , onClick <| ShowPlan plan.planText
                            ]
                        <|
                            text plan.name
              }
            , { header = el headerAttrs <| text "Creation time"
              , width = fill
              , view = .createdAt >> text
              }
            , { header = el headerAttrs <| text "Version"
              , width = fill
              , view = .version >> String.fromInt >> text
              }
            ]
        }






Finally, the main update function needs to deal with page-specific messages for these pages, similarly to the code we added for the registration page: 



update : Msg -> Model -> ( Model, Cmd Msg )
update msg ({ appState } as model) =
    case ( msg, model.currPage ) of
        ...
        ( Display pageMsg, DisplayPage pageModel ) ->
            let
                newPageModel =
                    Display.update pageMsg pageModel
            in
            ( { model | currPage = DisplayPage newPageModel }, Cmd.none )
        ...
        ( RequestSavedPlans, _ ) ->
            let
                ( pageModel, pageCmd ) =
                    SavedPlans.init appState.serverUrl appState.auth.sessionId
            in
            ( { model | currPage = SavedPlansPage pageModel }
            , Cmd.map SavedPlans pageCmd
            )

        ( SavedPlans pageMsg, SavedPlansPage pageModel ) ->
            let
                ( newPageModel, outMsg ) =
                    SavedPlans.update pageMsg pageModel

                newModel =
                    case outMsg of
                        SavedPlans.DisplayPlan planText ->
                            { model
                                | appState = { appState | currPlanText = planText }
                                , currPage = DisplayPage Display.init
                            }

                        _ ->
                            { model | currPage = SavedPlansPage newPageModel }
            in
            ( newModel, Cmd.none )

        ( SubmitPlan, InputPage ) ->
            ( { model | currPage = DisplayPage Display.init }, Cmd.none )        






User actions on the display page don’t cause page changes or side effects, so Display.update only needs to return a new model. SavedPlans.update returns a new model along with a message which can cause the main update to change the current page. 


Summary


A key point I’d like to reiterate is that throughout the refactoring process, we only dealt with types and functions. Even though there were quite a few steps involved, there was nothing more to it in terms of abstractions. We moved some of the definitions into separate modules and then we wired them into the main program. It didn’t require components, or design patterns, or any kind of framework. This approach is both conceptually simple and quite scalable. 


Another observation, which may not be apparent from reading this chapter, is that regardless of the type of the refactoring described here, actually performing it wasn’t a huge job, and didn’t require extensive changes to the code. A lot of stuff could simply be copy-pasted, and fixing up the resulting compiler errors was a mostly mechanical process. Once the code compiled again, it mostly worked, aside from issues like forgetting to add a link to the registration page to the menu panel. 


Finally, keep in mind that given how small this code base is, we’ve gone too far with modularising it. This was useful for learning purposes, but in real life, you should remember that it introduces additional complexity (such as the nested record updates in the update function), and should only be done when it actually improves maintainability on balance.








Squashing Bugs in Any Way Possible


Overview


This chapter is all about bugs: first, how you can prevent some of them, and second, how you can find the ones that steal sneak in. You will see several techniques for eliminating bugs by leveraging the language features and the type system. We will talk about testing because testing is a major tool for bug prevention, even in a language like Elm where some bugs can be avoided. Finally, we will take a look at Elm’s time travelling debugger and a couple more things like logging. 


Bug Prevention via Code Design


Elm does a great job of eliminating some classes of bugs by default. Since it’s a compiled languages, it prevents obvious mistakes like mistyped variable names and syntax errors. Its type system removes several other bug categories, for example because it doesn’t have the notion of null, and because a case statement must handle all the tags of a type. In addition, Elm’s type system provides us several other avenues for writing bug-free code by structuring types in particular ways. 


Type aliases


In Elm, we spend a lot of time looking at function signatures to work out how to pass arguments and generally get things to work together. Consider the signature of the login function from the previous chapter:



login : String -> String -> String -> Cmd Msg






A signature like this isn’t very helpful because it doesn’t explain what any of the arguments are, and it doesn’t prevent bugs caused by transposed arguments. For example, it would be very easy to mix up user name and password, and then we’d have to work out what’s wrong at runtime. We can use type aliases to add some clarity:



type alias ServerUrl = String
type alias UserName = String
type alias Password = String
 
login : ServerUrl -> UserName -> Password -> Cmd Msg






We’ve also been using String and Maybe String to represent the session ID in our application. We can use a SessionId type alias instead of plain String to clarify the code.


To be clear, this is really just compiler-parsed documentation, because we can still transpose arguments if we don’t look up the function signature, and the compiler will say nothing because both UserName and Password are simply different names for the same type: 



login "https://example.com" "password" "user"   -- Still compiles 






We can in fact go further by using single-constructor custom types instead of type aliases: 



type ServerUrl = ServerUrl String
type UserName = UserName String
type Password = Password String 

login : ServerUrl -> UserName -> Password -> Cmd Msg

-- ...

login 
    (ServerUrl "https://example.com") 
    (Password "password") 
    (UserName "user")  -- Produces a compiler error






Of course, now we’ll need to unwrap these arguments in the login function, and all this wrapping and unwrapping takes more effort, so in practice you will likely have a mix of custom types and aliases. We will talk more about single-constructor types later in the chapter. 


Custom types with multiple constructors


Custom types can often be used to achieve better representation of the data handled by the application. Specific examples of that are using custom types as a more robust substitute for boolean flags, as well as for managing dependencies between pieces of state. 


Imagine that we introduced a premium tier of functionality into our application which produces magical AI-driven recommendations for performance improvement based on a submitted query plan.


Now, when a user logs in, the server will return a premium status boolean flag so we can render sections of the UI appropriately, which means that we have two items of information: whether the user is logged in, and whether the user is a premium user. We could model it like this:



type alias Model = 
    { isLoggedIn : Boolean
    , isPremiumUser : Boolean
    ...
    }






However, this means that the model allows one possible state which should never occur in practice: isLoggedIn = False combined with isPremiumUser = True. As a result, we have to be diligent and set isPremiumUser to False whenever the user is logged out. Of course, diligence can only go so far, and eventually it’s likely that a bug will emerge somewhere. Instead, we can setup the model to eliminate the invalid state altogether: 



type AccessLevel 
    = Anon
    | Restricted
    | Full


type alias Model =
    { accessLevel : AccessLevel 
    ...
    }






Now, when we receive the response from the server, we transition from Anon to either Restricted or Full (depending on the value of the premium status flag), and on logout we transition to Anon. It’s no longer possible to get into an invalid state at all. 


Similarly, custom types can be used to avoid invalid states in situations where some information can only be supplied by the user depending on previous input. For example, we could show a list of configuration parameter fields based on the selected database version, or a list of cities based on the selected country followed by an additional street selector which is in turn based on the city. Once again, having separate fields in the model allows invalid state such as a selected city without a selected country, but modelling partially completed selections with custom types eliminates this problem. The custom types for these situations would look like this: 



type ConfigField
    = NotSelected
    | DbVersionSelected DbVersion
    | Selected DbVersion FieldName
    
    
type City
    = NotSelected
    | CountrySelected String
    | Selected String String






No requests without a valid session ID


The technique in the previous section can help us avoid invalid state, but what about preventing invalid actions? For example, can we get the compiler to stop us from issuing invalid requests to the server? 


Consider the SavedPlans.init and SavedPlans.getSavedPlans functions from the previous chapter: 



init : String -> Maybe SessionId -> ( Model, Cmd Msg )
init serverUrl sessionId =
    ( { lastError = "", savedPlans = [] }, getSavedPlans serverUrl sessionId )


getSavedPlans : String -> Maybe SessionId -> Cmd Msg
getSavedPlans serverUrl sessionId =
    Http.request
        { method = "GET"
        , headers =
            [ Http.header "SessionId" <| Maybe.withDefault "" sessionId ]
        , url = serverUrl ++ "plans"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectJson FinishSavedPlans decodeSavedPlans
        }






They are oblivious to whether there is a valid session ID. An obvious improvement is to only make the request when the session ID is valid: 



init : String -> Maybe SessionId -> ( Model, Cmd Msg )
init serverUrl sessionId =
    case sessionId of
        Just validId ->
            ( { lastError = "", savedPlans = [] }
            , getSavedPlans serverUrl validId
            )

        Nothing ->
            ( { lastError = "Invalid session ID", savedPlans = [] }, Cmd.none )


getSavedPlans : String -> SessionId -> Cmd Msg
getSavedPlans serverUrl sessionId =
    Http.request
        { method = "GET"
        , headers = [ Http.header "SessionId" sessionId ]
        , url = serverUrl ++ "plans"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectJson FinishSavedPlans decodeSavedPlans
        }






By properly handling both possibilities in the Maybe value, we stop requests without a session ID. But can we do better? Can we ensure that SavedPlans.init never even gets called without a valid session ID? 


Let’s start by changing the signature and implementation of this function to reflect this goal. Since init will  require a valid session ID, it should actually take a session ID as its argument instead of a Maybe value:



init : String -> SessionId -> ( Model, Cmd Msg )
init serverUrl sessionId =
    ( { lastError = "", savedPlans = [] }, getSavedPlans serverUrl sessionId )


getSavedPlans : String -> SessionId -> Cmd Msg
getSavedPlans serverUrl sessionId =
    Http.request
        { method = "GET"
        , headers = [ Http.header "SessionId" sessionId ]
        , url = serverUrl ++ "plans"
        , body = Http.emptyBody
        , timeout = Nothing
        , tracker = Nothing
        , expect = Http.expectJson FinishSavedPlans decodeSavedPlans
        }






Now the compiler will tell us to fix the call site in update where we’re still trying to pass a Maybe SessionId value as the argument: 



update : Msg -> Model -> ( Model, Cmd Msg )
update msg ({ appState } as model) =
    case ( msg, model.currPage ) of
        ...
        ( RequestSavedPlans, _ ) ->
            let
                ( pageModel, pageCmd ) =
                    -- Compilation error reported here:
                    SavedPlans.init appState.serverUrl appState.auth.sessionId
            in
            ( { model | currPage = SavedPlansPage pageModel }
            , Cmd.map SavedPlans pageCmd
            )
        ...






Again, we could try to handle both possibilities within this clause:



    ( RequestSavedPlans, _ ) ->
        case appState.auth.sessionId of
            Just id ->
                let
                    ( pageModel, pageCmd ) =
                        SavedPlans.init appState.serverUrl id
                in
                ( { model | currPage = SavedPlansPage pageModel }
                , Cmd.map SavedPlans pageCmd
                )

            Nothing ->
                ( model, Cmd.none )






This will work, however this code implies that the RequestSavedPlans message could be issued when the user isn’t logged in, and that would be a bug! In order to prevent this message from being generated, let’s require a session ID to be passed to the message constructor: 



type Msg
    ...
    | RequestSavedPlans SessionId
    ...






The compiler will direct me to fix the uses of this message constructor, for example when generating the side menu: 



menuPanel : AppState -> Element Msg
menuPanel appState =
    let
        items =
            [ el [ pointer, onClick CreatePlan ] <| text "New plan" ]
            ++ (case appState.auth.sessionId of
                    Just sessionId ->
                        [ el [ pointer, onClick (RequestSavedPlans sessionId) ] <|
                            text "Saved plans"
                        , el [ pointer, onClick RequestLogout ] <| text "Logout"
                        ]

                    Nothing ->
                        [ el [ pointer, onClick RequestLogin ] <| text "Login"
                        , el [ pointer, onClick RequestRegistration ] <|
                            text "Register"
                        ]
               )
    ...






All I need to do is pass the valid session ID to RequestSavedPlans because we already have a case statement to deal with the login status.


Finally, in the update function I need to use the session ID value carried by the message: 



update msg ({ appState } as model) =
    case ( msg, model.currPage ) of
        ...
        ( RequestSavedPlans sessionId, _ ) ->
            let
                ( pageModel, pageCmd ) =
                    SavedPlans.init appState.serverUrl sessionId
            in
            ( { model | currPage = SavedPlansPage pageModel }
            , Cmd.map SavedPlans pageCmd
            )
        ...






After these changes, the compiler makes it harder to issue an invalid request for saved plans. It’s still technically possible, but you would have to go to the effort of explicitly passing an invalid session ID such as an empty string to the message constructor. 


An additional effect is that the conditional logic dealing with Maybe SessionId is pushed towards the boundaries of the system (either the UI or server requests), while most of the other functions are simplified as they only need to deal with valid SessionId values. In general, this is what you should aim for with regards to Maybe. Try to contain it near the edges of the system, so to speak, so that it doesn’t spread through your code base and complicate the logic. 


Opaque Types


In the previous section, we were essentially trying to control the creation of values in one way or another. Elm provides another mechanism for controlling the creation of values and enforcing invariants. This is called opaque types and they are constructed with the help of module boundaries. If you can prevent invalid values from being created, then it stops certain kinds of bugs from appearing in your code. It potentially makes your code simpler too, as the validity checks are centralised in the functions you provide for working with values of an opaque type. 


Opaque types are especially useful when you create packages. The reason is that they give you some flexibility to change the implementation details of the package without changing the interface. This is helpful in a situation involving some sort of public API. However, it’s often a good idea to use them in non-package code as well, especially when the application gets larger and multiple developers are involved. It’s easier to enforce rules via the type system than to rely on documentation. 


Let’s consider the Plan type from the PlanParsers.Json module:



type Plan
    = PCte CteNode
    | PGeneric CommonFields
    | PResult ResultNode
    | PSeqScan SeqScanNode
    | PSort SortNode






We can export this type by exposing everything in the module, or by specifically exposing the type along with its constructors:



module PlanParsers.Json exposing (..)







module PlanParsers.Json exposing (Plan(..))






If we do either of these, then the users of the module are able to construct values of type Plan using its constructors PCte, PGeneric and so on. 


However, it results in a couple of issues when the type definition is changed: 



  	The code that’s using it breaks if constructors are removed or renamed

  	The code that’s using it breaks even if a new constructor is added, because case statements matching on the type tags are no longer exhaustive. 




But if we export the Plan type without explicitly exporting its constructors, it becomes an opaque type because its internals are no longer visible outside the module:



module PlanParsers.Json exposing (Plan)

type Plan
    = PCte CteNode
    | PGeneric CommonFields
    | PResult ResultNode
    | PSeqScan SeqScanNode
    | PSort SortNode






When we import PlanParsers.Json into another module, it allows Plan to appear in function signatures and to be used as a type of record fields, but we can’t construct values of this type directly because we can’t see its constructors. It’s only possible to construct Plan values if the module also exposes one or more functions that returns a Plan value. 


Additionally, by exporting it as an opaque type, we take away the ability to pattern match on the constructor in the code that subsequently imports Plan.


Single-constructor custom types


This works for single-constructor custom types as well. For example, we could turn SessionId into an opaque type and enforce the condition that a session ID can’t be an empty string: 



module Types exposing (SessionId, asString, makeSessionId)

type SessionId
    = SessionId String


makeSessionId : String -> Maybe SessionId
makeSessionId s =
    if String.isEmpty s then
        Nothing
    else
        Just (SessionId s)


asString : SessionId -> String
asString (SessionId id) =
    id






Now, the session ID string returned from the server can only be turned into a SessionId value with makeSessionId sessionIdStr, which will check for empty string. For example, when we get the session ID from the server when registering a user, we’ll try to construct a SessionId value, and only return a FinishSuccessfully message to the main update if the session ID is valid:



update : Msg -> { a | serverUrl : String } -> Model -> ( Model, Cmd Msg, PageMsg )
update msg { serverUrl } model =
    case msg of
        ...
        FinishRegistration (Ok sessionIdStr) ->
            case makeSessionId sessionIdStr of
                Just sessionId ->
                    ( model
                    , saveSessionId <| Just (asString sessionId)
                    , FinishSuccessfully sessionId
                    )

                Nothing ->
                    ( { model
                        | errors = [ "Invalid session ID returned from the server" ]
                      }
                    , Cmd.none
                    , DoNothing
                    )
        ...






When we send the session ID back to the server with a request or send it through the saveSessionId port, we need to extract the session ID string out of this value. Since the constructor is no longer visible outside of the Types module, we can’t use pattern matching to unwrap the string value, which is why we also have to provide asString (also used in the example above).


Record types


Record types exported via type alias cause similar issues when they are modified. Suppose we have an exported alias like this:



module PlanParsers.Json exposing (CteNode)

type alias CteNode =
    { common : CommonFields
    , alias_ : String
    , cteName : String
    }






If I just expose CteNode by exporting a type alias, then the user of my package has access to all the fields. If I have some kind of internal invariants for field values that have to be maintained, then the user could potentially violate them because they can set field values to anything they want. The other issue is that I’m not able to change the set of fields without causing a breaking change, even if I’m only doing it for internal implementation purposes.


That’s why it’s a good idea to wrap records in a single-constructor custom type and export that instead: 



module PlanParsers.Json exposing (CteNode)

type CteNode = CteNode
    { common : CommonFields
    , alias_ : String
    , cteName : String
    }






Whether we’re exporting an opaque custom type or a record type, since the user has no ability to construct values directly, we need to provide them with constructor functions: 



module PlanParsers.Json exposing 
    (CommonFields
    , CteNode
    , createCommonFields
    , createCteNode
    )

type CteNode = CteNode
    { common : CommonFields
    , alias_ : String
    , cteName : String
    }


createCteNode : String -> CommonFields -> CteNode
createCteNode name common = 
   CteNode { common = common, alias_ = name, cteName = name }






In the case of record fields, it might also be necessary to provide functions that allow some kind of modification of field values. 


Managing dependent parts of the state


Since we allowed the user to store multiple versions of the query plan as they work on optimising a particular query, it means that the user will be able to select a particular version to view, as well as delete versions and add new ones. 


The server sends a list of plan versions for a given query. We have to store that list as well as the selected version in the model in order to keep track of which version the user has open in the UI. We could model it in this manner: 



type alias PlanVersion =
    { version : Int
    , createdAt : String
    , planText : String
    }


type alias SavedPlan =
    { id : String
    , name : String
    , versions : List PlanVersion
    }


type alias Model = 
    { ...
    , savedPlans : List SavedPlan
    , currentPlan : Maybe SavedPlan
    , currentVersion : Maybe Int
    } 






The trouble is that we now have to be careful about keeping the available versions of the plan and the selected version in sync. For example, when the user deletes a version, the selected version can become invalid. 


Here, too, it’s possible to design the model in a way that makes invalid states impossible. We can achieve that by making the selected version a part of the versions list rather than something separate:



type alias Versions = 
  { priorVersions : List PlanVersion
  , selectedVersion : PlanVersion
  , followingVersions : List PlanVersion
  }






This structure is an example of a so-called zipper – a structure that makes traversing and updating a data structure convenient. It makes it impossible to have no version selected because the selectedVersion field always has to be set. So if a version is deleted, we have no choice but to set it to another valid version.


If there is only a single version of the plan, then we will have a record like this: 



{ priorVersions = []
, selectedVersion = { version = 1, ... }
, followingVersions = [] 
}






If, say, the last version is selected, then we will have a record like this: 



{ priorVersions = [ PlanVersion, PlanVersion, PlanVersion, PlanVersion ]
, selectedVersion = { version = 5, ... }
, followingVersions = [] }






In the SavedPlan record, we will use this Versions structure instead of a plain list: 



type alias SavedPlan =
    { id : String
    , name : String
    , versions : Versions
    }






Instead of hand-rolling our own data structure, we could use the rtfeldman/selectlist package, which implements a list zipper in a generic fashion, providing a number of convenience functions such as map.


Lastly, I’d like to point out that there is a tradeoff to consider here, as with most things. On one hand, using this data structure complicates the implementation. On the other hand, we get the benefit of never having to deal with a selection bug, and the benefit of compiler assistance when we make selection related changes. In any case, it’s definitely worth paying attention to opportunities for making invalid states unrepresentable in the program. 


Logging and Crashing


The Debug module from the Elm core library provides two non-pure functions which can be used to produce a log statement or an exception at runtime, with the caveat that these functions have to be removed from the code in order to make it compile with the --optimize flag.


You can log to the console using Debug.log, which we encountered briefly when dumping the model via a subscription in the previous chapter. You provide Debug.log with a label and a value to output, it converts the value to a string and writes it to the console (which is a side effect handled by the Elm runtime), and returns the value:



update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of 
        ...
        ToggleMenu -> 
            ( Debug.log "model" model, Cmd.none ) 
        ...






As it returns the value it logs, it can be conveniently used inside expressions. 


The other function in the Debug package is called crash. When called, it produces an exception (which is logged to the JavaScript console). Debug.crash is not particularly useful for debugging, but it is useful if you want to run your program and try things out while it’s still incomplete.


For example, you can use it to start testing without handling all the possible branches of a custom type:



planNodeToHtml plan =
    case plan of
        PCte cteNode ->
            genericPlanNodeToHtml plan
                cteNode.common
                [ text " on ", italic cteNode.cteName, text <| " (" ++ cteNode.alias\
_ ++ ")" ]

        _ ->
            Debug.crash "TODO: handle other node types"






Or you can even stub out a function completely: 



updatePage : String -> Msg -> Model -> ( Model, Cmd Msg )
updatePage = Debug.crash "update not implemented"






Note that because the return value of crash unifies with any type (including the type of update), I don’t need to mention the update arguments. 


Testing


Even though Elm reduces the need for testing compared to JavaScript, testing is still necessary and useful, particularly if you like to practise test-driven development. There are a number of libraries available for testing Elm programs. 


The main aspects of your code that you might want to test are: 



  	Functions in general (via unit or property tests)

  	The update function and model updates

  	Views

  	Code that deals with side effects (commands, tasks, subscriptions, ports).




The most popular solution for testing is an NPM package called elm-test and its companion Elm package, elm-explorations/test. It allows you to write both unit tests and fuzz tests. We’re going to take a closer look at using this package shortly.


There are a couple of packages that extend elm-test functionality specifically for the purposes of testing the update function with lists of messages. These are rtfeldman/test-update (which we’ll take for a spin in this chapter) and janiczek/elm-architecture-test.


Another interesting NPM package is called elm-verify-examples, and it sits at the intersection of tests and documentation. It allows you to write examples of using functions in comments (in a particular format), and then it converts these examples into elm-test tests which you can then run to validate the examples. 


The elm-explorations/test package additionally contains modules for testing Html values produced by your view functions by checking for the presence of particular elements, counting elements and so on. It also provides mechanisms for simulating interactions such as text input or mouse movements for the purpose of testing that your views generate the right messages. These modules used to be in a separate package called elm-html-test, and at the time of writing the integration process is ongoing, so the information in the section on view testing is preliminary. Keep in mind that this functionality is only useful if you’re constructing your views using the html package, but not elm-ui. However, elm-ui has a handy explain attribute which can help with debugging layout issues. 


When it comes to side effects, at this point it’s probably easiest to test them at the level of integration tests, which is going to involve non-Elm tools. I am not aware of any easy pure Elm solutions for this at present. The approaches that I’ve seen, such as avh4/elm-program-test, are all invasive, requiring modifications to the code to insert a layer between the code describing side effects and the Elm runtime. 


Finally, if you would like to do end-to-end testing by simulating a user, then you can use tools like Selenium or PhantomJS which are well known in the JavaScript world. 



  elm-test



Let’s start with the basics, and look at writing unit tests with the help of the elm-test NPM package. You can install it by running:



npm install -g elm-test






Then, you need to initialise your test harness in your project directory: 



elm-test init






This will add the necessary packages to elm.json and create tests/Example.elm with a sample test suite:



module Example exposing (..)

import Expect exposing (Expectation)
import Fuzz exposing (Fuzzer, int, list, string)
import Test exposing (..)


suite : Test
suite =
    todo "Implement our first test. See [link] for how to do this!"






You can execute the test suite by running elm-test without arguments.


elm-test will take care of compiling the tests and will report compiler errors if they occur. You can define multiple Test values in one file, and elm-test will execute those ones which are exposed. If you use elm-format, be aware that it always changes exposing (..) to an explicit list like exposing (suite, suite2). If you don’t remember to update this list when you add a Test value, the new suite will not be included in test runs. 


Let’s add a couple of tests for our saved plan JSON decoder (after renaming Example.elm to Tests.elm). First, we can verify that it works on a valid JSON string in this manner:



module Tests exposing (..)

import Expect exposing (Expectation)
import Json.Decode
import PlanParsers.Json exposing (..)
import Test exposing (..)

suite : Test
suite =
    describe "Saved plan JSON decoder"
        [ test "Decoder converts valid strings" <|
            \_ ->
                let
                    json =
                        """
                        [
                            { "id": "1"
                            , "name": "Project query plan"
                            , "versions": [ { "version": 1
                                            , "createdAt": "2017-12-16"
                                            , "planText": "{}" 
                                            }
                                          ]
                            }
                        ]
                        """

                    result =
                        Json.Decode.decodeString decodeSavedPlans json
                in
                Expect.equal result
                    (Ok
                        [ { id = "1"
                          , name = "Project query plan"
                          , versions = [ { version = 1
                                         , createdAt = "2017-12-16"
                                         , planText = "{}" 
                                         }
                                       ]
                          }
                        ]
                    )
        ]






The describe function allows you to group your tests in order to make the test report easier to read. describe calls can additionally be nested to create levels of subgroups. 


When tests fail, the report shows the group name along with the name of the failing test: 



elm-test 0.19.0-rev6
----------------------

Running 1 test. To reproduce these results, run: elm-test --fuzz 100 
  --seed 149453479

↓ Tests
↓ Saved plan JSON decoder
✗ Decoder converts valid strings

    Ok ([{ id = "1", name = "Project query plan", versions = 
    [{ version = 1, createdAt = "2017-12-16", planText = "{}" }] }])
    ╷
    │ Expect.equal
    ╵
    Ok ([{ id = "1", name = "Project query plan", versions = 
    [{ version = 1, createdAt = "2017-12-17", planText = "{}" }] }])

TEST RUN FAILED

Duration: 199 ms
Passed:   0
Failed:   1






We can also add a test to make sure that decoding fails if the input string is invalid: 



suite : Test
suite =
    describe "Saved plan JSON decoder"
        [ test "Decoder converts valid strings" <|
            \_ -> 
                ...
        , test "Decoder fails on missing fields" <|
            \_ ->
                let
                    json =
                        """
                        [{}]
                        """

                    result =
                        Json.Decode.decodeString decodeSavedPlans json
                in
                Expect.err result
        ]






When I run this test suite using elm-test, I get a report that all my tests have passed:



elm-test 0.19.0-rev6
----------------------

Running 2 tests. To reproduce these results, run: elm-test --fuzz 100 --seed 2146106\
055


TEST RUN PASSED

Duration: 612 ms
Passed:   2
Failed:   0






I used two expectations in my tests so far: Expect.equal and Expect.err. 


The full set of expectations is fairly basic. There are a few to deal with integer and floating point comparisons, and a few for checking equality between pairs of lists, sets and dicts. You do have the ability to write your own expectations using the functions pass and fail which return an expectation that always passes and an expectation that always fails, respectively. 


For example, we could write an expectation to make sure that the versions of a saved plan are sorted by version number: 



expectSortedVersions : SavedPlan -> Expectation
expectSortedVersions plan =
    let 
        versions = List.map .version plan.versions
    in
    if List.sort versions == versions then
        Expect.pass 
    else 
        Expect.fail "Versions are not sorted"






This expectation can then be used in a test like any other expectation: 



test "Version order" <|
    \_ ->
        expectSortedVersions 
            { id = "1"
            , name = "Plan"
            , versions = 
                [ { version = 1, createdAt = "2017-12-01", planText = "{}" }
                , { version = 5, createdAt = "2017-12-02", planText = "{}" }
                , { version = 3, createdAt = "2017-12-03", planText = "{}" }
                ] 
            }






Fuzz tests


elm-test supports one more type of tests called fuzz tests. A fuzz test is really a series of tests with randomised inputs. elm-test will generate these inputs and run the test function you supplied with each set of inputs. For example, we could write a test that checks the implementation of List.reverse:



fuzz (list int) "Twice applied List.reverse is a no-op" <|
    \fuzzList ->
         fuzzList
             |> List.reverse
             |> List.reverse
             |> Expect.equal fuzzList






Here is the cool thing: if it finds a failure, then starting from the input that caused it, elm-test will attempt to find the minimal input that still causes a failure: the shortest string, the shortest list, or the smallest number. This is called shrinking.


You may recognise this kind of testing as property based testing rather than fuzzing, but the distinction is a bit fuzzy. In any case, this type of tests is particularly useful when you perform complex calculations.


Let’s consider a more sophisticated fuzz test in the context of our application to make it a bit clearer how to define properties and how to generate input data for fuzz tests. Suppose that when showing our query plan tree, we want to show the portion of execution time spent in the current node, which requires taking the total time and subtracting all the time spent in child nodes, as well as accounting for loops. The calcDuration function below (together with a couple of helpers) shows how this can be expressed in Elm code:



unwrap : Plan -> CommonFields
unwrap plan =
    case plan of
        PCte p ->
            p.common

        PGeneric p ->
            p

        PResult p ->
            p.common

        PSeqScan p ->
            p.common

        PSort p ->
            p.common


calcNodeTime : CommonFields -> Float
calcNodeTime node =
    node.actualTotalTime * toFloat node.actualLoops


calcDuration : CommonFields -> Float
calcDuration node =
    let
        (Plans planList) =
            node.plans
    in
    calcNodeTime node
        - (List.sum <| List.map (unwrap >> calcDuration) planList)






Note that the time for a node is obtained by multiplying the time attribute by the number of loops executed in that node. Additionally, the way Postgres reports this data means that the parent node time includes the time of all child nodes. 


As a result, there are a couple of obvious postconditions (or properties) that should hold true for this function: 



  	The value it calculates should be less than the total time in the input node 

  	The value it calculates should be non-negative.




Let’s construct a test for the first property. The expectation is very simple:



    \node ->
        calcDuration (Debug.log "Node" node)
            |> Expect.atMost (calcNodeTime node)






The interesting part is how node is produced. Let’s start filling in the surrounding code that will exercise this expectation:



suite2 : Test
suite2 =
    describe "Node calculations"
        [ fuzz nodeFuzzer "Node duration should not exceed total time" <|
            \node ->
                calcDuration node
                    |> Expect.atMost (calcNodeTime node)
        ]






Instead of a built-in fuzzer, we’re using nodeFuzzer which has to produce a CommonFields record which may contain nested nodes in its plans field. To achieve that, I need four random inputs rather than one: 



  	Total actual time

  	Loop count

  	A list of times which I will use to populate sub-nodes of my root node

  	A list of loop counts which I will use to populate sub-nodes of my root node.




Therefore, nodeFuzzer can be written like this:



nodeFuzzer : Fuzz.Fuzzer CommonFields
nodeFuzzer =
    let
        timeRange =
            Fuzz.floatRange 0.001 1000.0

        loopRange =
            Fuzz.intRange 1 128
    in
    Fuzz.map4 makeNode
        timeRange
        loopRange
        (Fuzz.list timeRange)
        (Fuzz.list loopRange)






Fuzz.map4 maps a function of four arguments – makeNode in this case – over four fuzzers. I’ve constrained the  random values so that times are between 0.001 and 1000.0, and loop counts are between 1 and 128, mainly because negative values would be invalid and could cause spurious postcondition violations. 


Next, I need to write makeNode. It will receive four arguments and use them to generate a CommonFields record:



makeNode : Float -> Int -> List Float -> List Int -> CommonFields
makeNode actualTotalTime actualLoops subTimes subLoops =
    let
        makeSubNode ( time, loops ) =
            PGeneric
                { actualLoops = loops
                , actualTotalTime = time
                , nodeType = "Generic"
                , plans = Plans []
                , relationName = ""
                , schema = ""
                , startupCost = 0
                , totalCost = 0
                }

        plans =
            subLoops
                |> List.map2 Tuple.pair subTimes
                |> List.map makeSubNode
    in
    { actualLoops = actualLoops
    , actualTotalTime = actualTotalTime
    , nodeType = "Generic"
    , plans = Plans plans
    , relationName = ""
    , schema = ""
    , startupCost = 0
    , totalCost = 0
    }






When I run the test, it will test the expectation using various node values such as this: 



Node: { actualLoops = 66, actualTotalTime = 250.7000929596151, 
nodeType = "Generic", plans = Plans [], relationName = "", 
schema = "", startupCost = 0, totalCost = 0 }

Node: { actualLoops = 2, actualTotalTime = 219.9975220889591, 
nodeType = "Generic", plans = Plans [PGeneric { actualLoops = 53, 
actualTotalTime = 131.30076035569377, nodeType = "Generic", 
plans = Plans [], relationName = "", schema = "", startupCost = 0, 
totalCost = 0 }], relationName = "", schema = "", startupCost = 0,
totalCost = 0 }






If you want to see what kind of nodes are generated for this test (eg to check that all the fields have sensible values), you can add a Debug.log into the test: 



    \node ->
        calcDuration (Debug.log "Node" node)
            |> Expect.atMost (calcNodeTime node)






As you can see, this test is not trivial to construct, but the payoff is that it is less brittle than a unit test. Presumably, the property will not change regardless of how you choose to implement the calculation, and consequently this test should not require changes either. The test is also much broader because it tests a range of inputs and hopefully a number of edge cases. 


When it comes to testing the second property we noted earlier, which is that the result of calcDuration should be non-negative, things get even trickier. Here, we run into the issue that the input data is constrained in a complicated way: the sum of node time and loop count products across all sub-nodes should be less than (or maybe equal to) the same product for the root node. If we don’t constrain the data in this fashion, then of course calcDuration can produce negative values even if its calculations are correct. 


There is no easy way for us to generate data that conforms to these rules with the fuzzers provided in elm-test. We could possibly achieve it with a custom fuzzer (the package has a facility for that), but it is questionable whether the effort justifies the benefit, at least for a function like calcDuration. 


In my experience with fuzz testing, this tradeoff between the complexity of generating input data and the value of checking a particular property surfaces quite often. A factor in the tradeoff is that non-trivial amounts of test code are themselves subject to bugs. For example, you may not be generating inputs in the correct ranges or with correct distributions of values, leading to false positives or false negatives. 


Nevertheless, I recommend relying on fuzz testing where possible, as in many situations it effectively allows you to get a hundred tests for the price of one. 


Testing the update function with test-update



The rtfeldman/test-update package provides a couple of convenience functions for use with elm-test. However, at the time of writing it isn’t compatible with Elm 0.19. I’ve taken the code and made a few changes to make it compatible:



module TestUpdate exposing (fuzz, fuzzBasic, test)

import Expect exposing (Expectation)
import Fuzz exposing (Fuzzer)
import Test exposing (Test)
import Test.Runner

fuzz :
    (msg -> model -> ( model, cmd ))
    -> Fuzzer msg
    -> model
    -> String
    -> (model -> Expectation)
    -> Test
fuzz update =
    fuzzBasic (\msg model -> Tuple.first (update msg model))


fuzzBasic :
    (msg -> model -> model)
    -> Fuzzer msg
    -> model
    -> String
    -> (model -> Expectation)
    -> Test
fuzzBasic update msgFuzzer model description verify =
    Test.fuzz (Fuzz.list msgFuzzer) description <|
        test update model verify


test :
    (msg -> model -> model)
    -> model
    -> (model -> Expectation)
    -> List msg
    -> Expectation
test update model verify messages =
    case Test.Runner.getFailureReason (verify model) of
        Nothing ->
            testUpdateHelp Expect.pass update model verify messages

        Just { given, description, reason } ->
            [ "Initial model failed before any messages were applied."
            , ""
            , "Initial model was:"
            , ""
            , "    " ++ Debug.toString model
            , ""
            , "Failure was:"
            , ""
            , description
            ]
                |> String.join "\n"
                |> Expect.fail


testUpdateHelp :
    Expectation
    -> (msg -> model -> model)
    -> model
    -> (model -> Expectation)
    -> List msg
    -> Expectation
testUpdateHelp expectation update model verify messages =
    case messages of
        [] ->
            expectation

        msg :: rest ->
            let
                newModel =
                    update msg model
            in
            case Test.Runner.getFailureReason (verify newModel) of
                Nothing ->
                    testUpdateHelp expectation update newModel verify rest

                Just { given, description, reason } ->
                    [ "Model which passed:"
                    , ""
                    , "    " ++ Debug.toString model
                    , ""
                    , "Message applied to that model:"
                    , ""
                    , "    " ++ Debug.toString msg
                    , ""
                    , "Resulting model, which failed:"
                    , ""
                    , "    " ++ Debug.toString newModel
                    , ""
                    , "Failure:"
                    , ""
                    , description
                    ]
                        |> String.join "\n"
                        |> Expect.fail






One notable change is that I had to switch to Debug.toString for the purposes of showing messages and model, meaning that this module cannot be used together with the --optimize compiler flag. 


The module provides three functions for testing update: fuzzBasic, fuzz and test. 


fuzzBasic and fuzz both generate a list of messages (according to a fuzzer you define), and pass the model repeatedly through the update function along with each message. At the end, they check the expectation that you also provide. fuzzBasic expects the update function with a signature msg -> model -> model, whereas fuzz expects msg -> model -> ( model, cmd ) (although it doesn’t do anything with commands). 


The test function is similar, except it goes through a list of messages supplied by you instead of a randomly generated list. 


We can use the fuzz function to make sure that no messages other than RequestLogout reset the session ID. First, we have to define a fuzzer that will be used to generate a list of messages: 



msgFuzzer : Fuzz.Fuzzer Main.Msg
msgFuzzer =
    Fuzz.oneOf <|
        List.map Fuzz.constant
            [ Auth <| ChangePassword "a"
            , Auth <| ChangeUserName "a"
            , Auth <| StartLogin
            , ChangePlanText "{}"
            , CreatePlan
            , RequestLogin
            , RequestRegistration
            , SubmitPlan
            , ToggleMenu
            ]






Fuzz.oneOf takes a list of fuzzers and chooses one of them at random to generate a value. Fuzz.constant always returns the same value, so it’s not a useful fuzzer on its own but it comes handy in situations like this, where we are building a more complicated fuzzer. Unfortunately, there is no way in Elm to enumerate the tags in a custom type, so we have to type them all out (I only included a subset of tags here for brevity). 


Now we can apply this fuzzer in constructing the fuzz test:



suite3 : Test
suite3 =
    let
        ( model, cmd ) =
            init { sessionId = Just "123" }
    in
        describe "Update function"
            [ TestUpdate.fuzz update
                msgFuzzer
                model
                "Should not reset session ID wrongly"
              <|
                \model ->
                    model.appState.auth.sessionId |> Expect.notEqual Nothing
            ]






We pass the initial model into fuzz along with the update function and the message fuzzer, as well as the expectation describing the postcondition that should hold after going through all the messages. This test should pass, but if we included the RequestLogout message in the fuzzer, it would fail. 


Testing views with elm-test



elm-explorations/test includes a set of expectations and helper functions for checking the contents of Html values. They can be used in tests like other expectations. It’s aimed at testing views generated with the help of the html package, so it isn’t very useful for testing the elm-ui based views we’ve been working on. However, we can still look at some examples. 


Suppose that we have a login form view written like this: 



type Msg
    = Login


viewLogin : Html Msg
viewLogin =
    div [ class "login" ]
        [ input [ placeholder "Organisation", value "" ] []
        , input [ placeholder "User name", value "" ] []
        , input
            [ placeholder "Password"
            , type_ "password"
            , value ""
            ]
            []
        , button [ onClick Login ] [ Html.text "Login" ]
        ]






We can test that this function correctly generates the login button in this manner: 



import Test.Html.Query as Query
import Test.Html.Selector as Selector

-- ...

test "Login form has a Login button" <|
    \() ->
        viewLogin
            |> Query.fromHtml
            |> Query.find [ Selector.tag "button" ]
            |> Query.has [ Selector.text "Login" ]






First, we have to transform the Html value using Query.fromHtml. Then, we find the relevant element using Query.find, and finally check that its text is “Login”. The Query module provides a number of other functions as well. 


Another test could ensure that the password field has the type attribute set to “password” (so that it doesn’t accidentally get replaced with a text field):



test "Password field is of the right type" <|
    \() ->
        viewLogin
            |> Query.fromHtml
            |> Query.find [ Selector.attribute <| type_ "password" ]
            |> Query.has [ Selector.tag "input" ]






We can also test that clicking the login button produces the Login message (after importing Test.Html.Event): 



test "Login form generates the message" <|
    \() ->
        viewLogin
            |> Query.fromHtml
            |> Query.find [ tag "button" ]
            |> Event.simulate Event.click
            |> Event.expect Login






In this case, we have to find the button first (with Query.find), then simulate a click event on it with Event.simulate Event.click, and then verify that it results in the Login message. The Event module provides functions for simulating a number of other interactions. 



elm-ui explain attribute


You can add the Element.explain attribute to any element to make elm-ui render the border for the element and its immediate children (it doesn’t traverse children recursively). This can be helpful when debugging layout issues. 


In order to make sure that you don’t ship code with this attribute to production, it takes a Todo argument (produced by Debug.todo). So you would write:



row [ Element.explain Debug.todo ] rowElements






By sprinkling a few of these attributes in PlanTree.elm, we can see how elm-ui lays out the plan tree: 




  
    [image: Plan tree with visible element borders]
    Plan tree with visible element borders
  




Time Travelling Debugger


The last thing I would like to introduce in this chapter is Elm’s time travelling debugger. It’s currently sparsely documented and thus may not be easy to discover. If you compile your program in debug mode by passing the --debug flag to elm make, and then open it in the browser, you will see the debugger panel in the bottom right of the browser window. 




  
    [image: Debugger panel in the bottom right corner]
    Debugger panel in the bottom right corner
  




When you open this panel, you see the complete history of messages since page load along with the corresponding model values: 




  
    [image: Message history and model in the debugger]
    Message history and model in the debugger
  




You can click on any of the recorded messages and see both the model and the UI as it was at that point, which is extremely useful for reproducing bugs. Predictably, keeping track of all this information has memory and performance implications, so you wouldn’t normally use a debug build in production.  


The debugger also allows you to export and import the history of messages, which is great if you want to show somebody else what steps led to a bug. No need to write down the steps, just reproduce the bug and send them the exported history! In addition, after reproducing a bug you can change the code to add a fix for it, and then import and replay the exported messages using the new code, as long as those messages are still valid. This is great for verifying that the bug is fixed.


If you change the set of messages and Elm encounters an invalid message, it will tell you that it’s unable to replay the history.


Another thing to note is that none of the side effects will happen when you import history. This is because the messages are recorded along with the data they carry (for example, the session ID or the list of saved plans sent by the server in response to requests), and the model is reconstructed from that data. 


Summary


In this chapter, we looked at various ways of preventing bugs and making your code more robust. You learned about several language tools that can be used to prevent bugs: type aliases, custom types, as well as leveraging the type system by structuring types and function signatures to make certain bugs impossible. You got introduced to several test libraries available for testing different aspects of Elm code. Finally, Elm’s time travelling debugger got an honourable mention.








Other Options at the Elm/JavaScript Boundary


Overview


In this chapter, we will talk about a few additional techniques for managing the boundary between Elm code and JavaScript code. There are many types of applications where you don’t need to think about this, but you may find these approaches useful later on in your Elm journey, perhaps in the context of larger projects. 


We will discuss the approaches for integrating Elm into an existing JavaScript application, either with the goal of eventually replacing JavaScript, or just to implement a certain piece of functionality. In addition, you will see an example of a different mechanism for managing the interaction with JavaScript by means of custom elements.


Integration into an Existing JavaScript Application


Elm makes it quite easy to embed an Elm application into a larger JavaScript application. There are two scenarios where you might want to do it. One is if you want to use Elm for a specific part of your application, and the other one is if you are migrating JavaScript code to Elm but you don’t want to rewrite it all at once. 


As far as migration goes, the gradual approach is probably best in most cases because it’s a lot less risky. As for replacing parts of JavaScript with Elm, you might want to use Elm to handle complex logic, for example.  


There are a few options for selecting which parts of the application to implement in Elm:



  	If you have a multi-page application, then you could replace specific pages with Elm applications

  	You could have one or more Elm applications embedded into a single page 

  	You could use Elm to implement business logic, but manage the UI in JavaScript. 




Let’s look at these options in a bit more detail. 


Replacing a page


Aside from the fact that there are other pages without Elm, this is equivalent to the setup we’ve been using in this book. You could get elm make to generate an HTML page or just the JavaScript that you then include in your own HTML. 


Embedding Elm


While building the sample application in this book, we relied on Browser.document which takes over the body of the HTML document. It’s also possible for an Elm application to take over a single element instead. To do that, we can use either Browser.sandbox or Browser.element. 


Browser.sandbox is meant for simple programs that don’t need to interact with the outside world via flags, commands or subscriptions, while Browser.element is quite similar to Browser.document and allows the use of all these features. 


Let’s see how we can embed a simple counter into a page. The Elm code can be something like this:



module Main1 exposing (main)

import Browser
import Html exposing (Html, button, div, h1, text)
import Html.Attributes exposing (style)
import Html.Events exposing (onClick)


type alias Model =
    { count : Int }


initialModel : Model
initialModel =
    { count = 0 }


type Msg
    = Increment
    | Decrement


update : Msg -> Model -> Model
update msg model =
    case msg of
        Increment ->
            { model | count = model.count + 1 }

        Decrement ->
            { model | count = model.count - 1 }


view : Model -> Html Msg
view model =
    div
        [ style "border" "1px solid #c0c0c0"
        , style "margin" "10px"
        , style "padding" "10px"
        ]
        [ h1 [] [ text "Browser.sandbox" ]
        , button [ onClick Increment ] [ text "+1" ]
        , div [ style "font-size" "x-large", style "color" "#0055ee" ]
            [ text <| String.fromInt model.count ]
        , button [ onClick Decrement ] [ text "-1" ]
        ]


main : Program () Model Msg
main =
    Browser.sandbox
        { init = initialModel
        , view = view
        , update = update
        }






It needs to be compiled with elm make src/Main1.elm --output elm.js, and then we can use this custom index.html:



<!doctype html>
<html>
<head>
    <title>Elements</title>
    <script src="elm.js"></script>
</head>
<body>
<div id="el1"></div>
<script>
    let app1 = Elm.Main1.init({
        node: document.querySelector("#el1")
    })
</script>
</body>
</html>






Note that we passed a node in Elm.Main1.init arguments to tell the Elm runtime which element to use for the application. You can call init at any time, not just on page load, which makes it possible for you to start your Elm application in response to some user action.


Going further, there’s actually nothing stopping us from embedding more than one Elm application in this fashion. Let’s implement another counter using Browser.element instead of Browser.sandbox: 



module Main2 exposing (main)

import Browser
import Html exposing (Html, button, div, h1, text)
import Html.Attributes exposing (style)
import Html.Events exposing (onClick)


type alias Model =
    { count : Int }


type alias Flags =
    { count : Int }


init : Flags -> ( Model, Cmd Msg )
init flags =
    ( { count = flags.count }, Cmd.none )


type Msg
    = Increment
    | Decrement


update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        Increment ->
            ( { model | count = model.count + 1 }, Cmd.none )

        Decrement ->
            ( { model | count = model.count - 1 }, Cmd.none )


view : Model -> Html Msg
view model =
    div
        [ style "border" "1px solid #c0c0c0"
        , style "margin" "10px"
        , style "padding" "10px"
        ]
        [ h1 [] [ text "Browser.element" ]
        , button [ onClick Increment ] [ text "+1" ]
        , div [ style "font-size" "x-large", style "color" "#0055ee" ]
            [ text <| String.fromInt model.count ]
        , button [ onClick Decrement ] [ text "-1" ]
        ]


subscriptions : Model -> Sub Msg
subscriptions _ =
    Sub.none


main : Program Flags Model Msg
main =
    Browser.element
        { init = init
        , view = view
        , update = update
        , subscriptions = subscriptions
        }






Browser.element expects init to receive a Flags argument, which we are using to pass the initial value of the counter from JavaScript. Let’s alter index.html to embed both of these counters into the same page: 



<!doctype html>
<html>
<head>
    <title>Elements</title>
    <script src="elm.js"></script>
</head>

<body>
<div id="el1"></div>
<div id="el2"></div>
<script>
    let app1 = Elm.Main1.init({
        node: document.querySelector("#el1")
    })
    let app2 = Elm.Main2.init({
        flags: {count: 123},
        node: document.querySelector("#el2")
    })
</script>
</body>
</html>






Finally, we need to make sure both of these applications are included in elm.js by passing both modules to elm make:



elm make src/Main1.elm src/Main2.elm --output elm.js






Similarly, if you’re using elm-live, you can generate elm.js with this command: 



elm-live src/Main1.elm src/Main2.elm -- --output elm.js






As a result, only one copy of runtime will be included in elm.js, so there’s no extra overhead in terms of asset size from including more than one application. 


Note that both Main1.elm and Main2.elm in this scenario define a main function – it’s not a problem because the generated JavaScript function names include the module name. By the way, this also shows that the main module doesn’t have to be called Main.elm. 


The resulting page will contain both counters: 




  
    [image: Counters from two modules embedded into two divs on a page]
    Counters from two modules embedded into two divs on a page
  




Using Elm for business logic


What if you don’t want to write your UI code in Elm? Maybe you need particular widgets that are already implemented in JavaScript, or you’re just happy not to replace your existing UI. 


In that case, you can still use Elm for business logic, run your Elm code in headless mode which doesn’t interact with the DOM, and use ports to communicate with the UI portion of the application written in JavaScript. The setup is a little different in this situation, but it’s quite straightforward. First, instead of using one of the Browser module functions like Browser.document or Browser.application, we need to use Platform.worker:



Platform.worker
    { init = init
    , update = update
    , subscriptions = subscriptions
    }






The difference is that this function doesn’t expect a view field in the argument. Elm and JavaScript can communicate using ports. For example, we could get Elm to hash strings for us using the Skinney/murmur3 package:



port module Main3 exposing (main)

import Murmur3


port stringToHash : (String -> msg) -> Sub msg


port sendHash : Int -> Cmd msg


type Msg
    = Receive String


type alias Model =
    { seed : Int }


type alias Flags =
    Int


init : Flags -> ( Model, Cmd Msg )
init flags =
    ( { seed = flags }, Cmd.none )


update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        Receive s ->
            ( model, sendHash <| Murmur3.hashString model.seed s )


subscriptions : Model -> Sub Msg
subscriptions _ =
    stringToHash Receive


main : Program Flags Model Msg
main =
    Platform.worker
        { init = init
        , update = update
        , subscriptions = subscriptions
        }






There are two ports defined in this program: stringToHash is an incoming port for strings, and sendHash is an outgoing port for sending the resulting hash back to JavaScript. We get the seed value for the hash function via flags. In update, we produce a hash in response to a Receive message generated by subscriptions, and immediately send it back to JavaScript through the sendHash port. 


On the JavaScript side, we can use the init function to initialise the program and pass in the seed, as usual. We can then subscribe to the sendHash port, and send a string to hash: 



    let app3 = Elm.Main3.init({
        flags: 1234
    })
        
    app3.ports.sendHash.subscribe(hash => console.log(hash))
    app3.ports.stringToHash.send("String to be hashed")






And that’s it!


Custom Elements


There is one more way of interacting with JavaScript which we haven’t discussed yet. It relies on a fairly recent addition to the web platform called Web Components. It’s useful to know that Web Components are not a single feature but rather a set of four features:



  	Custom HTML elements 

  	Shadow DOM

  	HTML imports

  	HTML templates.




Custom elements are a mechanism for defining new HTML elements as well as customising built-in elements. The internal implementation of custom elements often relies heavily on the Shadow DOM, which allows you to scope HTML and CSS used to build up an element. This is achieved by having special DOM subtrees which are hidden from the regular DOM API functions. A shadow subtree is attached to an element using JavaScript (and that element is called the host). All the HTML and styles that comprise the subtree are scoped to that subtree. Regular DOM-related methods like document.querySelector don’t see any of the shadow DOM subtrees. 


HTML imports are a way of importing an HTML document into another HTML document: 



<link rel="import" href="custom-element.html">






custom-element.html referred to in this import can contain script tags, HTML templates and styles. HTML imports are only implemented in Google Chrome, and my understanding is that even there, they will eventually be replaced with something else, so I wouldn’t recommend using this feature.


Finally, HTML templates are defined inside the <template> tag. It’s not rendered in the DOM, but you can get its contents from JavaScript. Inside a template, you can use the <slot> tag to define placeholders for user-supplied content. Additionally, <style> blocks inside a template are scoped to the document fragments created from the template. 


Going back to custom elements, they provide a two-way communication mechanism between Elm and JavaScript. We can use properties and attributes to alter their state, and they can produce custom events which we can handle in Elm to change the model, create commands and so on. Thus, they can be a good alternative to ports when it comes to integrating complex JavaScript components into an Elm application. 


A custom element for a rich editor


To explore how this mechanism works in practice, let’s upgrade the input textarea in our sample application to a rich JSON editor to provide the users with a nicer experience. We will use the popular Ace editor which provides syntax highlighting and folding, among other things. As you would expect, a rich editor involves quite a lot of JavaScript and state management. We’ll define a custom element which is going to encapsulate all of that. 


I will omit most of the details pertaining to how the custom element is implemented as it’s beyond the scope of this book. The important part for us is this: given that we have a script which will register the custom element for the editor when included in the page, how do we work with this custom element in Elm code? 


First of all, let’s include the ace-editor.js script in index.html along with a couple of necessary polyfills:



<script src="js/custom-elements.min.js"></script>
<script src="js/shadydom.min.js"></script>
<script src="js/ace-editor.js" type="module"></script>






The implementation I have in js/ace-editor.js happens to be an ES6 module, hence the type="module" attribute. For Chrome, that’s all that’s needed, but in order to make the custom element work in Safari and Firefox, we also need to include polyfills for the custom element API and shadow DOM. These polyfills can be found at the Web Components web site and also on NPM, in the @webcomponents namespace. 


Polyfills also work for Microsoft Edge, and possibly even Internet Explorer 11 (although you might run into issues with components written using ES6 features or relying on other unsupported features). Keep in mind that polyfills do have a performance penalty, so you may need to benchmark custom elements before choosing them over ports. 


I only needed two polyfills, so I pulled them out from the NPM packages for simplicity.


Currently, inputPage looks like this: 



inputPage : AppState -> Element Msg
inputPage appState =
    column
        [ width (px 800)
        , spacingXY 0 10
        , centerX
        ]
        [ Input.multiline
            [ height (px 300)
            , Border.width 1
            , Border.rounded 3
            , Border.color lightCharcoal
            , padding 3
            ]
            { onChange = ChangePlanText
            , text = appState.currPlanText
            , placeholder = Nothing
            , label =
                Input.labelAbove [] <|
                    text "Paste the EXPLAIN output in JSON format:"
            , spellcheck = False
            }
        , Input.button
            (Attr.greenButton ++ [ width (px 200), height (px 40), alignRight ])
            { onPress = Just SubmitPlan
            , label = el [ centerX ] <| text "Go!"
            }
        ]






Let’s replace Input.multiline with the Ace editor:



import Html.Attributes as HtmlAttr exposing (attribute)

-- ...

inputPage : AppState -> Element Msg
inputPage appState =
    column
        [ width (px 800)
        , spacingXY 0 10
        , centerX
        ]
        [ text "Paste the EXPLAIN output in JSON format:"
        , el [ width fill, height fill ] <|
            html <|
                Html.node "ace-editor"
                    [ attribute "mode" "ace/mode/json"
                    , attribute "wrapmode" "true"
                    , HtmlAttr.style "height" "300px"
                    ]
                    []
        , Input.button
            (Attr.greenButton ++ [ width (px 200), height (px 40), alignRight ])
            { onPress = Just SubmitPlan
            , label = el [ centerX ] <| text "Go!"
            }
        ]






The html function from elm-ui allows us to insert an Html value amongst Element values. We used Html.node for an HTML tag with a custom name, in this case ace-editor (note that custom elements are required to have a dash in the tag name). We’ve also specified a couple of Ace-specific attributes: mode and wrapmode. mode specifies the language we want to work with, and setting wrapmode to true makes the editor wrap long lines. 


If we inspect the DOM, we will see something like this:



<ace-editor class="el none__3402792823" wrapmode="true" mode="ace/mode/json" 
style="pointer-events: auto; display: block; margin: 5px 0px; flex-shrink: 0; 
width: calc(100% - 0px); height: 300px; position: relative; left: 0px; 
top: 0px; box-sizing: border-box;"></ace-editor>






If we also make the Shadow DOM visible, we’ll see that the custom element code contains a bunch of other HTML elements and styles:




  
    [image: Shadow DOM nodes for ace-editor]
    Shadow DOM nodes for ace-editor
  




The attributes we specified are handled by JavaScript in ace-editor.js in this fashion:



let editor = ace.edit(container, options)
let session = editor.getSession()
session.setMode(this.getAttribute("mode"))
session.setUseWrapMode(this.hasAttribute("wrapmode"))






There is nothing complicated going on; the code retrieves attribute values and then uses the Ace API to apply them to the editor session. There is some additional code to handle attribute changes which I haven’t shown here. 


So far we’ve done enough to have the editor appear on the page:




  
    [image: Ace editor with highlighting]
    Ace editor with highlighting
  




However, the input in the editor is not going to be reflected in the Elm model at this point because we don’t have anything to generate the ChangePlanText message. 


We know that the custom element code generates change events:



editor.getSession().on("change", (event) => {
    this.textContent = this.editor.getSession().getValue()
    // ...
    element.dispatchEvent(new CustomEvent("change", 
        {bubbles: true, composed: true, detail: event}))
})






So we need to add a handler for these events and set currPlanText in the model, which we can do like this:



inputPage : AppState -> Element Msg
inputPage appState =
    let
        onChange =
            Json.Decode.string
                |> Json.Decode.at [ "target", "editorValue" ]
                |> Json.Decode.map ChangePlanText
                |> Html.Events.on "change"
    in
    column
        [ width (px 800)
        , spacingXY 0 10
        , centerX
        ]
        [ text "Paste the EXPLAIN output in JSON format:"
        , el [ width fill, height fill, htmlAttribute onChange ] <|
            html <|
                Html.node "ace-editor"
                    [ attribute "mode" "ace/mode/json"
                    , attribute "wrapmode" "true"
                    , HtmlAttr.style "height" "300px"
                    ]
                    []
        , Input.button
            (Attr.greenButton ++ [ width (px 200), height (px 40), alignRight ])
            { onPress = Just SubmitPlan
            , label = el [ centerX ] <| text "Go!"
            }
        ]






Note that we’ve added htmlAttribute onChange to the attributes of the element wrapping the custom element tag. In onChange, we supply Html.Events.on "change" with a JSON decoder which will extract the value of target.editorValue from the event data for us, and then wrap it in ChangePlanText. editorValue is a property which has a getter defined in the custom element class: 



window.customElements.define("ace-editor", 
    class AceEditor extends HTMLElement {
        get editorValue() {
            return this.textContent
        }
    
        // ...






This property is kept in sync with the text of the current Ace editor session, so it provides us whatever the user typed or pasted into the editor. So whenever the contents of the editor are changed, our Elm program will get the contents of the editor in a ChangePlanText message.


And that’s it! With this change in place, we’ve restored the original functionality (with improved user experience). The text from the editor is now fed into the model, and we can use it to generate the plan tree. 


Summary


In this chapter, we discussed additional approaches for managing the boundary between Elm code and JavaScript, including scenarios where Elm gradually replaces existing JavaScript and where it coexists with JavaScript. You also learned about interacting with JavaScript via custom elements. 


This is the end of the book. We covered a lot of ground, starting with many features of the language such as functions, records, pattern matching and destructuring. 


We looked at a number of tools such as Ellie, create-elm-app and elm-analyse. 


You learned about the different options available for implementing user interfaces. I demonstrated how to use the elm-ui package to create UI, and how to write complex JSON decoders, which is the number one difficulty encountered by people coming to Elm. 


We looked at performing asynchronous tasks and interacting with the JavaScript and the browser. We considered how code can be structured as your application grows, and how Elm code can coexist with JavaScript code. 


We also examined various techniques for dealing with bugs, from leveraging the type system to using the time travelling debugger. 


I hope that this book helped you become more comfortable with using Elm and working in the Elm ecosystem, and that you can use this knowledge to implement projects successfully, and to be more confident in choosing Elm as a viable alternative to JavaScript for front end implementations.

